

DANMARKS TEKNISKE UNIVERSITET



---

## Bachelor Projekt

---

FUNKTIONEL MODELLERING AF MATEMATISKE SYSTEMER I F#

Jonas Dahl Larsen (s205829)

21. maj 2024

---

# Indhold

<b>1</b>	<b>Introduktion</b>	<b>1</b>
<b>2</b>	<b>Fundamentale koncepter</b>	<b>3</b>
2.1	Introduktion til Funktions Programmering . . . . .	3
2.2	Typer . . . . .	4
2.2.1	Induktivt definerede typer . . . . .	5
2.3	Property Based Testing . . . . .	5
2.4	Overskrivning af operatorer . . . . .	6
2.5	Signaturfiler og implementeringsfiler . . . . .	6
<b>3</b>	<b>Symbolske udtryk</b>	<b>9</b>
3.1	Tal mængder . . . . .	9
3.1.1	Rationelle tal modul . . . . .	9
3.1.2	Komplekse tal-modul . . . . .	10
3.1.3	Tal modulet . . . . .	11
3.2	Matematiske udtryk . . . . .	14
3.2.1	Polsk notation . . . . .	14
3.2.2	Udtryk som træer . . . . .	14
3.2.3	Udtryksmodulet . . . . .	15
3.3	Evaluering af udtryk . . . . .	17
3.4	Konvertering mellem udtryksformer . . . . .	18
3.5	Reducering af udtryk . . . . .	19
3.6	Differentiering af udtryk . . . . .	22
3.7	Multivariable polynomier af første grad . . . . .	22
<b>4</b>	<b>Vektorer og Matricer</b>	<b>25</b>
4.1	Matrix operationer . . . . .	25
4.2	Matematiske operationer . . . . .	26
4.2.1	Skalering af en matrix . . . . .	26
4.2.2	Addition af matricer . . . . .	26
4.2.3	Matrix produkt . . . . .	27
4.2.4	Projektion af en vektor . . . . .	29
4.3	Gram-Schmidt . . . . .	30
4.4	Række-echelon form . . . . .	31
4.5	Lineært ligningssystem . . . . .	32
<b>5</b>	<b>PBT af programmet</b>	<b>35</b>
5.1	PBT af udtryk . . . . .	35
5.1.1	Tal modulet . . . . .	37
5.1.2	Homomorfisme af evaluering . . . . .	39
5.1.3	Simplifikation af udtryk . . . . .	39
5.1.4	Invers morfisme mellem infix og prefix . . . . .	40
5.1.5	Differentiering af udtryk . . . . .	41
5.2	PBT af vektorer og matricer . . . . .	43
5.2.1	PBT af matrix operationer . . . . .	43

---

5.2.2	PBT af Gram-Schmidt . . . . .	44
<b>6</b>	<b>Diskussion</b>	<b>48</b>
<b>7</b>	<b>Konklusion</b>	<b>51</b>
<b>8</b>	<b>Appendiks</b>	<b>53</b>
8.1	complex.fsi . . . . .	53
8.2	TreeGenerator.fs . . . . .	53
8.3	CommutativeAddSub.fs . . . . .	55
8.4	CommutativeMulDiv.fs . . . . .	57
8.5	Matrix.fs . . . . .	59

---

# 1 Introduktion

I 2023 besluttede Danmarks Tekniske Universitet (DTU) at anvende Python som et hjælpemiddel, der erstattede det matematiske softwareprogram Maple<sup>1</sup> i deres grundlæggende matematikkurser „01001 Matematik 1a“ og „01002 Matematik 1b“. I en tid, hvor programmeringssprog som Python dominerer i tekniske og videnskabelige miljøer<sup>2</sup>, undersøger dette projekt fordele og ulemper ved funktionel programmering i matematiske sammenhænge. Funktionel programmering introducerer et andet perspektiv og alternative metoder, som kan berige og muligvis forbedre de studerendes forståelse af matematiske koncepter.

Projektet har til formål at belyse, hvorvidt funktionel programmering, specifikt gennem F#<sup>3</sup>, kan anvendes til at opbygge og manipulere matematiske udtryk og systemer, der introduceres i den indledende matematiske undervisning. Ved at introducere læserne til grundlæggende såvel som avancerede funktioner og teknikker i F#, vil rapporten guide dem gennem opbygningen af funktionel modellering af matematiske systemer, der kan løse matematiske problemer fra de grundlæggende matematikkurser på DTU.

Rapporten vil først og fremmest dykke ned i konstruktionen af et specifikt modul for håndtering af symbolske matematiske udtryk og matrixmanipulering, samt deres anvendelser i forskellige matematiske kontekster. Projektets struktur og metodologi har til formål at give læseren en dybdegående forståelse af, hvordan funktionel programmering kan benyttes strategisk i matematiske discipliner, og hvordan det adskiller sig fra mere traditionelle imperative programmeringstilgange.

Gennem en systematisk tilgang til design og implementering af matematiske moduler vil rapporten udforske, hvordan matematiske principper kan integreres direkte i softwareudvikling gennem funktionel programmering. Dette vil ikke kun fremme en bedre forståelse af teoretiske koncepter og konstruktioner gennem praktisk anvendelse, men også demonstrere F#'s kapacitet og effektivitet i behandlingen af matematiske egenskaber.

---

<sup>1</sup>[10] *Maple*.

<sup>2</sup>[15] *Python: The most popular language*.

<sup>3</sup>[4] *fsharp.org*.

---

---

## 2 Fundamentale koncepter

Det forventes, at læseren har kendskab til programmering. Der gives derfor kun en kort beskrivelse af syntaks og notation, så læsere, der ikke er bekendt med F#, kan forstå de eksempler, der løbende vil forekomme i rapporten.

### 2.1 Introduktion til Funktions Programmering

F# er en del af ML (Meta Language) familien af funktionsprogrammeringssprog, som bygger på „typed lambda calculus“<sup>4</sup>. „Typed lambda calculus“ er en udvidelse af „lambda calculus“<sup>5</sup>, som er et formelt system til at udtrykke beregninger.

Vi begynder med at betragte funktionen for fakultet Ligning (1).

$$f(n) = \begin{cases} 1 & n = 0 \\ n \cdot f(n-1) & n > 0 \\ \text{undefined} & n < 0 \end{cases} \quad (1)$$

Et eksempel på en implementering af Ligning (1) i F# er givet i Listing 1.

```
1 // Fakultet i F#
2 let rec factorial n =
3     match n with
4     | 0          -> 1
5     | x when x > 0 -> x * factorial (x - 1)
6     | _         -> failwith "Negative argument"
```

Listing 1: Eksempel på Fakultet i F#

I F# anvendes `let` til at definere en ny variabel eller, i dette tilfælde, en funktion kaldet `factorial`. Næste nøgleord er `rec`, hvilket indikerer, at funktionen er rekursiv. Funktionen tager et argument `n`, og i linje 3 starter et match-udtryk. Her er `n` vores udtryk, og efter `with` begynder en række mønstre. Med tilhørende udtryk, der svare til de tre tilfælde i Ligning (1).

I F#, er det som udgangspunkt ikke nødvendigt at anvende parenteser som i andre programmeringssprog. Derfor vil de kun blive anvendt, hvor det er passende gennem rapporten, typisk i sammenhænge med kædning af funktioner. For at undgå brugen af parenteser kan man i F# benytte pipe-operatorerne, `|>` og `<|`, som fører resultatet fra en udledning direkte ind i den næste funktion. Nedenstående eksempel viser tre ækvivalente udtryk, der demonstrerer anvendelsen af disse operatører.

```
> factorial (factorial 3);;
val it: int = 720

> factorial <| factorial 3;;
val it: int = 720

> factorial 3 |> factorial;;
```

---

<sup>4</sup>[22] *Typed lambda calculus*.

<sup>5</sup>[9] *lambda calculus*.

---

```
val it: int = 720
```

Listing 2: Eksempel på anvendelse af pipe-operatorer i F# ved udregning af  $(3!)! = 6! = 720$ .

Vi kan beskrive evalueringen af udtrykket i Listing 2 som følgende, hvor  $e_1 \rightsquigarrow e_2$  betyder, at  $e_1$  evalueres til  $e_2$ :

```
factorial(factorial(3))
  ~> factorial(3 × factorial(3 − 1))
  ~> factorial(3 × 2 × factorial(2 − 1))
  ~> factorial(3 × 2 × 1 × factorial(1 − 1))
  ~> factorial(3 × 2 × 1 × 1)
  ~> factorial(6)
  ~> 6 × factorial(6 − 1)
  ~> 6 × 5 × factorial(5 − 1)
  ~> 6 × 5 × 4 × factorial(4 − 1)
  ~> 6 × 5 × 4 × 3 × factorial(3 − 1)
  ~> 6 × 5 × 4 × 3 × 2 × factorial(2 − 1)
  ~> 6 × 5 × 4 × 3 × 2 × 1 × factorial(1 − 1)
  ~> 6 × 5 × 4 × 3 × 2 × 1 × 1
  ~> 720
```

## 2.2 Typer

I F# har alle udtryk, inklusiv funktioner, en defineret type. Typen for funktionen i Listing 1 er  $int \rightarrow int$ . Det betyder, at det ikke er muligt at kalde funktionen med et argument, der ikke er af typen  $int$ .

I matematik kan vi beskrive funktioner som afbildninger ved brug af domæner og co-domæner på følgende måde:

$$\begin{aligned} f &: \mathbb{R} \rightarrow \mathbb{R} \\ g &: \mathbb{R}^n \rightarrow \mathbb{R} \end{aligned}$$

De tilsvarende funktioner i F# vil have typerne:

$$\begin{aligned} f &: float \rightarrow float \\ g &: float * float * \dots * float \rightarrow float \end{aligned}$$

Hvor typekonstruktøren „\*“ svarer til det kartesiske produkt, og refereres til som en tuple. I begge tilfælde skal senere definitioner og funktionsanvendelser overholde disse beskrivelser.

En tuple, der kun består af to typer, kaldes et par. Givet en funktion  $f : T_1 \rightarrow T_2 \rightarrow T_3$ , betyder dette, at den tager et udtryk af typen  $T_1$ , som giver en funktion af typen  $T_2 \rightarrow T_3$ , hvor evalueringen af funktionen resulterer i  $T_3$ .

Hvis en funktion kaldes med et argument, der ikke matcher funktionens type, genereres en fejlmeddelelse. For F#'s vedkommende kontrolleres dette på oversættelsestidspunktet, hvilket er en af de store forskelle mellem F# og Python, hvor typerne først kontrolleres under kørslen.

---

### 2.2.1 Induktivt definerede typer

I F# defineres lister og træer induktivt ved brug af rekursive typer. Et eksempel på en rekursivt defineret type for et træ er givet i Listing 3.

```
1 type Tree =  
2   | Leaf of int  
3   | Node of Tree * Tree
```

Listing 3: Eksempel på en rekursivt defineret type for et træ i F#

Rekursive typer lagres som en værdi og ikke en datastruktur.

### 2.3 Property Based Testing

Property Based Testing (PBT) er en teknik til at teste korrekthed af egenskaber som man ved altid skal være opfyldt. Ved PBT genereres en række tilfældige input til en funktion, hvorefter det kontrolleres, om en given egenskab holder. Fokuset ved PBT er de fundamentale egenskaber, som en funktion skal overholde, og ikke de specifikke tilfælde som ved eksempelvis unit tests<sup>6</sup>.

På DTU introduceres de studerende til logik, som det første emne i „01001 Matematik 1a“. Her lærer de at en udsagnslogisk formel er gyldig (en tautologi), hvis den altid er sand. Der findes mange teknikker til at påvise gyldigheden af en udsagnslogisk formel. I „01001 Matematik 1a“ anvendes blandt andet sandhedstabeller, som demonstrerer gyldigheden af en formel. Eksempelvis vises hvordan Formel (2) er gyldig.

$$P \wedge (Q \wedge Y) \iff (P \wedge Q) \wedge Y \quad (2)$$

Vi kan også bruge PBT til at undersøge, om Formel (2) holder, ved at definere egenskaben som en funktion af  $P, Q$  og  $Y$ , som vist i Listing 4.

```
1 #r "nuget: FsCheck"  
2 open FsCheck  
3  
4 // proposition formula: bool -> bool -> bool -> bool  
5 let propositional_formula P Q Y =  
6   (P && ( Q && Y)) = ((P && Q) && Y)  
7  
8 let propositional_formula_invalid P Q Y =  
9   P && ( Q && Y) = (P && Q) && Y
```

Listing 4: PBT af Formel (2). Begge sider skal være omgivet af parenteser, for at den er gyldig, da „=“ har en højere præcedens end „&&“

```
> let _ = Check.Quick propositional_formula;;  
Ok, passed 100 tests.
```

Listing 5: Output ved PBT af (2)

---

<sup>6</sup>[23] *Unit Testing*.



---

„Check.Quick“ er en del af „FsCheck“-biblioteket. Den tager en funktion som argument og genererer en række tilfældige input til funktionen baseret på funktionens type. Hvis funktionen returnerer „true“ for alle input, vil testen lykkes. Hvis funktionen returnerer „false“ for et input, vil testen fejle og give et eksempel på et input der fejlede. I Formel 4 er „Check.Quick“ anvendt til at teste, om Formel (2) er gyldig. Funktionen „Check.Quick“ returnerer „Ok, passed 100 tests,“ hvilket indikerer, at (2) er gyldig. Det er vigtigt at forstå, at dette ikke er det samme som at bevise, at Formel (2) er gyldig, da testen ikke garanterer, at alle muligheder er testet. Generelt vil der være flere kombinationer af argumenter til en funktion, end de 100 test cases som genereres.

Vi kan også teste den ikke gyldige funktion `propositional_formula_invalid`, som vist i Listing 6, er den ikke gyldig og vi får givet et eksempel på en input kombination der fejlede.

```
> let _ = Check.Quick propositional_formula_invalid;;
Falsifiable, after 2 tests (0 shrinks) (StdGen (68724885, 297333127)):
Original:
false
true
false
```

Listing 6: Output ved PBT af `propositional_formula_invalid`

I nogle tilfælde vil det være en fordel at opskrive en PBT før implementeringen af en funktion, som man ved skal overholde en egenskab. På den måde anvendes Test-Driven Development (TDD)<sup>7</sup> til at teste, om ens egenskab forbliver overholdt under implementeringen. I dette projekt vil vi anvende PBT til at validere, at de matematiske egenskaber bliver overholdt af programmet.

## 2.4 Overskrivning af operatorer

I F# er det muligt at overskrive standardoperatorer, så de kan anvendes på egne typer. Denne teknik vil blive benyttet igennem rapporten til at definere matematiske operationer for de typer, vi udvikler. Et eksempel på overskrivning af operatorer er givet i næste sektion.

## 2.5 Signaturfiler og implementeringsfiler

En implementeringsfil i F# er lavet med .fs extension. Implementeringsfilen kan have en tilhørende signaturfil med .fsi extension. Denne fil indeholder en beskrivelse af de typer og funktioner i implementeringsfilen, som er tilgængelige for andre filer. En signaturfil kan derfor bruges som en arbejdstegning for andre, der ønsker at anvende eller replicere implementeringsfilen. Til sidst er der scriptfiler med .fsx extension, som kan bruges til at køre F# kode uden at skulle kompilere den.

Som et eksempel på de forskellige filtyper kan vi se signaturfilen i Listing 7.

```
1 module MyInt
2
3 type MyInt = I of int
4 with
5     static member (*) : MyInt * MyInt -> MyInt
6     static member (-) : MyInt * MyInt -> MyInt
```

---

<sup>7</sup>[20] *Test-Driven Development*.

```
7
8 val factorial : MyInt -> MyInt
```

Listing 7: Eksempel på en signaturfil *MyInt.fsi*

Ud fra signaturfilen kan vi se, at implementeringsfilen i Listing 8 indeholder en type *MyInt*, hvor der er defineret overskrivning af operatorerne multiplication og subtraction. Derudover er der defineret en funktion *factorial*.

```
1 module MyInt
2 type MyInt = I of int
3
4 type MyInt with
5     static member (*) (I a, I b) = I (a * b)
6     static member (-) (I a, I b) = I (a - b)
7
8 let rec factorial (n:MyInt) =
9     match n with
10    | I 0 -> I 1
11    | I x when x > 0 -> n * factorial (n - I 1)
12    | _ -> failwith "Negative argument"
```

Listing 8: Eksempel på en implementeringsfil *MyInt.fs*

I denne implementeringsfil finder vi overskrivningerne af operatorerne på typen *MyInt*. Derudover er der også en implementering af en funktion *factorial*, som virker med typen *MyInt*.

Til sidst kan vi åbne modulet i en scriptfil, som vist i Listing 9.

```
1 #r "../bin/Release/net7.0/main.dll"
2 open MyInt
3 let a = I 3
4 factorial a |> printfn "%A"
```

Listing 9: Eksempel på en scriptfil *MyInt.fsx*

Vi kan derefter køre scriptfilen og få outputtet, som vist i Listing 10.

```
I 720
```

Listing 10: Output ved kørsel af *MyInt.fsx*

---

---

## 3 Symbolske udtryk

Det ønskes at kunne repræsentere simple udtryk som en type i F#. Vi skal derfor gennemgå en del teori og funktioner som er nødvendige for at kunne dette. Det vil give os et grundlæggende fundament for at kunne udføre matematiske evalueringer som differentiering i F#. Som de fleste andre programmer har F# kun float og int som kan repræsentere tal. Derfor vil vi begynde med at definere et modul som indeholder en type for tal. Tankegangen her at gennemgå en opbygning af en måde at kunne repræsentere udtryk samt reducere dem. Vi begrænset os selv til at kun have matematiske operationer som addition, subtraktion, negation, multiplikation og division.

### 3.1 Tal mængder

Vi begynder med opbygningen af et modul, der kan repræsentere en talmængde. Typen for tal består af tre konstruktører, henholdsvis for heltal, rationale tal og komplekse tal hvor real- og imaginærdelen er rationale.

#### 3.1.1 Rationelle tal modul

Repræsentationen af rationale tal kan laves ved hjælp af at danne et par af heltal, hvor det ene heltal er tælleren, og det andet er nævneren.

```
1 type rational = R of int * int
```

Listing 11: Typen for rationelle tal

Nedenfor er der givet en signaturfil for rational modulet 12. I implementeringsfilen overskrives de matematiske operatører ved hjælp af de klassiske regneregler for brøker<sup>8</sup>.

```
1 module rational
2
3
4 type rational = R of int * int
5 with
6     static member ( ~- ) : rational -> rational
7     static member ( + ) : rational * rational -> rational
8     static member ( + ) : int * rational -> rational
9     static member ( - ) : rational * rational -> rational
10    static member ( - ) : int * rational -> rational
11    static member ( * ) : int * rational -> rational
12    static member ( * ) : rational * int -> rational
13    static member ( * ) : rational * rational -> rational
14    static member ( / ) : rational * rational -> rational
15    static member ( / ) : int * rational -> rational
16    static member ( / ) : rational * int -> rational
17    static member ( / ) : int * int -> rational
18
19 val newRational      : int * int -> rational
20 val equal             : rational * rational -> bool
21 val positive         : rational -> bool
22 val toString         : rational -> string
23 val isZero           : rational -> bool
24 val isOne            : rational -> bool
```

---

<sup>8</sup>[17] *Rational Number wikipedia.*

---

```

25 val isInt      : rational -> bool
26 val makeInt    : rational -> int
27 val greaterThan : rational * rational -> bool
28 val isNegative  : rational -> bool
29 val absRational : rational -> rational

```

Listing 12: Signaturfilen for rational-modulet

For at kunne sammenligne og også for nemmere at undgå for store brøker, vil alle rationelle tal blive reduceret til deres simplest form. Dette gøres ved at finde den største fælles divisor (GCD)<sup>9</sup>. Derudover er det vigtigt at være opmærksom på ikke at foretage nul division. Derfor vil implementeringsfilen kaste en „System.DivideByZeroException“, hvis nævneren er eller bliver nul. Signaturfilen indeholder en række funktioner, som anvendes af andre filer. Det vil desuden være nødvendigt at kunne håndtere overflow, idet heltallene, der repræsenterer de rationelle tal, under eller efter operationen kan blive for store til korrekt at blive repræsenteret af 32-bit. Da denne rapport fokuserer på implementeringen af matematiske koncepters og ikke numeriske algoritmer, vil modulet blot slå en fejl, hvis der opstår overflow.

### 3.1.2 Komplekse tal-modul

Vi skal nu dykke lidt mere ned i implementeringen af et modul for komplekse tal. Signaturfilen *complex.fsi* givet i Appendiks 8.1. Først defineres en type for komplekse tal, som består af et rationelt tal par, henholdsvis for realdelen og imaginærdelen, se Listing 13.

```

1 type complex = C of rational * rational

```

Listing 13: Typen for komplekse tal

Der kan opskrives en række regneregler i Definition 1 for operationer på komplekse tal.

**Definition 1** (Regneregler for komplekse tal).

Lad  $a, b, c, d \in \mathbb{Q}$  så er følgende defineret omkring komplekse tal<sup>10</sup>:

1. **Addition**

$$(a + bi) + (c + di) = (a + c) + (b + d)i$$

2. **Subtraktion**

$$(a + bi) - (c + di) = (a - c) + (b - d)i$$

3. **Multiplikation**

$$(a + bi) \cdot (c + di) = (ac - bd) + (bc + ad)i$$

4. **Kvadratisk form**

$$(a + bi) \cdot (a - bi) = a^2 + b^2$$

5. **Konjugering**

$$\overline{a + bi} = a - bi$$

6. **Division**

$$\frac{a+bi}{c+di} = \frac{(a+bi) \cdot (c-di)}{c^2+d^2}$$

---

<sup>9</sup>[6] *Greatest common divisor wikipedia.*

<sup>10</sup>[11] *Mathematics 1a*, se. Definition 3.3 s. 54, Definition 3.5 s. 56, Definition 3.8 s. 57, ligning 3-2 s. 58.

Ved implementeringen, se Listing 14, af addition, subtraktion, multiplikation samt skalering med et rationelt tal. Skaleringen af et rationelt tal, kan udledes fra multiplikation, ved at lade den imaginære del være 0. Disse er simple operationer, som ikke behøver at defineres i særskilte funktioner. De kan defineres direkte på overskrivningen af deres respektive operatorer. Dog er det nødvendigt at definere multiplikation som en funktion, da den skal anvendes af divisionsfunktionen.

```

1 // complexDivRational: complex -> rational -> complex
2 let complexDivRational c (n) =
3     match c with
4     | _ when isZero n -> raise (System.DivideByZeroException("Complex.
5         divRational: Cannot divide by zero!"))
6     | C (a, b) -> C (a / n, b / n)
7
8 // mulConjugate: complex -> rational
9 let mulConjugate (C(a, b)) = a*a + b*b
10
11 // conjugate: complex -> complex
12 let conjugate (C (a, b)) = C (a, -b)
13
14 // mulComplex: complex -> complex -> complex
15 let mulComplex (C (a, b)) (C (c, d)) = C(a*c-b*d, b*c+a*d)
16
17 // divComplex: complex -> complex -> complex
18 let divComplex z1 z2 =
19     complexDivRational (mulComplex z1 (conjugate z2)) <| mulConjugate z2
20
21 type complex with
22     static member (+) (C(a, b), C(c, d)) = C(a + c, b + d)
23     static member (-) (C(a, b), C(c, d)) = C(a - c, b - d)
24     static member (*) (n, C(a, b)) = C(n * a, n * b)
25     static member (*) (C(a, b), n) = C(n * a, n * b)
26     static member (*) (z1, z2) = mulComplex z1 z2
27     static member (/) (z, n) = complexDivRational z n
28     static member (/) (z1, z2) = divComplex z1 z2
29     static member (~-) (C(a, b)) = C(-a, -b)

```

Listing 14: Overskrivning af operationer på komplekse tal

Bortset fra division af to komplekse tal, ligner de resulterende overskrivninger på operationerne deres respektive matematiske definitioner. Men når vi nærmere studerer divisionen af to komplekse tal, ser vi, at der blot er brug for få funktioner til at kunne udføre divisionen. Først konjugeres nævneren, derefter multiplikeres resultatet med tælleren. Til sidst divideres resultatet af multiplikationen med kvadratet af nævneren. Da komplekse tal som en talmængde indeholder heltal og rationale tal, vil vi i det følgende afsnit omkring tal modulet anvende komplekse tal til udføre de matematiske operationer i modulet.

### 3.1.3 Tal modulet

Vi har nu beskrevet en måde at kunne repræsentere bruger definere tal på ved brug af typer i F#. Det vil derfor være oplagt at have en type som indeholder alle de typer tal vi ønsker at kunne anvende i de matematiske udtryk vi er ved at opbygge. Fordelen ved at samle dem til en type er at vi kan lave en række funktioner blandt andet matematiske operationer som kan anvendes på alle type tal. Ved at samle dem til en type „Number“ kan vi også opskrive en række egenskaber i 1 som vi ønsker at de skal opfylde. Egenskaberne vil blive valideret i sektion 5.1.1.

---

**Egenskab 1** (Egenskaber for Number-typen).

Lad  $a, b, c \in \text{Number}$ , så gælder følgende<sup>11</sup>:

1. *Addition og multiplikation er **associative***  
 $a + (b + c) = (a + b) + c$  og  $a \cdot (b \cdot c) = (a \cdot b) \cdot c$
2. *Addition og multiplikation er **kommutive***  
 $a + b = b + a$  og  $a \cdot b = b \cdot a$
3. ***Distributivitet** af multiplikation over addition*  
 $a \cdot (b + c) = a \cdot b + a \cdot c$
4. *Addition og multiplikation har et **neutralt element***  
 $a + 0 = a$  og  $a \cdot 1 = a$
5. ***Omvendt funktion** eksisterer til addition*  
 $a + (-a) = 0$
6. ***Omvendt funktion** eksisterer til multiplikation for  $a \in \text{Number} \setminus \{0\}$*   
 $a \cdot a^{-1} = 1$

Vi begynder med at definere typen for tal i Listing 15, som indeholder konstruktører for de tal typer vi har definerede samt en for heltal.

```
1  type Number = | Int of int | Rational of rational | Complex of complex
```

Listing 15: Typen for Number

Typen Number er designet med henblik på, at den kan udvides med flere taltyper. Ved udvidelse er det et krav til den nye talmængde, at der er definerede matematiske operationer i form af addition, subtraktion, negation, multiplikation og division. Den nye Number type skal derudover forsat opfylde egenskaberne 1. En udvidelse kunne være for reelle tal, som kan håndtere „floating point errors“<sup>12</sup>, men for at undgå at komplicere programmet yderligere vil vi i denne opgave ikke inkludere decimal tal.

Betragtes signatur filen for Number modulet i Listing 16, ses det at der igen er defineret overskrivning af de anvendte matematiske operationer. Derudover er der defineret en række funktioner som kan anvendes på Number typen.

```
1  module Number
2  open rational
3  open complex
4
5  type Number =
6    | Int of int
7    | Rational of rational
8    | Complex of complex
9  with
10     static member ( + ) : Number * Number -> Number
11     static member ( - ) : Number * Number -> Number
12     static member ( * ) : Number * Number -> Number
13     static member ( / ) : Number * Number -> Number
14     static member ( ~- ) : Number -> Number
```

<sup>11</sup>[11] *Mathematics 1a*, se. Side 124 Definition 6.1.

<sup>12</sup>[3] *Floating-point error*.

---

```

15
16 val zero      : Number
17 val one       : Number
18 val two       : Number
19 val isZero    : Number -> bool
20 val isOne     : Number -> bool
21 val isNegative : Number -> bool
22 val absNumber : Number -> Number
23 val greaterThan : Number -> Number -> bool
24 val tryReduce  : Number -> Number
25 val toString  : Number -> string
26 val conjugate  : Number -> Number
27 val inv       : Number -> Number
28 val isInt     : Number -> bool

```

Listing 16: Signatur filen for Number modulet

Ved implementeringen af de matematiske operationer, hvis der eksisterer en konstruktør i `Number`, der repræsenterer en tal mængde hvor alle andre konstruktører er delmængder af denne mængde. Er det muligt at definere en enkelt funktion som kan udføre alle binære operationer. Som et eksempel er funktionen `operation` i Listing 17 givet, som tager to tal og en funktion i form af den ønskede binære operation som argumenter. Funktionen konverterer de to tal til vores komplekse tal type, hvortil den binære operation udføres på dem.

```

1 // makeComplex: Number -> complex
2 let makeComplex n =
3     match n with
4     | Int x -> newComplex (newRational(x, 1), newRational(0, 1))
5     | Rational x -> newComplex (x, newRational(0, 1))
6     | Complex x -> x
7
8 // operation: Number -> Number -> (complex -> complex -> complex) -> Number
9 let operation a b f =
10     f (makeComplex a) (makeComplex b) |> Complex

```

Listing 17: operation funktionen til udførelse af matematiske operationer

Det vil her til være oplagt, at have en funktion til at reducere tallet efter den matematiske operation, til den simpleste tal type. Den simpleste tal typer er i vores tilfælde heltal. Dette er gjort ved at anvende funktionen `tryMakeInt` på alle de matematiske operationers overskrivninger i Listing 18.

```

1 // tryMakeInt: Number -> Number
2 let tryMakeInt r =
3     match r with
4     | Rational a when isInt a -> Int (makeRatInt a)
5     | _ -> r
6
7 type Number with
8     static member (+) (a, b) = operation a b (+) |> tryMakeInt
9     static member (-) (a, b) = operation a b (-) |> tryMakeInt
10    static member (*) (a, b) = operation a b (*) |> tryMakeInt
11    static member (/) (a, b) = operation a b (/) |> tryMakeInt
12    static member (~-) (a) = neg a |> tryMakeInt

```

Listing 18: Overloadnings funktionerne for Number



---

Ved at have implementeret de matematiske operationer på denne måde, er det blandt andet muligt at fortage heltales division, da det vil resultere i et rationalt tal.

Dermed har vi et modul som kan repræsentere tal, samt udføre matematiske operationer på dem. Vi vil nu begynde at betragte hvordan vi kan anvende den i et matematisk udtryk.

## 3.2 Matematiske udtryk


### 3.2.1 Polsk notation

Matematiske udtryk som vi normalt kender dem er skrevet med infix notation. I infix notation skrives en binær operator mellem to operandere, kende tegnet for sproget er at det indeholder parenteser samt præcedens regler. Dette gør det generelt kompliceret at evaluere og håndtere matematiske udtryk i et programmeringssprog. Derfor er det mere oplagt at kunne anvende polsk notation (prefix) istedet, hvor operatoren skrives før operandere eller omvendt polsk notation (postfix). Da de hverken indeholder parenteser eller præcedens regler<sup>13</sup>. Følgende er der givet et simpelt eksempel på de 3 notationer.

Infix Notation:  $(A + B) \cdot C$   
Prefix Notation:  $\cdot + ABC$   
Postfix Notation:  $AB + C \cdot$

### 3.2.2 Udtryk som træer

Et matematisk udtryk kan repræsenteres som et binært træ, hvor bladene er operander i det anvendte matematiske rum og alle andre noder er operationer. Som eksempel kan udtrykket  $-(3 \cdot x - 5) \cdot 2$  repræsenteres som følgende træ i Figur 1.




Figur 1: Et binært træ der repræsenterer udtrykket  $-(3 \cdot x - 5) \cdot 2$

---

<sup>13</sup>[13] *Polsk notation wikipedia*.

Det skal bemærkes der er forskel på den unære og binære operator '-' i træet, den unære betyder negation og den binære er subtraktion. Givet et binært træ for et matematisk udtryk, vil det være muligt omdanne dem til infix, prefix eller postfix notation. Dette kan gøres ved at anvende modificeret Inorder, Preorder eller Postorder Traversal<sup>14</sup>, algorithmerne er illustreret i Figur 2.



Modificeret Inorder Traversal:  $-(3 \cdot x - 5) \cdot 2$

Preorder Traversal:  $\cdot - - \cdot 3 x 5 2$

Postorder Traversal:  $3 x \cdot 5 - - 2 \cdot$

Figur 2: Træet fra Figur 1 med forskellige travesal metoder

Vi vil i 3.2.3 betragte hvordan vi kan implementere et modul som kan repræsentere udtryk ved brug af prefix notation. Postorder Traversal bliver blandt andet anvendt til at kunne rekursivt reducere og evaluere udtryk.

Grundet præcedens regler i infix notation, er det nødvendigt at modificere Inorder Traversal, da unære noder altid skal håndteres før dens børn. Desuden vil det også være nødvendigt at implementere regler for at håndtere parenteser, hvis der ønskes et symbolsk udtryk. Den modificeret Inorder Traversal anvendes til at kunne visualisere udtrykket i infix notation.

### 3.2.3 Udtryksmodulet

Efter udviklingen af et modul til repræsentation af talmængder er vi nu klar til at udvide programmet med et modul for matematiske udtryk. Vi starter med at definere en polymorf type for udtryk, givet i Listing 19. Denne type omfatter flere konstruktører, hver tilknyttet specifikke

<sup>14</sup>[21] *Tree Traversal Techniques – Data Structure and Algorithm Tutorials*.

matematiske operationer vi ønsker at implementere. Desuden introducerer vi konstruktøren `N` til at repræsentere numeriske værdier ved at anvende talmængder defineret i Listing 15. Til sidst tilføjer vi konstruktøren `X` for variable. Således lagres matematiske udtryk i en rekursiv datatype, der skal repræsentere udtryks træerne i sektion 3.2.2. Dette skyldes at, hver konstruktør for en operation indeholder et eller to underudtryk af samme type.

```

1 type Expr<'a> =
2   | X of char
3   | N of 'a
4   | Neg of Expr<'a>
5   | Add of Expr<'a> * Expr<'a>
6   | Sub of Expr<'a> * Expr<'a>
7   | Mul of Expr<'a> * Expr<'a>
8   | Div of Expr<'a> * Expr<'a>

```

Listing 19: Typen for Expr

`Expr<'a>` typen er dermed en polymorfisk type, hvor `'a` er typen for den tal mængde hvor vi kan lave brugerdefinerede matematiske operationer. Et eksemplar på en `Expr<Number>` er givet i Listing 20. Her ses det at når udtrykket  $-(3 \cdot x - 5) \cdot 2$  visualiseres er det i prefix notation.

```

> tree "-(3*x-5)*2";;
val it: Expr<Number> =
  Mul (Neg (Sub (Mul (N (Int 3), X 'x'), N (Int 5))), N (Int 2))

```

Listing 20:  $-(3 \cdot x - 5) \cdot 2$  som et udtryks træ. Funktionen `tree` bliver beskrevet i 3.4.

Signatur filen indeholder overskrivninger på de matematiske operationer, så de kan anvendes på udtryk. Samt en funktion `eval` til at evaluere et udtryk beskrevet i 3.3.

```

1 module Expression
2 open Number
3
4 type Expr<'a> =
5   | X of char
6   | N of 'a
7   | Neg of Expr<'a>
8   | Add of Expr<'a> * Expr<'a>
9   | Sub of Expr<'a> * Expr<'a>
10  | Mul of Expr<'a> * Expr<'a>
11  | Div of Expr<'a> * Expr<'a>
12 with
13   static member ( ~- ) : Expr<Number> -> Expr<Number>
14   static member ( + ) : Expr<Number> * Expr<Number> -> Expr<Number>
15   static member ( - ) : Expr<Number> * Expr<Number> -> Expr<Number>
16   static member ( * ) : Expr<Number> * Expr<Number> -> Expr<Number>
17   static member ( / ) : Expr<Number> * Expr<Number> -> Expr<Number>
18
19 val eval : Expr<Number> -> Map<char,Number> -> Number
20 val isAdd : (Expr<Number> -> Expr<Number> -> Expr<Number>) -> bool
21 val isSub : (Expr<Number> -> Expr<Number> -> Expr<Number>) -> bool
22 val isMul : (Expr<Number> -> Expr<Number> -> Expr<Number>) -> bool
23 val isDiv : (Expr<Number> -> Expr<Number> -> Expr<Number>) -> bool
24 val isNeg : (Expr<Number> -> Expr<Number>) -> bool
25 val isZero : Expr<Number> -> bool
26 val containsX : Expr<Number> -> Expr<Number> -> bool

```

---

```

27 val getNumber : Expr<Number> -> Number
28 val getVariable : Expr<Number> -> char

```

Listing 21: Signatur filen for Expression modulet

De overskrevne matematiske operatører i implementeringsfilen for udtryk laver overflade evalueringer samt reduceringer på deres respektive argumenter. Overfalde evaluering vil sige at de individuelle funktioner kun betragter de to øverste niveauer på de udtryks træer de tager som input, implementeringerne af addition og multiplikation er givet i Listing 22. Lignende funktioner er implementeret for de andre operationer.

```

1 // mul: Expr<Number> -> Expr<Number> -> Expr<Number>
2 let rec mul e1 e2:Expr<Number> =
3   match e1, e2 with
4   | N a, N b                -> N (a * b)
5   | N a, b | b, N a when isOne a -> b
6   | N a, _ | _, N a when isZero a -> N zero
7   | Div(a, b), c | c, Div(a, b) -> Div (mul a c, b)
8   | Div(a, b), Div(c, d)        -> Div ((mul a c), (mul b d))
9   | Neg a, Neg b                -> mul a b
10  | _, _                       -> Mul(e1, e2)
11
12 // add: Expr<Number> -> Expr<Number> -> Expr<Number>
13 let rec add e1 e2:Expr<Number> =
14   match e1, e2 with
15   | N a, N b                -> N (a + b)
16   | N a, b | b, N a when isZero a -> b
17   | a, b when a = b         -> Mul (N two, b)
18   | Neg a, Neg b            -> neg (add a b)
19   | Neg a, b | b, Neg a     -> Sub (b, a)
20   | Mul(a, X b), Mul(c, X d) -> Mul(X b, a), Mul(c, X d)
21   | Mul(X b, a), Mul(c, X d) -> Mul(a, X b), Mul(X d, c)
22   | Mul(a, X b), Mul(X d, c) -> Mul(X b, a), Mul(X d, c)
23   | Mul(X b, a), Mul(X d, c) when b = d -> Mul(add a c, X b)
24   | _, _                     -> Add(e1, e2)

```

Listing 22: Addition og multiplikation af to udtryk

### 3.3 Evaluering af udtryk

Vi vil nu betragte hvordan vi kan evaluere et udtryk, ved hjælp af et miljø som indeholder værdier for variable som er indeholdt i udtrykket. Evalueringen af udtryk skal kunne opfylde følgende homomorfiske egenskaber 2. Egenskaben vil blive valideret i sektion 5.1.2.

**Egenskab 2** (Homomorfisme af evaluering).

Lad  $\oplus \in \{+, -, \times, /\}$  sættet af binære operationer,  $e1$  og  $e2$  være udtryk, så gælder følgende:

$$eval(e1 \oplus e2) = eval(e1) \oplus eval(e2)$$

Derudover skal det om negation også gælde at:

$$eval(-e) = -eval(e)$$

Funktionen `eval` i Listing 23 tager et udtryk og et miljø som input og evaluere udtrykket til en numerisk værdi. Funktionen kører en Postorder Traversal på udtrykket og evaluerer dermed udtrykket nedefra og op, ved at foretage matematiske operationer defineret i `Number` modulet.

```

1 // eval: Expr<Number> -> Map<char, Number> -> Number
2 let rec eval (e:Expr<Number>) (env) =
3   match e with
4   | X x -> Map.find x env
5   | N n -> n
6   | Neg a -> - eval a env
7   | Add (a, b) -> eval a env + eval b env
8   | Sub (a, b) -> eval a env - eval b env
9   | Mul (a, b) -> eval a env * eval b env
10  | Div (a, b) -> eval a env / eval b env

```

Listing 23: Evaluering af et udtryk

### 3.4 Konvertering mellem udtryksformer

Det er ønsket at kunne konvertere udtryk frem og tilbage mellem prefix notation, repræsenteret af `Expression`-typen, og den standard infix notation. Dette ønske skyldes, at infix notation er lettere for os at læse og skrive. Derfor er det essentielt, at de to konverteringsfunktioner fungerer som hinandens inverse. Dette krav er yderligere uddybet i egenskab 3. Egenskaben bliver valideret i sektion 5.1.4.

**Egenskab 3** (Invers morphism<sup>15</sup> mellem infix og prefix).

Lad  $Q^n$  være mængden af rationelle infix udtryk repræsenteret som en string, med  $n$  variable, så defineres følgende:

$$\begin{aligned} tree &: Q^n \rightarrow Expr \\ tree^{-1} &: Expr \rightarrow Q^n \end{aligned}$$

Dermed gælder følgende egenskaber

$$\begin{aligned} tree^{-1} \circ tree &= id_{Q^n} \\ tree \circ tree^{-1} &= id_{Expr} \end{aligned}$$

Hvor  $id_x$  er identitetsfunktionen på mængden  $x$ .

Vi begynder med at betragte den inverse funktion, som konverterer et udtryk til infix notation. Funktionen `etf` i Listing 24 fortager denne konvertering ved at lave den modificeret Inorder Traversal på udtrykket, som blev beskrevet i 3.2.2. Den modificeret del er at håndtere parentes samt negation, som var det en binær node i træet hvor det venstre barn er et tomt udtryk.

```

1 // parenthesis: bool -> string -> string
2 let parenthesis b f = if b then "(" + f + ")" else f
3
4 // etf: Expr<Number> -> bool -> string
5 let rec etf e p =
6   match e with

```

<sup>15</sup>[7] Inverse function.

```

7 | N a when not <| isInt a -> parenthesis p <| toString a
8 | N a -> toString a
9 | X a -> string a
10 | Neg a -> parenthesis p <| "-" + etf a (not p)
11 | Add(a, b) -> parenthesis p <| etf a false + "+" + etf b false
12 | Sub(a, b) -> parenthesis p <| etf a false + "-" + etf b true
13 | Mul(a, b) -> parenthesis p <| etf a true + "*" + etf b true
14 | Div(a, b) -> parenthesis p <| etf a true + "/" + etf b true
15
16
17 // infixExpression: Expr<Number> -> string
18 let infixExpression e = etf e false

```

Listing 24: konvertering fra expression til infix notation

Funktionen `tree`, som foretager konverteringen fra infix notation til et udtrykstræ, er baseret på algoritmen beskrevet i [1]<sup>16</sup>. Først konverteres en udtryksstreng til en liste af tokens. Disse tokens beskriver, om en karakter i udtrykket er en operand, en operator, eller en konstant, hvor en operator også indeholder information om præcedens og associativitet<sup>17</sup>. Typen for disse tokens kan ses i Listing 25. Herefter anvendes to stacks: én for operatore og én for udtryk. Der anvendes en række regler, som beskrevet i [1], for hvornår der skal udføres pop og push på disse to stacks. Det skal bemærkes, at når en operator pushes til udtryksstacken, da navnet på operatorkonstruktøren på udtrykket står skrevet fra venstre mod højre, vil udtryksstacken være i prefix notation og ikke postfix notation som beskrevet i kilden. Funktionen `tree` er at finde i Appendix 8.2.

```

1 type Associative = | Left | Right
2 type Precedence = int
3 type Operator = char * Precedence * Associative
4 type Token =
5   | Operand of char
6   | Operator of Operator
7   | Constant of int
8 type OperatorList = Operator list

```

Listing 25: Konvertering fra infix til udtrykstræ

### 3.5 Reducering af udtryk

Vi skal nu betragte en systematisk metode til at kunne reducere matematiske udtryk, ved hjælp af simple algebraiske regler. Dette er en nødvendighed at kunne for at bruge udtrykkene i en matematisk sammenhæng, da det vil kunne medføre både en reduktion i kompleksitet og en forbedring i læsbarhed når udtrykkene visualiseres. Før vi betragter metoden, kan vi opskrive en egenskab som reduceringen skal overholde. Egenskaben vil blive valideret i sektion 5.1.3, det er en nødvendighed at evaluere udtrykket før og efter reduktion, da det er en kompleks opgave at skulle sammenligne om to udtryk er ækvivalente.

#### Egenskab 4 (Reducering af udtryk).

Lad  $e$  være et udtryk, så gælder følgende:

$$eval(e) = eval(simplifyExpr(e))$$

<sup>16</sup>[1] Convert Infix expression to Postfix expression.

<sup>17</sup>[12] Operator Precedence and Associativity in C.

Vi begynder med at betragte funktionen `simplifyExpr` i Listing 26, som reducerer et udtryk ved at foretage en Postorder Traversal på udtrykket. Hvilket betyder når en node i udtrykstræet skal reduceres, vil dens børn allerede være reduceret.


```

1 //simplifyOperation: Expr<Number> -> Expr<Number> -> (Expr<Number> -> Expr<
    Number> -> Expr<Number>) -> Expr<Number>
2 let rec simplifyOperation e1 e2 f =
3   match f e1 e2 with
4   | Neg a ->
5     commutativeMulDiv.applyCommutative (Neg a) |> commutativeAddSub.
      applyCommutative
6   | Add(a, b) when isAdd f -> commutativeAddSub.applyCommutative (Add(a, b))
7   | Sub(a, b) when isSub f -> commutativeAddSub.applyCommutative (Sub(a, b))
8   | Mul(a, b) when isMul f -> commutativeMulDiv.applyCommutative (Mul(a, b))
9   | Div(a, b) when isDiv f -> commutativeMulDiv.applyCommutative (Div(a, b))
10  | Add(a, b) -> simplifyOperation a b (+)
11  | Sub(a, b) -> simplifyOperation a b (-)
12  | Mul(a, b) -> simplifyOperation a b (*)
13  | Div(a, b) -> simplifyOperation a b (/)
14  | a -> a
15
16 //simplifyExpr: Expr<Number> -> Expr<Number>
17 let rec simplifyExpr e =
18   match e with
19   | N a when Number.isNegative a -> Neg (N (Number.absNumber a))
20   | N (Rational(R(a, b))) ->
21     simplifyOperation (simplifyExpr (N (Int a))) (simplifyExpr (N (Int b)))
22     (/)
23   | Neg a -> - (simplifyExpr a)
24   | Add(a, b) -> simplifyOperation (simplifyExpr a) (simplifyExpr b) (+)
25   | Sub(a, b) -> simplifyOperation (simplifyExpr a) (simplifyExpr b) (-)
26   | Mul(a, b) -> simplifyOperation (simplifyExpr a) (simplifyExpr b) (*)
27   | Div(a, b) -> simplifyOperation (simplifyExpr a) (simplifyExpr b) (/)
28   | _ -> e


```

Listing 26: Reducering af et udtryk

Det er `simplifyOperation`, som foretager selve reduceringen af et givet udtryk. Funktionen tager som argumenter to udtryk samt den binære operation, der skal anvendes på disse. Funktionen anvendes på udtrykkene, hvorefter overfladisk reducering, som blev beskrevet i 3.2.3, udføres. Hvis overfladisk reducering resulterer i en ændring af den anvendte operation, kalder funktionen sig selv rekursivt med de to udtryk og den nye operation. Hvis overfladisk reducering ikke resulterer i ændringer i operationen, vil funktionen foretage en dybere reducering af de to udtryk. Denne dybere reducering udføres af funktionerne `applyCommutative` fra filerne *CommutativeAddSub.fs* og *CommutativeMulDiv.fs*, som findes i Appendiks 8.3 og 8.4. Disse funktioner arbejder efter samme princip, hvor de starter med at flade udtrykstræerne ud ifølge de kommutative regler for henholdsvis addition og multiplikation. Derefter sorterer de udtrykstræet, hvilket muliggør at foretage overfladisk når ved at gendanne træet. For multiplikation anvendes samme metode i nævneren for division, og der undersøges, om der er fælles udtryk i det udfladede træ, som fremtræder i nævneren af en division og i det udfladede træ, der indeholder divisionen. Et eksempel på anvendelse af den kommutative multiplikationsreducering på et udtryk er givet i Figur 3, som viser det visuelle input og det resulterende svar fra funktionen, samt Figur 4, der viser det udfladede træ og sorteringen af det fladede træ.



Figur 3: Før og efter reducere af et udtryk ved brug af `CommutativeMulDiv.applyCommutative`



Figur 4: Det udfladede og sorterede af udtrykket  $(3 \cdot y \cdot (-2) / -x) \cdot x$  i processen af kommutativ reducere

Generelt, når det gælder reducere af udtryk, skal man være opmærksom på ikke at ende i et uendeligt loop. Derfor er det vigtigt ikke at definere nogle overfladiske reducere, som er hinandens inverse funktioner. Desuden forsøger funktionerne i dette program altid at skubbe negation af udtryk så langt ud som muligt i håb om, at de kan ophæve hinanden. Dette illustreres blandt andet i figur 4, hvor ved udfladning af træet, hvis et af de kommutative udtryk for multiplikation er negativt, fjernes negationen, og der tilføjes i stedet en negation af tallet 1, som ved sortering skubbes til venstre.



### 3.6 Differentiering af udtryk

Vi kan nu betragte den første implementering, der benytter vores udtryksmodul, som samtidig vil understrege vigtigheden af at kunne reducere udtryk. Vi begynder igen med at opskrive nogle algebraiske linearitetsegenskaber, som differentieringen skal overholde. Disse egenskaber vil blive testet i sektion 5.1.5.

**Egenskab 5** (Linearitetsbetingelserne<sup>18</sup>).

Lad  $f$  og  $g$  være udtryk, og  $a$  være tal fra talmodulet, så gælder følgende:

1. **Skaleringsreglen**

$$\frac{d}{dx}(af) = a \frac{df}{dx}$$

2. **Sumreglen**

$$\frac{d}{dx}(f + g) = \frac{df}{dx} + \frac{dg}{dx}$$

3. **Subtraktionsreglen**

$$\frac{d}{dx}(f - g) = \frac{df}{dx} - \frac{dg}{dx}$$

4. **Produktreglen**

$$\frac{d}{dx}(fg) = \frac{df}{dx}g + f\frac{dg}{dx}$$

5. **Kvotientreglen**

$$\frac{d}{dx}\left(\frac{f}{g}\right) = \frac{\frac{df}{dx}g - f\frac{dg}{dx}}{g^2}$$

Funktionen for differentiering `diff` i Listing 27, som tager et udtryk og en variabel som input og differentierer udtrykket med hensyn til variabelen. Dette er en af de store fordele ved at anvende et funktionelt programmeringssprog, da det ses, hvordan fire af reglerne fra egenskab 5 er implementeret direkte, som de er beskrevet.

```
1 // diff: Expr<Number> -> char -> Expr<Number>
2 let rec diff e dx =
3     match e with
4     | X f when f = dx -> N (Int 1)
5     | X _ -> N (Int 0)
6     | N _ -> N (Int 0)
7     | Neg f -> diff (Mul (N (Int -1), f)) dx
8     | Add(f, g) -> Add(diff f dx, diff g dx)
9     | Sub(f, g) -> Sub(diff f dx, diff g dx)
10    | Mul(f, g) -> Add(Mul(diff f dx, g), Mul(f, diff g dx))
11    | Div(f, g) -> Div(Sub(Mul(diff f dx, g), Mul(f, diff g dx)), Mul(g, g))
```

Listing 27: Differentiering af et udtryk

### 3.7 Multivariable polynomier af første grad

Da vi senere i projektet skal betragte matricer, vil vi i den forbindelse også lave en løsning af lineære ligningssystemer i sektion 4.5. Det kræver derfor, at vi kan isolere variable i et multivariable polynomium af første grad. Vi betragter derfor nu to simple funktioner til at udføre denne isolation, se Listing 28. Funktionen `isolateX` undersøger først, om den variabel, som skal isoleres, befinder sig på højre eller venstre side, derefter kaldes funktionen `expressionOnX`, som fungerer

<sup>18</sup>[2] *Differentiation Rules*.

ved at evaluere til en funktion, der giver den omvendte operation af den operation, som variabelen er involveret i. Dertil anvendes den omvendte funktion på begge sider af ligningen, hvor hvis variabelen er isoleret, gives et udtrykspar, hvor det første udtryk er den isolerede variabel. Hvis variabelen ikke er isoleret, kaldes funktionen rekursivt med de nye højre og venstre sider.

```

1 // expressionOnX: Expr<'a> -> Expr<'a> -> (Expr<'a> -> Expr<'a>)
2 let rec expressionOnX hs x =
3   match hs with
4   | N _ | X _ -> fun e -> e
5   | Neg(a) when a = x -> fun e -> Neg e
6   | Sub(a, b) when a = x -> fun e -> Add(e, b)
7   | Div(a, b) when a = x -> fun e -> Mul(e, b)
8   | Div(_, a) when a = x -> fun e -> Mul(e, a)
9   | Mul(a, b) | Mul(b, a) when a = x -> fun e -> Div(e, b)
10  | Add(a, b) | Add(b, a) | Sub(b, a) when a = x -> fun e -> Sub(e, b)
11  | Add(a, b) | Sub(a, b) | Mul(a, b) | Div(a, b) -> fun e -> expressionOnX a
12  | Neg(a) -> fun e -> expressionOnX a x e
13
14 // isolateX: Expr<Number> -> Expr<Number> -> Expr<Number> -> Expr<Number> *
15 // Expr<Number>
16 let rec isolateX lhs rhs x =
17   let operation =
18     if containsX lhs x
19     then expressionOnX lhs x
20     elif containsX rhs x
21     then expressionOnX rhs x
22     else
23       failwith "Variable not found in either side of the equation"
24   match operation lhs |> simplifyExpr, operation rhs |> simplifyExpr with
25   | a, b | b, a when a = x -> (a, b)
26   | a, b -> isolateX a b x

```

Listing 28: Isolering af variable i et udtryk

Da funktionen kun betragter operationen på den variable, der ønskes isoleret, eksisterer der mange tilfælde, hvor funktionen ikke vil kunne isolere variabelen. Men den fungerer til at løse ligninger af formen  $a_1x_1 + a_2x_2 + \dots + a_nx_n = b$ , hvilket er tilstrækkeligt for vores formål.

---

---

## 4 Vektorer og Matricer

Vi vil nu betragte opbyggelsen af et modul for vektorer og matricer. Eftersom en vektor også kan opfattes som en matrix, vil vi i det følgende, når begge dele omtales, udelukkende referere til matricer. For systematik at kunne håndtere matricer, starter vi med at definere en type for lagringsordning Listing 29.

```
1 type Order = | R | C
```

Listing 29: Typen for order

Typen Order, anvendes til at angive, om en matrix er i rækkefølge (row-major) eller kolonnefølge (column-major)<sup>19</sup>. En vektor, der er lagret i rækkefølge, kan betragtes som den transponeret kolonnefølge vektor. Vi kan derfor nu definere en type for matricer, ved hjælp af en type for vektore i Listing 30.

```
1 type Vector = V of list<Number> * Order
2 type Matrix = M of list<Vector> * Order
```

Listing 30: Typen for Matricer

Derudover er det en fordel at kunne kende dimissionen af en matrix. Derfor er der også defineret en type for dimissionen se Listing 31.

```
1 // Rows x Cols
2 type Dimension = D of int * int
```

Listing 31: Typen for dimissionen

### 4.1 Matrix operationer

Der vil i denne sektion beskrives en række funktioner som er nødvendige før vi kan betragte nogle funktion for anvendelse af matricer. Da modulet indeholder mange hjælpe funktioner, vil der fokuseres på de funktioner med matematisk relevans.

Det muligt at definere en funktion til at finde dimissionen af en matrix se Listing 32. Funktionen laver et kald til `matrixValidMajor` som genere en fejl hvis ikke alle vektorer og matrien har samme lagringsordning. `matrixVectorLength` finder længden på vektor listen en i matricen.

```
1 // dimMatrix : Matrix -> Dimension
2 let dimMatrix (M(vl, o)) =
3   if vl = [] then D (0, 0)
4   else
5     let _ = matrixValidMajor (M(vl, o))
6     let d1 = List.length vl
7     let d2 = matrixVectorLength (M(vl, o))
8     match o with
9     | R -> D (d1, d2)
10    | C -> D (d2, d1)
```

Listing 32: Funktion til at finde dimissionen af en matrix

---

<sup>19</sup>[18] Row- and column-major order.

Hvis en matrix er gemt som rækkefølge, vil antallet af rækker være længden af en vektor og antallet af kolonner være længden af vektor listen, og omvendt for kolonnefølge.

## 4.2 Matematiske operationer

I denne sektion bør det bemærkes, at flere listings ikke inkluderer fejlhåndtering; dette er udeladt for at forbedre læsbarheden. De funktioner, der anvendes i det implementerede modul, har passende fejltjek, herunder dimensionstjek på matricerne. Den fulde implementering med fejlhåndtering kan findes i appendiks 8.5.

Før vi implementere funktioner til at udføre de ønskede matematiske operationer, vil vi først definere nogle egenskaber matricerne skal opfylde i egenskab 6.

### Egenskab 6 (Vektor Aksiomer).

Lad  $c, d \in \mathbb{F}$  og  $v_i \in \mathbb{F}^n$  for  $i = 1 \dots m$  så gælder:<sup>20</sup>

1.  $(v_1 + \dots + v_{m-1}) + v_m = v_1 + (v_2 + \dots + v_m)$
2.  $c \cdot \left( d \cdot \begin{bmatrix} | & & | \\ v_1 & \dots & v_m \\ | & & | \end{bmatrix} \right) = (c \cdot d) \cdot \begin{bmatrix} | & & | \\ v_1 & \dots & v_m \\ | & & | \end{bmatrix}$
3.  $c \cdot (v_1 + \dots + v_m) = c \cdot v_1 + \dots + c \cdot v_m$

### 4.2.1 Skalering af en matrix

Vi begynder med at betragte en funktion til at skalere en matrix se Listing 33.

```
1 // scalarVector : Number -> Vector -> Vector
2 let scalarVector (n: Number) (V (v1, o)) =
3   V ((List.map (fun x -> x * n) v1), o)
4
5 // scalarMatrix : Matrix -> Number -> Matrix
6 let scalarMatrix (M (v1, o)) n =
7   M ((List.map (fun x -> scalarVector n x) v1), o)
```

Listing 33: Funktion til at skalere en matrix

Det at skalere en matrice er svare til at skalere hvert element i matricen. Derfor ved at have en funktion `scalarVector`, der skalere hvert element i en givet vektor bliver `scalarMatrix` at skalere hver vektor i en givet matrice. `List.map` svare til at lave en list comprehension i Python<sup>21</sup>.

### 4.2.2 Addition af matricer

```
1 // addVector : Vector -> Vector -> Vector
2 let addVector (V (v1, o1)) (V (v2, _)) =
3   V ((List.map2 (+) v1 v2), o1)
4
5 // addMatrix : Matrix -> Matrix -> Matrix
6 let addMatrix (M(v11, o)) (M(v12, _)) =
```

<sup>20</sup>[11] *Mathematics 1a*, Theorem 7.2 s. 155.

<sup>21</sup>[14] *Python - List Comprehension*.

---

```

7   M (List.map2 addVector v11 v12, o)
8
9   // subVector : Vector -> Vector -> Vector
10  let subVector x y =
11      scalarVector (-one) y |> addVector x
12
13  // sumRows : Matrix -> Matrix
14  let rec sumRows m =
15      if not <| correctOrderCheck m C
16      then sumRows <| correctOrder m C
17      else
18          let zeroVector = vectorOf zero <| matrixVectorLength m
19          let (M(v1, _)) = m
20          matrix [List.fold (addVector) zeroVector v1]

```

Listing 34: Funktion til at addere matricer og subtraktion af vektorer

Addition af vektorer reduceres til at udføre additionen elementvis, som vist i Listing 34, ved brug af List-funktionen `map2`. Vi kan bruge `addVector` til at definere matrix addition og subtraktion af vektorer. Vektor addition bruges også til at summere rækkerne i en matrix (`sumRows`), hvilket vil blive anvendt i implementeringen af matrix produkt i næste sektion og Gram-Schmidt-processen i sektion 4.3. Funktionen bliver yderligere beskrevet i definition 2.

**Definition 2** (Summering af rækker i en matrix).

Lad  $A$  være en matrix med  $m$  rækker og  $n$  søjler, hvor

$$A = \begin{bmatrix} a_{11} & a_{12} & \cdots & a_{1n} \\ a_{21} & a_{22} & \cdots & a_{2n} \\ \vdots & \vdots & \ddots & \vdots \\ a_{m1} & a_{m2} & \cdots & a_{mn} \end{bmatrix}$$

Så gælder om `sumRows` at

$$\text{sumRows}(A) = v = \begin{bmatrix} \sum_{j=1}^n a_{1j} \\ \sum_{j=1}^n a_{2j} \\ \vdots \\ \sum_{j=1}^n a_{mj} \end{bmatrix}$$

Dermed er  $v_i = \sum_{j=1}^n a_{ij}$  for  $i = 1, 2, \dots, m$ .

### 4.2.3 Matrix produkt

**Definition 3** (Matrix-Vektor Produkt).

Lad  $A$  være en matrix med  $m$  rækker og  $n$  søjler, hvor

$$A = \begin{bmatrix} a_{11} & a_{12} & \cdots & a_{1n} \\ a_{21} & a_{22} & \cdots & a_{2n} \\ \vdots & \vdots & \ddots & \vdots \\ a_{m1} & a_{m2} & \cdots & a_{mn} \end{bmatrix}$$

og  $\mathbf{v} = (v_1, v_2, \dots, v_n)^T$  være en vektor med  $n$  elementer. Så er matrix-vektor  $A\mathbf{v}$  produktet defineret som

$$\begin{bmatrix} a_{11} & a_{12} & \cdots & a_{1n} \\ a_{21} & a_{22} & \cdots & a_{2n} \\ \vdots & \vdots & \ddots & \vdots \\ a_{m1} & a_{m2} & \cdots & a_{mn} \end{bmatrix} \begin{bmatrix} v_1 \\ v_2 \\ \vdots \\ v_n \end{bmatrix} = \begin{bmatrix} a_{11}v_1 + a_{12}v_2 + \cdots + a_{1n}v_n \\ a_{21}v_1 + a_{22}v_2 + \cdots + a_{2n}v_n \\ \vdots \\ a_{m1}v_1 + a_{m2}v_2 + \cdots + a_{mn}v_n \end{bmatrix} = \begin{bmatrix} \sum_{j=1}^n a_{1j}v_j \\ \sum_{j=1}^n a_{2j}v_j \\ \vdots \\ \sum_{j=1}^n a_{mj}v_j \end{bmatrix}$$

**Sætning 1** (Matrix-Vektor Produkt).

Lad  $A$  være en matrix med  $m$  rækker og  $n$  søjler, og lad  $\mathbf{v}$  være en vektor med  $n$  elementer. Så gælder der

$$A\mathbf{v} = \text{sumRows} \left[ \begin{array}{c|c|c|c} & & & \\ \hline a_1v_1 & a_2v_2 & \cdots & a_nv_n \\ \hline & & & \end{array} \right]$$

*Bevis.* Lad  $B$  være resultatet af at skalere søjlerne i matrix  $A$  med de tilsvarende elementer i vektoren  $\mathbf{v}$ . Vi har

$$\begin{aligned} B &= \left[ \begin{array}{c|c|c|c} & & & \\ \hline a_1v_1 & a_2v_2 & \cdots & a_nv_n \\ \hline & & & \end{array} \right] \\ &= \begin{bmatrix} a_{11}v_1 & a_{12}v_2 & \cdots & a_{1n}v_n \\ a_{21}v_1 & a_{22}v_2 & \cdots & a_{2n}v_n \\ \vdots & \vdots & \ddots & \vdots \\ a_{m1}v_1 & a_{m2}v_2 & \cdots & a_{mn}v_n \end{bmatrix} \end{aligned}$$

Ved brug af definition 2 for `sumRows` og definition 3 ses det at

$$\text{sumRows}(B) = \begin{bmatrix} \sum_{j=1}^n a_{1j}v_j \\ \sum_{j=1}^n a_{2j}v_j \\ \vdots \\ \sum_{j=1}^n a_{mj}v_j \end{bmatrix} = A\mathbf{v}$$

□

Vi kan dermed anvende sætning 1 til at definere en funktion `matrixVectorProduct` til matrix-vektor produkt se Listing 35. Denne funktion skalerer først søjlerne i matrixen med de tilsvarende elementer i vektoren og summer derefter rækkerne i matrixen.

```
1 // matrixVectorProduct : Matrix -> Vector -> Matrix
2 let rec matrixVectorProduct (M(vl, _)) (V(nl, _)) =
3   M (List.map2 (fun mc n -> scalarVector n mc) vl nl, C)
4   |> sumRows
```

Listing 35: Funktion til matrix-vektor produkt

---

**Definition 4** (Matrix produkt).

Lad  $A \in \mathbb{F}^{m \times n}$  og  $B \in \mathbb{F}^{n \times \ell}$ . Lad søjlerne i  $B$  er være givet ved  $b_1, \dots, b_\ell \in \mathbb{F}^n$ , dermed<sup>22</sup>

$$B = \begin{bmatrix} | & & | \\ b_1 & \cdots & b_\ell \\ | & & | \end{bmatrix}.$$

Så defineres matrixproduktet som

$$A \cdot B = \begin{bmatrix} | & & | \\ A \cdot b_1 & \cdots & A \cdot b_\ell \\ | & & | \end{bmatrix}.$$

Ud fra definition 4 ses det, at funktionen `matrixProduct` i Listing 36 til at tage produktet af to matricer, ved at udfører matrix-vektor produkt på hver søjle i matricen. Da `matrixVectorProduct` evaluere til en matrix, skal der bruges en funktion til at konvertere matricen til en vektor, hvilket `matrixToVector` gør.

```
1 // matrixProduct : Matrix -> Matrix -> Matrix
2 let rec matrixProduct a (M(vlb, _)) =
3   let product = List.map (
4     fun bv -> matrixVectorProduct a bv |> matrixToVector ) vlb
5   M(product, C)
```

Listing 36: Funktion til at tage produktet af to matricer

#### 4.2.4 Projektion af en vektor

**Definition 5** (Projektion af en vektor).

Projektionen af en vektor på en linje defineres i som følgende, hvor  $Y = \text{span}\{y\}$ <sup>23</sup>.

$$\text{proj}_Y : V \rightarrow V, \quad \text{proj}_Y(x) = \frac{\langle x, y \rangle}{\langle y, y \rangle} y \quad (3)$$

hvor det standard indre produkt er defineret som:

$$\langle x, y \rangle = y^* x = \sum_{k=1}^n x_k \bar{y}_k \quad (4)$$

For at kunne projekte en vektor, som defineret i destination 5 skal vi først kunne konjugere en vektor. Funktionen `conjugateVector` konjugerer elementerne i en vektor. Derudover defineres en funktion til at multiplicere to vektorer elementvis (`vectorMulElementWise`).

```
1 // vectorMulElementWise : Vector -> Vector -> Vector
2 let vectorMulElementWise (V(u, o1)) (V(v, o2)) =
3   V (List.map2 (*) u v, o1)
4
5 // conjugateVector : Vector -> Vector
```

---

<sup>22</sup>[11] *Mathematics 1a*, Definition 7.12 s. 162.

<sup>23</sup>[8] *Mathematics 1b - Functions of several variables*, s. 40.



---

```

6 let conjugateVector (V(v, o)) =
7   V (List.map conjugate v, o)
8
9 // innerProduct : Vector -> Vector -> Number
10 let innerProduct u v =
11   let (V(w, _)) = conjugateVector v |> vectorMulElementWise u
12   List.fold (+) zero w
13
14 // proj : Vector -> Vector -> Vector
15 let proj y x =
16   scalarVector (innerProduct x y / innerProduct y y) y

```

Listing 37: Funktioner til at projicere en vektor på en anden

Evalueringen af det standard indre produkt `innerProduct` mellem to vektore, bliver derfor at konjugere den ene vektor og derefter multiplicere elementvis med den anden vektor. Hvortil summen af elementerne i den resulterende vektor er det indre produkt.

Til sidst kan funktionen `proj` skrives direkte som den er defineret i definition 5.

### 4.3 Gram-Schmidt

Vi kan nu betragte implementeringen af Gram-Schmidt processen. Denne proces kan anvendes rekursivt til at finde en ortonormal basis for et underrum udspændt af en liste af vektorer  $v_1, v_2, \dots, v_n$ . Processen kan implementeres rekursivt idet de nye vektorer  $w_k$  for  $k = 2, 3, \dots, n$  konstrueres baseret på alle de tidligere vektorer  $w_1, \dots, w_{k-1}$ .

Før vi implementerer Gram-Schmidt processen, er vi dog begrænset af vores Number type fra Listing 15, idet  $x \in \{\text{Number}\} \not\Rightarrow \sqrt{x} \in \{\text{Number}\}$ . Derfor vil vi ikke normalisere vektorerne, hvilket medfører, at vi kun vil finde en ortogonal basis, fremfor en ortonormal basis.

Der er også 2 egenskaber 7 som vi i sektion 5.2.2 vil teste for at sikre at vores implementering af Gram-Schmidt processen er korrekt.

#### Egenskab 7 (Gram-Schmidt).

Lad  $w_1, w_2, \dots, w_n$  være de nye vektore som er dannes ud fra gram-schmidt processen på  $v_1, v_2, \dots, v_n$  som er lineært uafhængige vektorer. Så gælder der:

1.  $w_1, w_2, \dots, w_n$  er ortogonale.
2.  $w_1, w_2, \dots, w_n$  udspænder det samme underrum som  $v_1, v_2, \dots, v_n$ . dvs.  
 $\text{span}\{v_1, v_2, \dots, v_n\} = \text{span}\{w_1, w_2, \dots, w_n\}$ .

```

1 // orthogonalBasis : Matrix -> Matrix
2 let orthogonalBasis m =
3   if not <| correctOrderCheck m C
4   then orthogonalBasis (correctOrder m C)
5   else
6
7   // Gram_Schmidt : Matrix -> (Vector list -> Matrix) -> Matrix
8   let rec Gram_Schmidt vm acc_wm =
9     match acc_wm [], vm with
10    | x, M([], _) -> x
11    | M([], _), M(v1::vrest, o) ->
12      Gram_Schmidt (M(vrest, o))

```

```

13   <| fun x -> extendMatrix (M([v1], C)) x
14   | M(w, _), M(vk::vrest, o) ->
15     let (V(wk, _)) = vk - sumProj w vk
16     Gram_Schmidt (M(vrest,o))
17   <| fun x -> extendMatrix (acc_wm wk) x
18
19   // sumProj : Vector list -> Vector -> Vector
20   and sumProj w vk =
21     List.map (fun x -> proj x vk) w
22     |> matrix
23     |> sumRows
24     |> matrixToVector
25
26   Gram_Schmidt m (fun _ -> M([], C))

```

Listing 38: Dannelsen af en ortogonal basis, ved hjælp af Gram-Schmidt processen

Listing 38 viser implementeringen af Gram-Schmidt processen. Lavet ud fra beskrivelse side 45 i 'Mathematics 1b'<sup>24</sup>.

Funktionen `sumProj` tager en liste med vektorer  $w$ , som i Gram-Schmidt-processen er de tidligere behandlede vektorer  $w_1, \dots, w_{k-1}$ , og en vektor  $v_k$  som er den  $k$ 'te vektor.  $v_k$  projiceres på alle vektorerne i  $w$ , hvorefter der tages summen af disse projektioner.

Funktionen `Gram_Schmidt`, tager en matrix hvor søjlerne er de vektore som ønskes at finde en ortogonal basis for. Der udover tager den en akkumulerende funktion som indeholder de behandlede vektorer. Hvis der ikke er flere vektorer i matrixen, gives den akkumulerede funktion. Hvis der ikke er nogle vektorer i akkumulatoren, tages den første vektor fra matrixen og tilføjes til akkumulatoren. Hvis der er vektorer i både akkumulatoren og matrixen, kaldes `sumProj` på den akkumulerede liste og den første vektor i matrixen. Resultatet trækkes fra den første vektor i matrixen, og dette bliver den nye vektor som tilføjes til akkumulatoren.

Funktionen `orthogonalBasis` tager en matrix og tjekker om matrixen er i kolonnefølge, hvis ikke kalder funktionen sig selv, med den korrekte lagringsordning. Ellers kaldes `Gram_Schmidt` med matrixen og en tom akkumulator. Resultatet bliver derfor en matrix med en ortogonal basis for underrummet udspændt af de givne vektorer, givet at vektorerne er lineært uafhængige.

## 4.4 Række-echelon form

I forbindelse med udførelsen af PBT til Gram-Schmidt metoden vil vi blandt andet anvende række-echelon form til at sikre os, at et sæt af vektorer er lineært uafhængige. Derfor vil vi i denne sektion betragte implementeringen af række-echelon form, ud fra „Algorithm 9 for computing a row echelon form of a matrix“ angivet på side 142 af noterne til „01001 Mathematics 1a“<sup>25</sup>. Efter at have udført PBT af Gram-Schmidt metoden, slog den i første omgang fejl, hvilket efter nærmere undersøgelse viste sig at være grundet en mindre fejl i Algorithm 9. Linje 14 burde være „ $b \leftarrow$  the  $j$ 'th entry of the  $i$ -th row of  $B$ “. Listing 39 viser implementeringen af række-echelon form, lavet ud fra pseudokoden „Algorithm 9“.

```

1 // rowEchelonForm : Matrix -> Matrix
2 let rec rowEchelonForm A =

```

<sup>24</sup>[8] *Mathematics 1b - Functions of several variables*, s. 45.

<sup>25</sup>[11] *Mathematics 1a*, s. 142.

```

3   if not <| correctOrderCheck A R then rowEchelonForm (correctOrder A R)
4   else
5   let (D(r, c)) = dimMatrix A
6   match A with
7   | M([], _) -> A
8   | _ when isZeroMatrix A -> A
9   | M(v:_, _) when r = 1 -> firstNonZero v |> inv |> scalarMatrix A
10  | M(_, o) ->
11      let (i, j) = firstNonZeroIndexMatrix A 0 (-1, -1)
12      let (M(B, _)) = swapFirstWith A i
13      let b = List.head B |> firstNonZero
14      let (M(B, _)) = scalarIthVector (inv b) 0 (M(B, o))
15      let R1 = List.head B
16      let R2m = List.tail B
17      let B = rowOps j 1 r R1 (M(R2m, o))
18      let (M(Cm, _)) = rowEchelonForm B
19      M(R1::Cm, o)
20
21  // rowOps: int -> int -> int -> Vector -> Matrix -> Matrix
22  and rowOps column i nrows R1 acc_m =
23      if i >= nrows then acc_m else
24      let Ri = getMatrixIthVector (i - 1) acc_m
25      let b = getVectorIthNumber column Ri
26      rowOps column (i + 1) nrows R1 <| replaceMatrixIthVector (i-1) acc_m (Ri
    - b * R1)

```

Listing 39: Funktion til at finde række-echelon form

## 4.5 Lineært ligningssystem

Som den sidste del i anvendelsen af matricer vil vi beskrive, hvordan vi kan løse et lineært ligningssystem af formen  $Ax = b$ . Ved at foretage række-echelon form på totalmatricen  $[A | b]$  kan vi finde en løsning til ligningssystemet, hvis ranken af  $A$  er forskellig fra ranken af den totale matrix<sup>26</sup>.

Efter at have foretaget række-echelon form på totalmatricen, skal vi tage matrix-vektor-produktet af  $\text{ref}(A)$  og  $x$  for at kunne have  $m$  ligninger med  $n$  ubekendte, hvis vi antager, at  $A \in \mathbb{F}^{m \times n}$ .

Den nemmeste tilgang vil være at introducere to nye typer for at repræsentere en vektor og matrix, som kan indeholde variable. Dette kan gøres ved hjælp af vores type for udtryk fra Listing 19. Vi definerer derfor følgende typer i Listing 40.

```

1 type ExprVector = list<Expr<Number>>
2 type ExprMatrix = list<ExprVector>

```

Listing 40: Typer for at repræsentere en vektor og matrix med variable

Det er her undladt at definere lagringsordningen for matricerne af udtryk, da vi ikke bruger dem til mere end en mellemregning i løsningen af ligningssystemet. Men alle funktioner, vi har lavet i Listing 30, ville man også kunne lave for udtryksmatricer, hvis man anvender en lagringsordning.

Ved at have udtryksmatricer defineret som en liste af udtryk, kan vi anvende den samme teknik til at foretage matrix-vektor-produktet mellem  $\text{ref}(A)$  og  $x$ , som vi benyttede i Listing 35. Dette

<sup>26</sup>[11] *Mathematics 1a*, Corollary 6.27 s. 146.

---

vil resultere i en liste af udtryk, som vi ved hjælp af funktionen `isolateX` i Listing 28 kan bruge til at isolere  $x_i$  i den  $i$ 'te ligning ved at sætte den lig med  $b_i$  og indsætte det i ligningerne 1 til  $i - 1$ .

Den fulde implementering er at finde i appendiks 8.5. Det er værd at bemærke, at denne implementering vil resultere i en vektor af `Number`. Derfor vil den fejle, hvis der er flere ubekendte end ligninger, hvilket betyder, at koefficientmatricen  $A$  skal have fuld rank.

---

---

## 5 PBT af programmet

Vi skal nu validere vores program ved hjælp af PBT, ved at lave funktioner, som kan undersøge de egenskaber der løbende er blevet beskrevet i rapporten.

### 5.1 PBT af udtryk

Før vi begynder at udføre PBT på alle egenskaber vedrørende udtryk, skal vi først bygge en generator for vores talmængde og udtryk. Da vi ikke altid kan sammenligne, om to udtrykstræer er ækvivalente, vil vi i alle vores PBT, hvor vi ønsker at sammenligne træer, gøre dette ved brug af et miljø, som indeholder værdier for variablene, og derefter evaluere udtrykene ved brug af `eval`-funktionen.

Vi begynder med at definere en række generatorer, som kan generere blade i vores udtrykstræer i Listing 41.

```
1 let max = 3
2 let min = -3
3
4 // noneZeroGen: Gen<int>
5 let noneZeroGen =
6   Gen.oneof [
7     Gen.choose(1, max) ;
8     Gen.choose(min, -1)]
9
10 // numberGen: Gen<Number>
11 let numberGen =
12   Gen.oneof [
13     Gen.map2 (fun x y -> newRational(x, y) |> Rational |> tryReduce ) (Gen
14   .choose(min, max)) noneZeroGen;
15     Gen.map (fun x -> Int x) (Gen.choose(min, max));
16     Gen.map4 (fun a b c d -> newComplex (newRational(a, b), newRational(c,
17   d)) |> Complex |> tryReduce ) (Gen.choose(min, max)) noneZeroGen (Gen.
18   choose(min, max)) noneZeroGen]
19
20 // numberInExprGen: Gen<Expr<Number>>
21 let numberInExprGen =
22   Gen.map (fun x -> N x) numberGen
23
24 // randomListElement: list<'a> -> Gen<'a>
25 let randomListElement xlist =
26   gen { let! i = Gen.choose(0, List.length xlist - 1)
27     return xlist.[i] }
28
29 // variableGen: list<char> -> Gen<Expr<'a>>
30 let variableGen xlist = Gen.map X (randomListElement xlist)
31
32 // leafGen: list<char> -> Gen<Expr<Number>>
33 let leafGen xlist =
34   if xlist <> [] then
35     Gen.oneof [numberInExprGen; variableGen xlist]
36   else
37     numberInExprGen
38
39 // onlyIntleafGen: list<char> -> Gen<Expr<Number>>
40 let onlyIntleafGen xlist : Gen<Expr<Number>> =
```

```

38     if xlist <> [] then
39         Gen.oneof [Gen.map (fun x -> N <| Int x) (Gen.choose(-10, 10));
40         variableGen xlist]
41     else
42         Gen.map (fun x -> N <| Int x) (Gen.choose(-10, 10))
43
44 // charsSeqGen: char -> char -> seq<Gen<char>>
45 let charsSeqGen c1 c2 = seq { for c in c1 .. c2 do
46     yield gen { return c } }
47
48 // charGen: Gen<char>
49 let charGen = gen { return! Gen.oneof (charsSeqGen 'A' 'Z')}
50
51 // smallEnvGen: Gen<Map<char, Number> * list<char>>
52 let smallEnvGen =
53     gen {
54         let! i = Gen.choose (0, 5)
55         let! xlist = Gen.listOfLength i charGen
56         let! ns = Gen.listOfLength i numberGen
57         return (Map.ofList (List.zip xlist ns), xlist) }
58
59 // exprGen: 'a -> int -> ('a -> Gen<Expr<'b>>) -> Gen<Expr<'b>>
60 let rec exprGen xlist n leafType =
61     if n = 0 then
62         leafType xlist
63     else
64         Gen.oneof [
65             // leaf occurs twice because leaf is X or N giving the same
66             // probability for each expression
67             leafType xlist;
68             leafType xlist;
69             Gen.map2 (fun x y -> Add (x, y)) (exprGen xlist (n/2) leafType) (
70                 exprGen xlist (n/2) leafType);
71             Gen.map2 (fun x y -> Mul (x, y)) (exprGen xlist (n/2) leafType) (
72                 exprGen xlist (n/2) leafType);
73             Gen.map2 (fun x y -> Div (x, y)) (exprGen xlist (n/2) leafType) (
74                 exprGen xlist (n/2) leafType);
75             Gen.map2 (fun x y -> Sub (x, y)) (exprGen xlist (n/2) leafType) (
76                 exprGen xlist (n/2) leafType);
77             Gen.map (fun x -> Neg x) (exprGen xlist (n/2) leafType)]
78
79 type SmallEnv = Map<char, Number> * char list
80 type SmallEnvGen =
81     static member SmallEnv() =
82         {new Arbitrary<SmallEnv>() with
83             override _.Generator = smallEnvGen
84             override _.Shrinker _ = Seq.empty}
85
86 type NumberGen =
87     static member Number() =
88         {new Arbitrary<Number>() with
89             override _.Generator = numberGen
90             override _.Shrinker _ = Seq.empty}

```

Listing 41: Generatorene anvendt til PBT af udtryk

Vi begynder med at definere to variable, som alle funktioner har til rådighed, `max` og `min`, som definerer intervallet for de heltal, der kan anvendes i genereringen af Numbers. Den første

---

funktion, `noneZeroGen`, er en generator for et tilfældigt heltal fra sættet  $S_1$ :

$$S_1 = \{x \mid x \in \mathbb{Z} \setminus \{0\}, \quad \min \leq x \leq \max\}.$$

Vi kan dermed bruge `noneZeroGen`, da vi ikke ønsker at generere rationale tal med nævneren 0. For at definere en generator for Numbers i form af `numberGen`, som genererer et gyldigt Number fra sættet  $S_5$ :

$$S_2 = \{x \mid x \in \mathbb{Z}, \quad \min \leq x \leq \max\}$$

$$S_3 = \left\{ \frac{x}{y} \mid x \in S_1, y \in S_2 \right\}$$

$$S_4 = \{x + yi \mid x, y \in S_3\}$$

$$S_5 = S_2 \cup S_3 \cup S_4$$

Funktionen `numberInExprGen` er en generator, som ved brug af `numberGen` konverterer et Number til et `Expr<Number>`. Det er disse generatorer, der anvendes til at generere tal til vores talmængde.

Dernæst kommer nogle generatorer til generering af variable. Først har vi `randomListElement`, som tager en liste af en vilkårlig type og udvælger et tilfældigt element fra listen. Derudover har vi `variableGen`, som tager en liste af karakterer og bruger `randomListElement` til at udvælge en af karaktererne og konvertere den til et `Expr<Number>`.

Dermed er det nu muligt at lave en generator, som kan generere enten et tal eller en variabel fra konstruktørerne af `Expr<Number>`. Disse to konstruktører er også bladene i vores udtrykstræer. `leafGen` genererer et tilfældigt blad i vores udtrykstræ ud fra en liste af karakterer. `onlyIntleafGen` fungerer ud fra det samme princip, men genererer kun heltal fra intervallet -10 til 10.

Funktionerne `charGen` og `charsSeqGen` er generatorer, som sammen genererer en tilfældig karakter mellem 'A' og 'Z'. Endelig har vi en miljøgenerator, `smallEnvGen`, som genererer et par af et miljø samt liste af variable i miljøet. Miljøet indeholder variable og deres tilsvarende værdier.

Den sidste generator, `exprGen`, vil i vores PBT tage en liste af variable, som den må generere blade ud fra, samt hvilken generator der skal anvendes til at generere bladene. Derudover vil den maksimale dybde af udtrykket være  $\log_2(n)$ .

Til sidst defineres tre typer, hvor `SmallEnvGen` og `NumberGen` gør det muligt inden kørsel af vores PBT at registrere generatorerne.

Vi har dermed nu lavet fundamentet, til at kunne teste vores egenskaber vedrørende udtryk.

### 5.1.1 Tal modulet

De 6 egenskaber fra 1 kan nu testes med PBT. Egenskaberne kan oversættes direkte til funktioner, grundet de overskrivninger vi har lavet på number typen i Listing 42.

```
1 #r "../bin/Release/net7.0/main.dll"
2 #r "nuget: FsCheck"
3 open FsCheck
4 open Number
5 open Generators
```



```

6
7 Arb.register<NumberGen>()
8
9 // Addition og multiplikation er associative
10 let associative (a:Number) (b:Number) (c:Number) =
11     a + (b + c) = (a + b) + c && a * (b * c) = (a * b) * c
12
13 // Addition og multiplikation er kommutative
14 let commutative (a:Number) (b:Number) =
15     a + b = b + a && a * b = b * a
16
17 // Distributivitet af multiplikation over addition
18 let distributive (a:Number) (b:Number) (c:Number) =
19     a * (b + c) = a * b + a * c
20
21 // Addition og multiplikation har et neutralt element
22 let neutralAdditive (a:Number) =
23     a + zero = a && a * one = a
24
25 // Omvendt funktion eksisterer til addition
26 let inverseAdditive (a:Number) =
27     a + (-a) = zero
28
29 // Omvendt funktion eksisterer til multiplikation
30 let inverseMultiplicative (a:Number) =
31     let res =
32         try
33             if a * inv a = one then 1 else 0
34         with
35             | :? System.DivideByZeroException -> 2
36     (res = 1 || res = 2)
37     |> Prop.classify (res = 1) "PropertyHolds"
38     |> Prop.classify (res = 2) "DivideByZeroExceptions"
39
40 let _ = Check.Quick associative
41 let _ = Check.Quick commutative
42 let _ = Check.Quick distributive
43 let _ = Check.Quick neutralAdditive
44 let _ = Check.Quick inverseAdditive
45 let _ = Check.Quick inverseMultiplicative

```

Listing 42: *numberPBT.fsx* - funktioner til test af egenskaberne i 1

I *inverseMultiplicative* anvender vi „Prop.classify“ til at tillade, at egenskaben kan slå bestemte fejl. I dette tilfælde tillades det, at der opstår en „DivideByZeroException“. Dette sker, fordi generatoren godt kan generere 0, som der ikke kan tages en invers af. For at teste ovenstående funktioner, køres *.fsx* filen, og outputtet kan ses i Listing 43.

```

Ok, passed 100 tests.
Ok, passed 100 tests.
Ok, passed 100 tests.
Ok, passed 100 tests.
Ok, passed 100 tests.
Ok, passed 100 tests.
94% PropertyHolds.
6% DivideByZeroExceptions.

```

Listing 43: Outputtet fra PBT af Number typer, ved kørsel af Listing 42

---

Dermed viser testen at alle egenskaberne fra 1 holder.

### 5.1.2 Homomorfisme af evaluering

Vi skal nu teste egenskaben fra 2. Testen er lavet i Listing 44. Dette er gjort ved, at der genereres et tilfældigt miljø, hvori der samples to udtryk ud fra variablene i miljøet. For alle operatorer testes det, om de overholder egenskaben om homomorfisme.

```
1 // evalOperation: Expr<Number> -> Expr<Number> -> Map<char, Number> -> (Expr<
  Number> -> Expr<Number> -> Expr<Number>) -> bool
2 let evalOperation e1 e2 env f =
3   eval (f e1 e2) env = (getNumber <| f (eval e1 env |> N) (eval e2 env |> N))
4
5 let evalPBT ((env ,xlist):SmallEnv) =
6   let result =
7     try
8       let exprList = Gen.sample 1 2 (exprGen xlist 10 leafGen)
9       let e1::[e2] = exprList
10      let prop = evalOperation e1 e2 env
11      let negation = eval (-e1) env = - eval e1 env
12      if negation && prop ( + ) && prop ( - ) && prop ( * ) && prop ( /
    ) then 1 else 0
13    with
14      | :? System.DivideByZeroException as _ -> 2
15      | :? System.OverflowException as _ -> 3
16    (result = 1 || result = 2 || result = 3)
17    |> Prop.classify (result = 1) "Property Holds"
18    |> Prop.classify (result = 2) "DivideByZeroExceptions"
19    |> Prop.classify (result = 3) "OverflowException"
```

Listing 44: PBT af egenskaben omkring Homomorfisme fra 2

Outputtet fra testen kan ses i Listing 45. Testen viser, at egenskaben holder for alle operationer.

```
> Arb.register<SmallEnvGen>()
- let _ = Check.Quick evalPBT;;
Ok, passed 100 tests.
76% Property Holds.
18% DivideByZeroExceptions.
6% OverflowException.
```

Listing 45: Outputtet fra PBT af egenskaben 2

Vi ser her en større mængde af tests, som bliver klassificeret som „DivideByZeroExceptions“. Dette skyldes, at vores generator godt kan generere udtryk som  $1/(0 \cdot X)$  og lignende, hvilket ikke er et lovligt udtryk. Derudover forekommer der også „OverflowExceptions“, hvilket skyldes, at ved evaluering af udtryk kan vi godt ende med en kombination af operationer, som giver rationale tal, der, som beskrevet i sektion 3.1.1, kan resultere i en overflow. Samme form for klassificeringer vil vi løbende se i de kommende tests.

### 5.1.3 Simplifikation af udtryk

Det er vigtigt, at vores simplifikation altid er lig med det oprindelige udtryk. Vi kan derfor teste egenskab 4 ved at simplificere et udtryk og sammenligne evalueringen af det med det oprindelige udtryk ved hjælp af et miljø. Dette er gjort i Listing 46.

```

1 // compareSimpExpr: Map<char, Number> -> Expr<Number> -> bool
2 let compareSimpExpr env (e:Expr<Number>) =
3     eval (simplifyExpr e) env = eval e env
4
5 // simpEqualEval: SmallEnv -> int
6 let simpEqualEval (env, xlist) =
7     try
8         if Gen.sample 1 1 (exprGen xlist 10 leafGen)
9             |> List.head
10            |> compareSimpExpr env
11        then 1 else 0
12    with
13        | :? System.DivideByZeroException as _ -> 2
14        | :? System.OverflowException as _ -> 3
15
16 // simpPBT: SmallEnv -> Property
17 let simpPBT (se:SmallEnv) =
18     let result = simpEqualEval se
19     (result = 1 || result = 2 || result = 3)
20     |> Prop.classify (result = 1) "Equal"
21     |> Prop.classify (result = 2) "DivideByZeroExceptions"
22     |> Prop.classify (result = 3) "OverflowException"

```

Listing 46: PBT af egenskaben 4

Outputtet fra testen kan ses i Listing 47. Testen viser, at egenskaben holder.

```

> Arb.register<SmallEnvGen>()
- let _ = Check.Quick simpPBT;;
Ok, passed 100 tests.
94% Equal.
6% DivideByZeroExceptions.

```

Listing 47: Outputtet fra PBT af egenskaben 4

#### 5.1.4 Invers morfisme mellem infix og prefix

Nu skal det undersøges, hvorvidt egenskab 3 holder. Vi beskrev i Listing 24 den inverse funktion til `tree`-funktionen, som benyttes til at opskrive egenskaben i Listing 48.

```

1 // generatesCorrectTree: Map<char, Number> -> Expr<Number> -> bool
2 let generatesCorrectTree env (e:Expr<Number>) =
3     eval e env = eval
4         (simplifyExpr e
5             |> infixExpression
6             |> tree
7             |> infixExpression
8             |> tree ) env
9
10 // treeEqualEval: SmallEnv -> int
11 let treeEqualEval (env, xlist) =
12     try
13         if Gen.sample 1 1 (exprGen xlist 10 onlyIntleafGen)
14             |> List.head
15             |> generatesCorrectTree env
16        then 1 else 0
17    with

```

```

18 | :? System.DivideByZeroException as _ -> 2
19 | :? System.OverflowException as _ -> 3
20
21 // treePBT: SmallEnv -> Property
22 let treePBT (se:SmallEnv) =
23     let result = treeEqualEval se
24     (result = 1 || result = 2 || result = 3)
25     |> Prop.classify (result = 1) "Equal"
26     |> Prop.classify (result = 2) "DivideByZeroExceptions"
27     |> Prop.classify (result = 3) "OverflowException"

```

Listing 48: PBT af egenskaben 3

Funktionen `generatesCorrectTree` tager et miljø samt et udtryk og undersøger, om:

$$\text{eval}(e) = \text{eval}(\text{tree}(\text{tree}^{-1}(\text{tree}(\text{tree}^{-1}(\text{simplifyExpr}(e)))))) \quad (5)$$

Det er gjort på denne måde, da det ikke er muligt igennem vores program at generere matematiske udtryk i infix-notation, som med sikkerhed har gyldig notation. I stedet vil vi anvende vores udtryksgenerator til at generere et udtryk, som vi så ved hjælp af et miljø kan tjekke, at egenskaben 3 holder for alle  $x \in \text{Expr}\langle \text{Number} \rangle$ .

Udtrykket bliver simplificeret, før det konverteres til infix-notation, da der er valgt ikke at placere parenteser rundt om tal i udtrykket, for visualiseringens skyld. Dette betyder, at hvis udtrykket ikke simplificeres, kan der skabes udtryk som `Neg(N -1)`, som i infix-notation er `--1`, hvilket ikke er en gyldig notation uden parenteser. Simplifikationens sørger for, at negative tal repræsenteres ved notationen af et positivt tal.

Outputtet fra kørsel af testen kan ses i Listing 49. Testen viser som forventet, at egenskaben holder for alle udtryk.

```

> Arb.register<SmallEnvGen>()
- let _ = Check.Quick treePBT;;
Ok, passed 100 tests.
94% Equal.
6% DivideByZeroExceptions.

```

Listing 49: Outputtet fra PBT af egenskaben 3

### 5.1.5 Differentiering af udtryk

Selv om differentieringsfunktionen `diff` er implementeret ud fra egenskaberne i 5, er det stadig vigtigt at teste, om funktionen overholder de samme egenskaber. Dette er gjort i `diffPBT.fsx`, se Listing 50.

```

1 #r "../bin/Release/net7.0/main.dll"
2 #r "nuget: FsCheck"
3 open FsCheck
4 open Expression
5 open Number
6 open Generators
7 open SymbolicManipulation
8 open Differentiation
9 Arb.register<SmallEnvGen>()

```

```

10 Arb.register<NumberGen>()
11
12 // Skaleringsreglen
13 let scaleP f a env dx =
14     eval (diff (a * f) dx) env = eval (a * diff f dx) env
15
16 // Sumreglen
17 let additionP f g env dx =
18     eval (diff (f + g) dx) env = eval (diff f dx + diff g dx) env
19
20 // Subtraktionsreglen
21 let subtractionP f g env dx =
22     eval (diff (f - g) dx) env = eval (diff f dx - diff g dx) env
23
24 // Produktreglen
25 let multiplicationP f g env dx =
26     eval (diff (f * g) dx) env = eval (diff f dx * g + f * diff g dx) env
27
28 // Kvotientreglen
29 let divisionP f g env dx =
30     eval (diff (f / g) dx) env = eval ((g * diff f dx - f * diff g dx) / (g *
31     g)) env
32
33 // Property test of differentiation
34 let diffPBT ((env ,xlist):SmallEnv) (a:Number)=
35     let result =
36         try
37             let a = N a
38             let dx = if xlist <> [] then List.head xlist else 'X'
39             let exprList = Gen.sample 1 2 (exprGen xlist 10 leafGen)
40             let e1::[e2] = exprList
41             let e1 = simplifyExpr e1
42             let e2 = simplifyExpr e2
43             if
44                 scaleP e1 a env dx && scaleP e2 a env dx &&
45                 additionP e1 e2 env dx &&
46                 subtractionP e1 e2 env dx &&
47                 multiplicationP e1 e2 env dx &&
48                 divisionP e1 e2 env dx
49             then 1
50             else 0
51         with
52             | :? System.DivideByZeroException as _ -> 2
53             | :? System.OverflowException as _ -> 3
54     (result = 1 || result = 2 || result = 3)
55     |> Prop.classify (result = 1) "Property Holds"
56     |> Prop.classify (result = 2) "DivideByZeroExceptions"
57     |> Prop.classify (result = 3) "OverflowException"
58
59 printfn "Differentiation property based testing"
60 let _ = Check.Quick diffPBT

```

Listing 50: *diffPBT.fsx* - funktioner til test af egenskaberne i 5

Outputtet fra kørsel af testen kan ses i Listing 51. Testen viser, at egenskaberne holder.

```

Differentiation property based testing
Ok, passed 100 tests.
72% Property Holds.

```

```
15% DivideByZeroExceptions.  
13% OverflowException.
```

Listing 51: Outputtet fra PBT af differentiering af udtryk

## 5.2 PBT af vektorer og matricer

Ligesom ved PBT af udtryk begynder vi med at lave end generatorer for matricer.

```
1 // vectorGen : int -> Gen<Vector>  
2 let vectorGen n =  
3   Gen.listOfLength n numberGen |> Gen.map (fun x -> vector x)  
4  
5 // matrixGen : Gen<Matrix>  
6 let matrixGen =  
7   gen {  
8     let! row = Gen.choose(1, 6)  
9     let! col = Gen.choose(1, 6)  
10    let! vectors = Gen.listOfLength col (vectorGen row)  
11    return matrix vectors  
12  }  
13  
14 type MatrixGen =  
15   static member Matrix() =  
16     {new Arbitrary<Matrix>() with  
17       override _.Generator = matrixGen  
18       override _.Shrinker _ = Seq.empty}
```

Listing 52: Generatorerne anvendt til PBT af matrixoperationer

Først defineres `vectorGen`, som laver en liste ved hjælp af den tidligere definerede funktion `numberGen`. Den laver en liste med en given længde af tilfældige tal fra vores talmængde. Dernæst passes listen videre til `vector`-funktionen, som laver en søjlelagret vektor ud fra listen. Funktionen `matrixGen` genererer en matrix af tilfældig størrelse, hvor antallet af rækker og kolonner er mellem 1 og 6 ved hjælp af `vectorGen`. Til sidst kan vi definere en type for matrixgeneratorerne, som gør det muligt at registrere dem før kørsel af PBT.

### 5.2.1 PBT af matrix operationer

Det er nu muligt at opstille nogle PBT af der sikre at matricerne overholder matematiske egenskaber i sætning 6.

Vi kan dermed nu lave definere egenskaberne fra 6 som nogle funktioner i Listing 53.

```
1 //vectorCom : Matrix -> bool  
2 let vectorCom m =  
3   sumRows m = sumRows (flip m)  
4  
5 //vectorScalarAss : Matrix -> Number -> Number -> bool  
6 let vectorScalarAss (m:Matrix) (n1:Number) (n2:Number) =  
7   n1 * (n2 * m) = (n1 * n2) * m  
8  
9 //vectorAssCom : Matrix -> Number -> bool  
10 let vectorAssCom m (c:Number) =
```

---

```

11  c * (sumRows m) = sumRows (c * m)

```

Listing 53: Egenskaberne fra sætning 6 som funktioner

Listing 54 viser outputtet fra kørsel af testene. Testene viser, at egenskaberne holder for alle matricer.

```

> Arb.register<MaxtrixGen>()
- let _ = Check.Quick vectorCom
- let _ = Check.Quick vectorScalarAss
- let _ = Check.Quick vectorAssCom;;
Ok, passed 100 tests.
Ok, passed 100 tests.
Ok, passed 100 tests.

```

Listing 54: Outputtet fra PBT af vektor Listing 53

### 5.2.2 PBT af Gram-Schmidt

Udfordringen ved at lave en PBT af Gram-Schmidt er, at vektorsættet skal være lineært uafhængigt. Derfor laves der en generator, som ved at udføre tilfældige rækkeoperationer på en diagonal matrix med  $m$  rækker og  $n$  søjler, hvorom det gælder, at  $n \geq m$ . Dette sikrer os, at rækkerne i matricen er lineært uafhængige. Hertil vil den transponerede matrix have lineært uafhængige søjler, som vi kan bruge til at teste Gram-Schmidt processen<sup>27</sup>. Listing 55 viser de forskellige generatorer, som anvendes til PBT af Gram-Schmidt-processen.

```

1  // performRowOperationGen: Matrix -> Gen<Matrix>
2  let performRowOperationGen m =
3      let (D(n, _)) = dimMatrix m
4      gen {
5          let! i = Gen.choose(1, n)
6          let! j = match i with
7              | 1 -> Gen.choose(2, n)
8              | _ when i = n -> Gen.choose(1, n-1)
9              | _ -> Gen.oneof [Gen.choose(1, i-1); Gen.choose(i+1, n)]
10         let! a = numberGen
11         return rowOperation i j a m }
12
13 // multipleRowOperationsGen: Matrix -> int -> Gen<Matrix>
14 let rec multipleRowOperationsGen m count =
15     if count <= 0 then Gen.constant m
16     else
17         gen {
18             let! newMatrix = performRowOperationGen m
19             return! multipleRowOperationsGen newMatrix (count - 1)
20         }
21
22 // getDiagonalMatrixGen: int -> Gen<Matrix>
23 let getDiagonalMatrixGen maxRows =
24     gen {
25         let! m = Gen.choose(2, maxRows)
26         let! n = Gen.choose(m, m + 3)
27         return fullrankedDiagonalMatrix m n }

```

---

<sup>27</sup>[16] *Rank (linear algebra)*.

---

```

28
29 // getIndependentBasisGen: Gen<Matrix>
30 let getIndependentBasisGen =
31     gen {
32         let! A = getDiagonalMatrixGen 5
33         let! numberOfOperations = Gen.choose(1, 10)
34         let! span = multipleRowOperationsGen A numberOfOperations
35         return span |> transposeMatrix }
36
37 type IndependentBasisMatrix = Matrix
38 type IndependentBasisMatrixGen =
39     static member IndependentBasisMatrix() =
40         {new Arbitrary<Matrix>() with
41             override _.Generator = getIndependentBasisGen
42             override _.Shrinker _ = Seq.empty}

```

Listing 55: Generatorene anvendt til PBT af Gram-Schmidt

Funktionen `performRowOperationGen` tager en matrix, hvori der udvælges to tilfældige rækker,  $i$  og  $j$ , hvorefter der udføres en rækkeoperation på  $R_j$ , således at  $R_j \leftarrow R_j - aR_i$ , hvor  $a$  er et tilfældigt tal. `multipleRowOperationsGen` gentager denne proces et givet antal gange. Funktionen `getDiagonalMatrixGen` laver en diagonal matrix med  $m$  rækker og  $n$  søjler, hvor  $n \geq m$ . `getIndependentBasisGen` genererer en matrix, hvis søjler er lineært uafhængige. Vi kan dermed definere en funktionen til at undersøge om en matrix er en ortogonal basis, som vist i Listing 56.

```

1 // isOrthogonalBasis : Matrix -> bool
2 let rec isOrthogonalBasis m =
3     (transposeMatrix m |> conjugateMatrix) * m |> isDiagonalMatrix

```

Listing 56: Funktion til at tjekke om søjlerne i en matrix er en ortogonal basis

Dernæst skal vi bruge en funktion til at tjekke, om en matrix er en ortogonal basis. `isOrthogonalBasis` i Listing 56 tjekker, om alle vektorerne i en matrix er ortogonale med hinanden. Dette kan gøres ved hjælp af følgende sætning:

### Sætning 2.

Lad  $A$  have søjlerne  $v_1, v_2, \dots, v_n$ , som udgør en ortogonal basis. Per definition gælder det, at det indre produkt mellem to ortogonale vektorer  $v_i$  og  $v_j$  er:

$$\langle v_i, v_j \rangle = 0 \quad \text{for } i \neq j \quad (6)$$

Lad så  $D$  være en kvadratisk diagonal matrix. Da gælder det, at:

$$D = A^* A \quad (7)$$

Bevis. Fra definitionen af matrixprodukt 4 ved vi, at:

$$A^* A = \begin{bmatrix} | & & | \\ A^* \cdot v_1 & \cdots & A^* \cdot v_n \\ | & & | \end{bmatrix} \quad (8)$$



Fra definitionen af matrix-vektorprodukt 3 ved vi, at:

$$A^*v_i = \begin{bmatrix} - & \bar{v}_1 & - \\ & \vdots & \\ - & \bar{v}_n & - \end{bmatrix} \begin{bmatrix} v_{1i} \\ \vdots \\ v_{ni} \end{bmatrix} = \begin{bmatrix} \sum_{j=1}^m \bar{v}_{j1}v_{ji} \\ \sum_{j=1}^m \bar{v}_{j2}v_{ji} \\ \vdots \\ \sum_{j=1}^m \bar{v}_{jn}v_{ji} \end{bmatrix} \quad (9)$$

Hvilket ifølge definitionen af indre produkt 5 betyder

$$A^*v_i = \begin{bmatrix} \langle v_i, v_1 \rangle \\ \langle v_i, v_2 \rangle \\ \vdots \\ \langle v_i, v_n \rangle \end{bmatrix} \quad (10)$$

Hvilket betyder, at:

$$A^*A = \begin{bmatrix} \langle v_1, v_1 \rangle & \langle v_2, v_1 \rangle & \cdots & \langle v_n, v_1 \rangle \\ \langle v_1, v_2 \rangle & \langle v_2, v_2 \rangle & \cdots & \langle v_n, v_2 \rangle \\ \vdots & \vdots & \ddots & \vdots \\ \langle v_1, v_n \rangle & \langle v_2, v_n \rangle & \cdots & \langle v_n, v_n \rangle \end{bmatrix} \quad (11)$$

Derfor, hvis  $m > n$ , så gælder det, at  $A^*A = D$ , da hvis  $m < n$ , ville  $v_1, v_2, \dots, v_n$  ikke være lineært uafhængige vektorer.  $\square$

Der udover skal vi også undersøge om spannet er det samme dette gøres ved at undersøge følgende:

### Sætning 3.

Lad  $w_1, w_2, \dots, w_n$  være de nye vektorer efter Gram-Schmidt processen på  $v_1, v_2, \dots, v_n$ , som er lineært uafhængige. Vi vil undersøge om  $\text{span}(v_1, v_2, \dots, v_n) = \text{span}(w_1, w_2, \dots, w_n)$

Lad derfor  $B$  være echelonformen af  $A$ . Så gælder det, at

$$\text{rank}(B_{1..n}) = \text{rank}(B_{n+1..2n}) = n, \quad (12)$$

hvor

$$A = \begin{bmatrix} | & | & & | & | & | & & | \\ w_1 & w_2 & \cdots & w_n & v_1 & v_2 & \cdots & v_n \\ | & | & & | & | & | & & | \end{bmatrix} \quad (13)$$

og

$$A = \begin{bmatrix} | & | & & | & | & | & & | \\ v_1 & v_2 & \cdots & v_n & w_1 & w_2 & \cdots & w_n \\ | & | & & | & | & | & & | \end{bmatrix}. \quad (14)$$

Da for  $k = 1, 2, \dots, n$  er  $w_k$  udelukkende lineært afhængig af  $v_1, v_2, \dots, v_k$ .

Funktionen `hasSameSpan` i Listing

---

```

1 // hasSameSpan : Matrix -> Matrix -> bool
2 let hasSameSpan m1 m2 =
3     let (D(r, c)) = dimMatrix m1
4     if (D(r, c)) <> dimMatrix m2 then failwith "Matrices must have the same
5         dimension"
6     else
7         let hSS m1 m2 =
8             let s1, s2 = extendMatrixWithMatrix m1 m2 |> rowEchelonForm |>
9                 splitMatrix C (c-1)
10             hasFullRank s1 && hasFullRank s2
11         hSS m1 m2 && hSS m2 m1

```

Listing 57: Funktion til at undersøge om søjlerne i to matricer har samme span

Dermed kan vi opskrive en PBT af Gram-Schmidt processen i Listing 58.

```

1 let gramSchmidtIsOrthogonal (m:independentBasisMatrix) =
2     let res =
3         try
4             let um = orthogonalBasis m
5             if isOrthogonalBasis um && hasSameSpan m um then
6                 1
7             else
8                 0
9         with
10            | :? System.DivideByZeroException as _ -> 2
11            | :? System.OverflowException as _ -> 3
12
13     (res = 1 || res = 2 || res = 3)
14     |> Prop.classify (res = 1) "PropertyHolds"
15     |> Prop.classify (res = 2) "DivideByZeroExceptions"
16     |> Prop.classify (res = 3) "OverflowException"

```

Listing 58: PBT af Gram-Schmidt processen

```

> Arb.register<independentBasisMatrixGen>()
- let _ = Check.Quick gramSchmidtIsOrthogonal;;
Ok, passed 100 tests.
69% PropertyHolds.
31% OverflowException.

```

Listing 59: Output fra PBT af Gram-Schmidt processen

Outputtet fra PBT af Gram-Schmidt processen kan ses i Listing 59. Som sædvanligt indikerer testen kun korrekthed, men medfører ikke garanti for det.

Som nævnt i sektion 4.3 var der en slåfejl i Algorithm 9 for echelonform i noterne til „01001 Mathematics 1a“<sup>28</sup>. I forbindelse med at finde frem til denne fejl blev der lavet en PBT af echelonform af generatoren for lineært uafhængige vektorsæt. Egenskaben i ?? skal sikre, at både generatorens implementering er korrekt, og at funktionen til at finde echelonformen er det.

---

<sup>28</sup>[11] *Mathematics 1a*.

---

### Egenskab 8.

Lad  $A$  være en matrix, hvor søjlerne er lineært uafhængige. Da gælder det, at  $A$  har fuld rank.

Vi kan dermed opskrive en PBT af generatoren i Listing 60.

```
1 // independentBasisMatrixHasFullRank : independentBasisMatrix -> bool
2 let independentBasisMatrixHasFullRank (m:independentBasisMatrix) =
3     let res =
4         try
5             if hasFullRank m then 1 else 0
6         with
7             | :? System.OverflowException -> 2
8     (res = 1 || res = 2)
9     |> Prop.classify (res = 1) "PropertyHolds"
10    |> Prop.classify (res = 2) "OverflowException"
```

Listing 60: PBT af generatoren for lineært uafhængige vektorsæt

Outputtet fra PBT af generatoren for lineært uafhængige vektorsæt kan ses i Listing 61. Testen viser, at egenskaben holder.

```
> Arb.register<independentBasisMatrixGen>()
- let _ = Check.Quick independentBasisMatrixHasFullRank;;
Ok, passed 100 tests.
87% PropertyHolds.
13% OverflowException.
```

Listing 61: Output fra PBT af generatoren for lineært uafhængige vektorsæt

## 6 Diskussion

Vi begyndte rapporten med at informere læseren om, at Python er blevet indført som et hjælpemiddel i matematikkurserne på DTU. En af fordelene ved Python er klart, at det er et meget mere udbredt programmeringssprog, hvilket gør det nemmere at finde hjælp og vejledninger til opbygningen af et program eller løsning af en opgave. Under 1% af udviklere i 2023 anvender F#, hvilket er markant mindre end de næsten 50% som bruger Python. Derfor har det været en udfordring i udviklingen af dette program at finde vejledning til de problemstillinger, der er opstået undervejs.

Derimod, eftersom F# er et stærkt typet sprog, har det været markant nemmere at finde fejl i programmet inden det bliver kørt, hvilket er en af de store fordele ved F# frem for Python og Ruby<sup>29</sup>, som på mange måder ligner Python. Ruby er også det sprog, jeg arbejder med til daglig. Ruby tillader, ligesom Python, at kalde funktioner med objekter uden at definere en type for objektet, så længe objektet har de metoder, som funktionen forventer. Det medfører at man ofte skal tjekke om en metode er tilstede på et objekt, hvilket vi ikke behøver i F#. Derudover kan objekter få tilføjet metoder under kørslen, hvilket kan have sine fordele, men som udvikler medfører det flere problemstillinger, blandt andet at når et objekt ikke har den forventede metode, vil programmet først fejle under kørslen.

Mangel på typer sammenlignet med F# betyder også, at det tager længere tid og kræver flere kommentarer at forstå, hvad et program gør. Denne udfordring har man ikke med F#, så længe

---

<sup>29</sup>[19] Ruby.

---

funktionsnavnet er sigende for, hvad funktionen gør. Kombineret med at kunne se, typen for funktionen, behøver man ofte ikke at læse selve koden for at forstå, hvad funktionen. Dette medfører, at man som udvikler kan være mere effektiv. Hvilket også giver rent performens mæssige fordele ved kørsel af et F# program.

Vedrørende syntaksen af F# sammenlignet med Python/Ruby, er de mere kendt for at have en mere læsbar syntaks, hvilket gør dem mere begynder venlige. Især da F#'s syntaks er forholdsvis anderledes end objektorienterede programmeringssprog. Dog ligger syntaksen for mange af funktionerne i dette program meget tæt op ad matematiske notationer, især grundet muligheden for nemt at kunne overskrive operatorer på egne typer. Netop det, at mange af funktionerne ligner de matematiske funktioner, gør, at implementeringen af dem burde medføre en bedre forståelse af, hvorfor mange af de metoder, man lærer at udføre i hånden er korrekte. Jeg har personligt fået en bedre forståelse af især matrixoperationer, og hvordan matrix-matrix produkt er bygget på matrix-vektor produkt. Sammenhængen mellem matematiske domæner og typerne i F# gør, at man også kan forstå domæner fra et andet synspunkt, hvilket kan være med til at give en bedre forståelse herom.

Når det kommer til selve programmet, er der også både nogle udvidelser og forbedringer, der kunne laves. Først og fremmest fortrød jeg sent, at matricemodulet ikke var bygget med følgende typer i Listing 62.

```
1 type Vector = V of list<Expr<Number>> * Order
2 type Matrix = M of list<Vector> * Order
```

Listing 62: Eksempel på alternative typer for matrixmodulet

Dette ville ikke have betydet store ændringer i funktionerne for modulet uden at ændre på funktionaliteten. Desuden ville det have gjort det nemmere at implementere løsninger til et lineært ligningssystem.

Modulet for komplekse tal kunne også have haft en polymorfisk type som i Listing 63.

```
1 type Complex<'a, 'b> = C of 'a * 'b
```

Listing 63: Eksempel på alternative typer for komplekse tal modulet

Som ville medføre, at Number-typen ville blive:

```
1 type Number =
2 | Int of int
3 | Rational of Rational
4 | Complex of Complex<Number, Number>
```

Listing 64: Eksempel på alternative typer for Number-typen

Denne ændring ville selvfølgelig medføre, at Number ville blive en rekursivt defineret type, hvilket skulle håndteres ansvarligt.

---

---

## 7 Konklusion

Det primære fokuspunkt for rapporten er, at en læser skulle opnå en bedre forståelse af matematiske koncepter gennem at opbygge et matematisk program ved hjælp af funktionsprogrammering. Min erfaring fra samtaler med mine medstuderende er, at rekursive funktioner generelt er et svært koncept for mange. Derfor tror jeg også, at introducere de studerende til rekursivitet gennem et funktionelt programmeringssprog som F# ved hjælp af matematiske koncepter, som de møder i kurserne „01001 Matematik 1a“ og „01002 Matematik 1b“, vil gavne de studerendes forståelse af både matematikken og programmeringssproget. Derudover vil kendskabet til funktionsprogrammering forbedre de studerendes evne til at skrive bedre programmer, for eksempel i Python. Noget jeg ser hos mine medstuderende på „Matematik og Teknologi“, som har lært at programmere i Python gennem de to grundlæggende matematikkurser, er blandt andet de ikke definerer funktioner for at genbruge kode. I stedet kopierer og indsætter de kode, hvilket ikke er en god praksis.

Selvom syntaksen for F# kan være sværere at lære for nogle, tror jeg på, at gevinsten ved at lære den er stor. Implementeringen af dette program, fra hvordan man udfører operationer af komplekse tal til hvordan man ud fra simple matematiske regler kan reducere udtryk, vil gavne forståelsen af matematikken. De studerende vil forstå, hvorfor diverse funktioner virker. Hvilket ikke altid er tilfældet i Python, hvor mange funktioner kan håndteres som en sort boks.

Konklusionen er, at F# måske ikke vil kunne erstatte Python i matematikkurserne på DTU, men det vil kunne være et godt supplement til at forbedre de studerendes forståelse af matematikken og programmering.

---

## Litteratur

- [1] *Convert Infix expression to Postfix expression*. URL: <https://www.geeksforgeeks.org/%20convert-infix-expression-to-postfix-expression/> (hentet 17.02.2024).
- [2] *Differentiation Rules*. URL: [https://en.wikipedia.org/wiki/Differentiation\\_rules](https://en.wikipedia.org/wiki/Differentiation_rules) (hentet 10.05.2024).
- [3] *Floating-point error*. URL: [https://en.wikipedia.org/wiki/Floating-point\\_error\\_mitigation](https://en.wikipedia.org/wiki/Floating-point_error_mitigation) (hentet 30.03.2024).
- [4] *fsharp.org*. URL: <https://fsharp.org/> (hentet 30.03.2024).
- [5] *Github repository for the project*. URL: <https://github.com/Larsen00/funktionsprogrammeringForIndledend> (hentet 24.05.2024).
- [6] *Greatest common divisor wikipedia*. URL: [https://en.wikipedia.org/wiki/Greatest\\_common\\_divisor](https://en.wikipedia.org/wiki/Greatest_common_divisor) (hentet 30.03.2024).
- [7] *Inverse function*. URL: [https://en.wikipedia.org/wiki/Inverse\\_function](https://en.wikipedia.org/wiki/Inverse_function) (hentet 06.05.2024).
- [8] Ole Christensen og Jakob Lemvig. *Mathematics 1b - Functions of several variables*. URL: [https://01002.compute.dtu.dk/\\_assets/notesvol2.pdf](https://01002.compute.dtu.dk/_assets/notesvol2.pdf) (hentet 09.02.2024).
- [9] *lambda calculus*. URL: [https://en.wikipedia.org/wiki/Lambda\\_calculus](https://en.wikipedia.org/wiki/Lambda_calculus) (hentet 20.05.2024).
- [10] *Maple*. URL: <https://www.maplesoft.com/products/Maple/> (hentet 19.05.2024).
- [11] *Mathematics 1a*. URL: [https://01001.compute.dtu.dk/\\_assets/enotesvol1.pdf](https://01001.compute.dtu.dk/_assets/enotesvol1.pdf) (hentet 09.02.2024).
- [12] *Operator Precedence and Associativity in C*. URL: <https://www.geeksforgeeks.org/operator-precedence-and-associativity-in-c/> (hentet 17.02.2024).
- [13] *Polsk notation wikipedia*. URL: [https://en.wikipedia.org/wiki/Polish\\_notation](https://en.wikipedia.org/wiki/Polish_notation).
- [14] *Python - List Comprehension*. URL: [https://www.w3schools.com/python/python\\_lists\\_comprehension.asp](https://www.w3schools.com/python/python_lists_comprehension.asp) (hentet 31.03.2024).
- [15] *Python: The most popular language*. URL: <https://datascientest.com/en/python-the-most-popular-language#:~:text=Python%20is%20also%20the%20most%20widely%20used%20language%20for%20Data,libraries%20and%20other%20numerical%20algorithms.> (hentet 19.05.2024).
- [16] *Rank (linear algebra)*. URL: [https://en.wikipedia.org/wiki/Rank\\_\(linear\\_algebra\)#:~:text=The%20column%20rank%20of%20A,row%20rank%20are%20always%20equal.](https://en.wikipedia.org/wiki/Rank_(linear_algebra)#:~:text=The%20column%20rank%20of%20A,row%20rank%20are%20always%20equal.) (hentet 18.05.2024).
- [17] *Rational Number wikipedia*. URL: [https://en.wikipedia.org/wiki/Rational\\_number](https://en.wikipedia.org/wiki/Rational_number) (hentet 30.03.2024).
- [18] *Row- and column-major order*. URL: [https://en.wikipedia.org/wiki/Row-\\_and\\_column-major\\_order](https://en.wikipedia.org/wiki/Row-_and_column-major_order) (hentet 31.03.2024).
- [19] *Ruby*. URL: <https://www.ruby-lang.org/en/> (hentet 19.05.2024).
- [20] *Test-Driven Development*. URL: [https://en.wikipedia.org/wiki/Test-driven\\_development](https://en.wikipedia.org/wiki/Test-driven_development) (hentet 31.03.2024).
- [21] *Tree Traversal Techniques - Data Structure and Algorithm Tutorials*. URL: <https://www.geeksforgeeks.org/tree-traversals-inorder-preorder-and-postorder/> (hentet 30.03.2024).
- [22] *Typed lambda calculus*. URL: [https://en.wikipedia.org/wiki/Typed\\_lambda\\_calculus](https://en.wikipedia.org/wiki/Typed_lambda_calculus) (hentet 20.05.2024).
- [23] *Unit Testing*. URL: [https://en.wikipedia.org/wiki/Unit\\_testing](https://en.wikipedia.org/wiki/Unit_testing) (hentet 19.05.2024).

---

## 8 Appendiks

Hele projektet kan findes på Github, på følgende reference [5]. Udvalgte filer som omtales fra projektet er vedlagt i følgende sektioner.

### 8.1 complex.fsi

```
1 module complex
2 open rational
3
4 type complex = C of rational * rational
5 with
6     static member ( + ) : complex * complex -> complex
7     static member ( - ) : complex * complex -> complex
8     static member ( * ) : rational * complex -> complex
9     static member ( * ) : complex * rational -> complex
10    static member ( * ) : complex * complex -> complex
11    static member ( / ) : complex * rational -> complex
12    static member ( / ) : complex * complex -> complex
13    static member ( ~- ) : complex -> complex
14
15 val newComplex : rational * rational -> complex
16 val isGreater : complex * complex -> bool
17 val realPart : complex -> rational
18 val isReal : complex -> bool
19 val isZero : complex -> bool
20 val toString : complex -> string
21 val isNegative : complex -> bool
22 val absComplex : complex -> complex
23 val conjugate : complex -> complex
```

Listing 65: complex.fsi

### 8.2 TreeGenerator.fs

```
1 module TreeGenerator
2 open Number
3 open Expression
4
5 type Associative = | Left | Right
6 type Precedence = int
7 type Operator = char * Precedence * Associative
8 type Token =
9     | Operand of char
10    | Operator of Operator
11    | Konstant of int
12 type OperatorList = Operator list
13
14 // Converts a infix string to a list of tokens
15 let rec infixToTokenList s =
16     mapToToken (Seq.toList s) true false
17
18 // maps a token to its corresponding token type
19 and mapToToken l allowUnary allowOperator=
20     match l with
21     | [] -> []
```



```

22 | x::tail when x = ' ' -> mapToToken tail allowUnary allowOperator
23 | x::tail when allowOperator && x = '/' || x = '*' -> Operator (x, 2, Left)
24 | x::tail when allowOperator && (x = '+' || (x = '-' && not allowUnary))
25 -> Operator (x, 1, Left)::mapToToken tail false false
26 | x::tail when (x = '-' && allowUnary) -> Operator ('~', 2, Right)::
27 mapToToken tail true false
28 | x::tail when x = '(' -> Operator (x, -1, Left)::mapToToken tail true
29 true
30 | x::tail when x = ')' -> Operator (x, -1, Left)::mapToToken tail false
31 true
32 | x::_ when System.Char.IsDigit(x) ->
33 let (k, tail) = foundInt 1 ""
34 Konstant (int k):: mapToToken tail false true
35 | x::tail when System.Char.IsLetter x -> Operand x::mapToToken tail false
36 true
37 | x::_ -> failwith ("Invalid syntax at: " + string x)
38
39
40 // Allowing more than one digit in a number
41 and foundInt 1 s =
42 match 1 with
43 | x::tail when System.Char.IsDigit(x) -> foundInt tail (s + string x)
44 | _ -> (s, 1)
45
46
47 // pops the last operator from the stack and adds it to the prefix list
48 let popprefixStack op prefix =
49 match op, prefix with
50 | x, e1::e2::tail when x = '+' -> Add(e2, e1)::tail
51 | x, e1::e2::tail when x = '-' -> Sub(e2, e1)::tail
52 | x, e1::e2::tail when x = '*' -> Mul(e2, e1)::tail
53 | x, e1::e2::tail when x = '/' -> Div(e2, e1)::tail
54 | x, e::tail when x = '~' -> Neg(e)::tail
55 | x, _ when x = '(' -> prefix
56 | _, _ -> failwith "match not found"
57
58
59 // Runs the algorithm to generate the expression tree
60 let rec generateExpression c (stack:OperatorList) prefix =
61 match c, stack with
62 | [], [] -> prefix
63 | [], (s,_,_):stack_tail -> generateExpression c stack_tail (
64 popprefixStack s prefix)
65 | Operand x :: tail, _ -> generateExpression tail stack (X x::prefix)
66 | Konstant x :: tail, _ -> generateExpression tail stack (N (Int x)::prefix)
67 )
68 | Operator (x, prec, lr)::tail, _
69 when x = '('
70 -> generateExpression tail ((x, prec, lr)::stack) prefix
71 | Operator (x, _, _):tail, _
72 when x = ')'
73 -> match stack with
74 | [] -> generateExpression tail stack prefix
75 | (s,_,_):stack_tail
76 when s = '('
77 -> generateExpression tail stack_tail prefix
78 | (s,_,_):stack_tail
79 -> generateExpression c stack_tail (popprefixStack s prefix)
80 | Operator e::tail, [] -> generateExpression tail (e::stack) prefix
81 | Operator e::tail, s::stack_tail

```

```

73     -> match e, s with
74     | (x, precX, lr), (y, precY, _)
75         when precX < precY || (precX = precY && lr = Left)
76         -> generateExpression c stack_tail (popprefixStack y prefix)
77     | _, _ -> generateExpression tail (e::stack) prefix
78
79 // Converts a infix string to a expression tree
80 let tree s =
81     match generateExpression (infixToTokenList s) [] [] with
82     | [] -> failwith "Tree is empty"
83     | tree::_ -> tree
84
85 // Adds parenthesis to a string if a boolean is true
86 let parenthesis b f =
87     if b then "(" + f + ")" else f
88
89 // Converts a expression tree to a infix string
90 let rec etf e p =
91     match e with
92     | N a when not <| isInt a -> parenthesis p <| toString a
93     | N a -> toString a
94     | X a -> string a
95     | Neg a -> parenthesis p <| "-" + etf a (not p)
96     | Add(a, b) -> parenthesis p <| etf a false + "+" + etf b false
97     | Sub(a, b) -> parenthesis p <| etf a false + "-" + etf b true
98     | Mul(a, b) -> parenthesis p <| etf a true + "*" + etf b true
99     | Div(a, b) -> parenthesis p <| etf a true + "/" + etf b true
100
101 // Converts a expression tree to a infix string
102 let infixExpression e = etf e false

```

Listing 66: TreeGenerator.fs

### 8.3 CommutativeAddSub.fs

```

1 module commutativeAddSub
2 open Expression
3 open Number
4
5
6 // rank when sorting a commutative expression
7 let expressionSortRank e1 =
8     match e1 with
9     | N _ -> 1
10    | Neg (N _) -> 2
11    | X _ -> failwith "Variables should not be in the list" // all variables
    are multiplied with 1
12    | Add _ -> failwith "Addition should not be in the list" // all addition
    should be reduced
13    | Sub _ -> failwith "Subtraction should not be in the list" // all
    subtraction should be reduced
14    | Mul _ -> 6
15    | Div _ -> 7
16    | Neg _ -> 8
17
18 // Flattens a expression tree with respect to addition and subtraction
19 let rec flatTree e =
20     match e with

```

---

```

21 | Add (a, b) -> flatTree a @ flatTree b
22 | Sub (a, b) -> flatTree a @ flatTree (Neg b)
23 | Neg (Add(a, b)) -> flatTree (Neg a) @ flatTree (Neg b)
24 | Neg (Sub(a, b)) -> flatTree (Neg a) @ flatTree b
25 | X a -> [Mul(N one, X a)]
26 | Div (a, b) -> [a / b]
27 | Neg (Neg a) -> flatTree a
28 | N _ | Div _ | Mul _ | Neg _ -> [e]
29
30
31
32 let rec sort l = List.sortBy (fun e -> expressionSortRank e) l
33
34 // Reduces a sorted commutative list for addition
35 let rec reduceNumbers l =
36   // printfn "rn %A" l
37   match l with
38   | [] -> []
39   | N a :: N b :: tail -> reduceNumbers (N a + N b :: tail)
40   | N a :: Neg (N b) :: tail -> reduceNumbers (N a - N b :: tail)
41   | Neg (N a) :: Neg (N b) :: tail -> reduceNumbers (Neg (N a + N b) :: tail)
42   | Neg (N a) :: tail -> Neg (N a) :: tail
43   | N a :: tail -> N a :: tail
44   | _ -> l
45
46 // given a list of commutative expressions, and a variable, sums the
47 // coefficients of the variable
48 let rec sumInstancesOfVariable x l =
49   match x, l with
50   | _, [] -> ([], x)
51   | Mul(N n1, X x1), Mul(N n2, X x2) :: tail
52   | Mul(N n1, X x1), Mul(X x2, N n2) :: tail
53   | Mul(X x1, N n1), Mul(N n2, X x2) :: tail
54   | Mul(X x1, N n1), Mul(X x2, N n2) :: tail
55     when x1 = x2 -> sumInstancesOfVariable (Mul(N (n1 + n2), X x1)) tail
56   | _, head :: tail ->
57     let (l_new, x_new) = sumInstancesOfVariable x tail
58     (head :: l_new, x_new)
59
60
61
62 let rec reduceVariables l =
63   match l with
64   | [] -> []
65   | Mul(N a, X b) :: tail
66   | Mul(X b, N a) :: tail
67     ->
68     let (l_new, x_new) = sumInstancesOfVariable (Mul(N a, X b)) tail
69     x_new :: reduceVariables l_new
70   | head :: tail -> head :: reduceVariables tail
71
72 let rec rebuildTree l =
73   // printfn "rt %A" l
74   match l with
75   | [] -> N zero
76   | Neg x :: tail -> (rebuildTree tail) - x
77   | Mul (a, b) :: tail -> (rebuildTree tail) + (a * b)

```

---

```

78 | Div (a, b)::tail -> (rebuildTree tail) + (a / b)
79 | x::tail -> rebuildTree tail + x
80
81 let applyCommutative e =
82   match e with
83   | Sub _ | Add _ -> flatTree e |> sort |> reduceNumbers |> reduceVariables
84   |> rebuildTree
85   | _ -> e

```

Listing 67: CommutativeAddSub.fs

## 8.4 CommutativeMulDiv.fs

```

1 module commutativeMulDiv
2 open Expression
3 open Number
4
5
6 ///////////////////////////////////////////////////////////////////
7 /// Commutative RULES For multiplication and division ///
8 ///////////////////////////////////////////////////////////////////
9
10 // rank when sorting a commutativ expression
11 let expressionSortRank e1 =
12   match e1 with
13   | Neg _ -> 1
14   | N _ -> 2
15   | X _ -> 3
16   | Add _ -> 4
17   | Sub _ -> 5
18   | Mul _ -> 6
19   | Div _ -> 7
20
21 // sorts the commutativ list
22 let rec sort l = List.sortBy (fun e -> expressionSortRank e) l
23
24
25 // determines the sign of a commutativ list
26 let rec signList l s =
27   match l with
28   | [] -> s
29   | Neg _::tail -> signList tail (-1*s)
30   | _::tail -> signList tail s
31
32
33 // flattens the tree to a commutativ list
34 let rec flatTree e =
35   match e with
36   | N _ | X _ | Add _ | Sub _ -> [e]
37   | Neg a -> Neg (N one) :: flatTree a
38   | Mul (a, b) -> flatTree a @ flatTree b
39   | Div (N a, N b) -> [N (a / b)]
40   | Div (Neg a, b) | Div (a, Neg b) -> Neg (N one) :: flatTree (Div (a, b))
41   | Div (a, N b) -> N (one / b) :: flatTree a
42   | Div (a, b) -> Div (N one, b) :: flatTree a
43
44
45 // multiplies all Div elements in a commutativ list

```

---

```

46 let rec mulDivElements l =
47   // division will always be in the end of a sorted list, and numarator will
   be 1
48   match l with
49   | [] -> []
50   | Div (_, b) :: Div (_, c) :: tail -> mulDivElements (Div(N one, Mul(b, c)
   ) :: tail)
51   | x::tail -> x :: mulDivElements tail
52
53 // removes a element from a list
54 let rec removeElem e l =
55   match l with
56   | [] -> []
57   | x::tail when x = e -> tail
58   | x::tail -> x :: removeElem e tail
59
60 // reduces a commutativ list
61 let rec reduce l =
62   let sorted = divCancelling (sort l)
63   if signList sorted 1 > 0 then rebuildTree sorted else Neg (rebuildTree
   sorted)
64
65 // initiates the division cancelling
66 and divCancelling l =
67   // printfn "DivCancelling: %A" l
68   match List.rev l with
69   | [] -> l
70   | Div(_, b)::tail ->
71     let (numerator, denominator) = cancelEquality (List.
   rev tail) (sort (flatTree b))
72     sort (flatTree (reduce numerator / reduce denominator)
   )
73   | _ -> l
74
75 // cancels out equal elements in the numerator and denominator
76 and cancelEquality nu de =
77   // printfn "cancelEquality: %A / %A" nu de
78   match nu with
79   | [] -> ([], de)
80   | n::ntail when List.contains n de -> cancelEquality ntail (removeElem n
   de)
81   | n::ntail -> let (numerator, denominator) = cancelEquality ntail de
82     (n::numerator, denominator)
83
84
85
86 // rebuilds a commutativ list to a tree
87 and rebuildTree l =
88   match l with
89   | [] -> N one
90   | Neg (N a)::tail when Number.isOne a -> rebuildTree tail
91   | Neg a::tail -> rebuildTree (a::tail)
92   | N a::N b::tail -> rebuildTree (N (a * b) :: tail)
93   | N a :: Add(b, c) :: tail -> rebuildTree (Add(reduce (flatTree (Mul(N a,
   b))), reduce (flatTree (Mul(N a, c)))) :: tail)
94   | N a :: Sub(b, c) :: tail -> rebuildTree (Sub(reduce (flatTree (Mul(N a,
   b))), reduce (flatTree (Mul(N a, c)))) :: tail)
95   | a::tail -> a * rebuildTree tail
96

```

---

```

97
98 // applies the commutativ rules to a tree
99 let applyCommutative e:Expr<Number> =
100     match e with
101     | Mul _ | Div _ -> reduce (flatTree e)
102     | _ -> e

```

Listing 68: CommutativeMulDiv.fs

## 8.5 Matrix.fs

```

1 module Matrix
2 open Number
3 open Expression
4 open SymbolicManipulation
5
6 // row or coloumn major ordere: default of a vector and matric is Column major
  // order.
7 type Order = | R | C
8
9 // Rows x Cols
10 type Dimension = D of int * int
11
12 // a "Normal" is C major and a Transposed one is R major
13 type Vector = V of list<Number> * Order
14 type Matrix = M of list<Vector> * Order
15
16 // Construct a vector
17 let vector nl = V (nl, C)
18
19 // The dimension of a vector
20 let dimVector (V(v, o)) =
21     let len = List.length v
22     match o with
23     | R -> D (1, len)
24     | C -> D (len, 1)
25
26 // Makes sure that a matrix has the same major order as its vectors
27 let matrixValidMajor (M(m, o)) =
28     match List.head m, o with
29     | V (_, R), R -> true
30     | V (_, C), C -> true
31     | _, _ -> failwith "Matrix's vector's dont have same major order"
32
33 // The list length of a vector
34 let vectorLength (V(v, _)) = List.length v
35
36 // The list length of a matrix's vectors
37 let matrixVectorLength (M(m, _)) =
38     match m with
39     | [] -> 0
40     | x::_ -> vectorLength x
41
42 // The dimension of a matrix
43 let dimMatrix (M(vl, o)) =
44     if vl = [] then D (0, 0)
45     else
46         let _ = matrixValidMajor (M(vl, o))

```

---

```

47 let d1 = List.length v1
48 let d2 = matrixVectorLength (M(v1, o))
49 match o with
50 | R -> D (d1, d2)
51 | C -> D (d2, d1)
52
53 // Multiplication of a scalar and a vector
54 let scalarVector (n:Number) (V (nl, o)) =
55   V ((List.map (fun x -> x * n) nl), o)
56
57 // Multiplication of a scalar and a matrix
58 let scalarMatrix (M (v1, o)) n =
59   M ((List.map (fun x -> scalarVector n x) v1), o)
60
61 // Addition two vectors
62 let addVector x y =
63   let (V (v1, o1)) = x
64   let (V (v2, o2)) = y
65   if o1 <> o2
66   then failwith "Vectors must have the same major order"
67   elif dimVector x <> dimVector y
68   then failwith "Vectors must have the same dimension"
69   else
70     V ((List.map2 (+) v1 v2), o1)
71
72 // Subtraction of two vectors
73 let subVector x y =
74   scalarVector (-one) y |> addVector x
75
76
77 // Construct a vector of n with length len
78 let vectorOf n len = V ((List.init len (fun _ -> n)), C)
79
80 // Construct a matrix of n with dimension d
81 let matrixOf n (D (r, c)) = M ((List.init c (fun _ -> vectorOf n r)), C)
82
83 // Transposes a vector
84 let transposeVector (V(v, o)) =
85   match o with
86   | R -> V(v, C)
87   | C -> V(v, R)
88
89 // adds a dimension to a vector
90 let extendVector (V(v, o)) n =
91   V(v @ [n], o)
92
93 // Extend a matrix with a number list
94 let extendMatrix (M(m, o)) nl =
95   if m <> [] && nl <> [] && List.length nl <> matrixVectorLength (M(m, o))
96   then failwith "The list must have the same length as the matrix's vectors"
97   elif nl = [] then M(m, o)
98   else
99     M (m @ [(V (nl, o))], o)
100
101 // extend a matrix with a matrix
102 let extendMatrixWithMatrix (M(m1, o1)) (M(m2, o2)) =
103   if o1 <> o2 then failwith "Matrices must have the same major order"
104   elif matrixVectorLength (M(m1, o1)) <> matrixVectorLength (M(m2, o2)) then
105     failwith "Matrices must have the same dimension"

```

```

104     else M (m1 @ m2, o1)
105
106
107 // Alternates the Major order
108 let alternateOrder o =
109     match o with
110     | R -> C
111     | C -> R
112
113 // Alternates the Major order of a matrix
114 let alternateOrderMatrix (M(m, o)) =
115     M (m, alternateOrder o)
116
117 // The i'th number of a vector
118 let getVectorIthNumber i (V(v, _)) = v.[i]
119
120 // The i'th vector of a matrix
121 let getMatrixIthVector i (M(m, _)) = m.[i]
122
123 // replaces the i'th vector of a matrix
124 let replaceMatrixIthVector i (M(m, o)) v =
125     M (m.[0..i-1] @ [v] @ m.[i+1..], o)
126
127 // pops a vector
128 let seperateFistNumberFromVector (V(v, o)) =
129     (v.Head, V(v.Tail, o))
130
131 // first element of a vector
132 let headVector (V(v, _)) = List.head v
133
134 // Gives a vector of all the first elements of the vectors in a matrix
135 let rec firstElemetsVectors (M(m, o)) v_acc m_acc =
136     match m with
137     | [] -> (v_acc, M(m_acc, o))
138     | x::xs ->
139         let (n, tail) = seperateFistNumberFromVector x
140         firstElemetsVectors (M(xs, o)) (v_acc @ [n]) (m_acc @ [tail])
141
142 // Helper function for changeOrderMatrix
143 let rec chaingingOrderMatrix (M(m, o)) m_acc =
144     match matrixVectorLength (M(m, o)) with
145     | 0 -> m_acc
146     | _ ->
147         let (v, m_new) = firstElemetsVectors (M(m, o)) [] []
148         extendMatrix m_acc v |> chaingingOrderMatrix m_new
149
150 // Changes the order of a matrix
151 let changeOrderMatrix (M(m, o)) =
152     alternateOrderMatrix (M([], o)) |> chaingingOrderMatrix (M(m, o))
153
154 // Makes sure that two matrices have the same major order, if diffrent then
155 // changes the order of the second matrix
156 let giveMatrixHaveSameOrder (M(_, o1)) (M(m2, o2)) =
157     if o1 <> o2 then changeOrderMatrix (M(m2, o2)) else M(m2, o2)
158
159 // Transposes a matrix
160 let transposeMatrix (M(vl, o)) =
161     M(List.map (fun x -> transposeVector x) vl, alternateOrder o)

```



```

162 // Adds two matrices elemwise
163 let addMatrix m1 m2 =
164   let m3 = giveMatrixHaveSameOrder m1 m2
165   if dimMatrix m1 <> dimMatrix m3
166   then failwith "addMatrix: Matrices must have the same dimension"
167   else
168     let (M(vl1, o)) = m1
169     let (M(vl3, _)) = m3
170     M (List.map2 addVector vl1 vl3, o)
171
172 // Construct a matrix
173 let matrix vl =
174   let rec mc vl m =
175     match vl with
176     | [] -> m
177     | (V(x, _))::xs -> mc xs (extendMatrix m x)
178   mc vl (M([], C))
179
180 // Changes the order of af matrix to x
181 let correctOrder (M(m, o)) x =
182   if o = x then M(m, o) else changeOrderMatrix (M(m, o))
183
184 // Boolean value of if a matrix has the correct order
185 let corectOrderCheck (M(_, o)) x =
186   o = x
187
188 // Sum the rows of a matrix
189 let rec sumRows m =
190   if not <| corectOrderCheck m C
191   then sumRows <| correctOrder m C
192   else
193     let zeroVector = vectorOf zero <| matrixVectorLength m
194     let (M(vl, _)) = m
195     matrix [List.fold (addVector) zeroVector vl]
196
197
198
199 // matrix vector product  $A.v = b$  - Definition 7.10
200 let rec matrixVectorProduct m v =
201   let (D(rv, _)) = dimVector v
202   let (D(_, cm)) = dimMatrix m
203   if rv <> cm
204   then failwith "matrixVector: the number of columns of the matrix has to be
205     the same as the number of entries in the vector."
206   elif not <| corectOrderCheck m C then matrixVectorProduct (correctOrder m
207     C) v
208   else
209     let (M(vl, _)) = m
210     let (V(nl, _)) = v
211     M (List.map2 (fun mc n -> scalarVector n mc) vl nl, C)
212     |> sumRows
213
214 // Converts a matrix to a vector if possible
215 let matrixToVector m =
216   match dimMatrix m, m with
217   | D(r, c), M(v::_, _) when r = 1 || c = 1 -> v
218   | _, _ -> failwith "mactrixToVector: Matrix is not a vector"
219
220 // Multiplies two matrices - Definition 7.12

```

---

```

219 let rec matrixProduct a b =
220     let (D(_, ca)) = dimMatrix a
221     let (D(rb, _)) = dimMatrix b
222     if ca <> rb
223     then failwith "matrixProduct: matrix product A .* B is defined only if the
224         number of columns of A is the same as the number of rows of B"
225     elif not <| correctOrderCheck b C then matrixProduct a (correctOrder b C)
226     else
227         let (M(vlb, _)) = b
228         let product = List.map (
229             fun bv -> matrixVectorProduct a bv |> matrixToVector ) vlb
230         M(product, C)
231
232 type Vector with
233     static member (+) (v1, v2) = addVector v1 v2
234     static member (-) (v1, v2) = subVector v1 v2
235     static member (*) (n, v) = scalarVector n v
236     static member (*) (v, n) = scalarVector n v
237     static member (~-) (v) = scalarVector (-one) v
238
239 type Matrix with
240     static member (+) (m1, m2) = addMatrix m1 m2
241     static member (+) (m, n) = dimMatrix m |> matrixOf n |> addMatrix m
242     static member (+) (n, m) = dimMatrix m |> matrixOf n |> addMatrix m
243     static member (*) (n, m) = scalarMatrix m n
244     static member (*) (m, n) = scalarMatrix m n
245     static member (*) (m, v) = matrixVectorProduct m v
246     static member (*) (m1, m2) = matrixProduct m1 m2
247     static member (/) (m, n) = scalarMatrix m (inv n)
248
249
250 // Horizontal flip of a matrix
251 let flip m =
252     let (M(m_new, _)) = correctOrder m C
253     List.rev m_new |> matrix
254
255 // extracts the first vector of a matrix
256 let extractfirstVector (M(m, o)) =
257     match m with
258     | [] -> failwith "Matrix is empty"
259     | x::xs -> x, M(xs, o)
260
261 // Last vector of a matrix
262 let rec extractlastVector m =
263     let (v, mf) = flip m |> extractfirstVector
264     (v, flip mf)
265
266 // Multiplies two vectors element wise
267 let vectorMulElementWise (V(u, o1)) (V(v, o2)) =
268     if o1 <> o2
269     then failwith "Vectors must have the same major order"
270     elif List.length u <> List.length v
271     then failwith "Vectors must have the same dimension"
272     else
273         V (List.map2 (*) u v, o1)

```

```

277
278 // Conjugates a vector
279 let conjugateVector (V(v, o)) =
280     V (List.map conjugate v, o)
281
282 // Conjugates a matrix
283 let conjugateMatrix (M(m, o)) =
284     M (List.map conjugateVector m, o)
285
286 // Inner product of two vectors
287 let innerProduct u v =
288     let (V(w, _)) = conjugateVector v |> vectorMulElementWise u
289     List.fold (+) zero w
290
291 // Projection of a vector on another vector
292 let proj y x =
293     scalarVector (innerProduct x y / innerProduct y y) y
294
295 // Dot product of two vectors
296 let dotProduct (V(u, ou)) (V(v, ov)) =
297     if List.length u <> List.length v
298     then failwith "dotProduct: Vectors must have same length."
299     else
300
301         match ou, ov with
302         | R, C -> innerProduct (V(u, C)) (V(v, C))
303         | _ -> failwith "Missing implementation for this case."
304
305
306 // Orthogonal basis using the Gram-Schmidt process
307 let rec orthogonalBasis m =
308     if not <| correctOrderCheck m C
309     then orthogonalBasis (correctOrder m C)
310     else
311
312         // Gram-Schmidt process but without the normalization
313         let rec Gram_Schmidt vm acc_wm =
314             match acc_wm [], vm with
315             | x, M([], _) -> x
316             | M([], _), M(v1::vrest, o) ->
317                 Gram_Schmidt (M(vrest,o))
318                 <| fun x -> extendMatrix (M([v1], C)) x
319             | M(w, _), M(vk::vrest, o) ->
320                 let (V(wk, _)) = vk - sumProj w vk
321                 Gram_Schmidt (M(vrest,o))
322                 <| fun x -> extendMatrix (acc_wm wk) x
323
324         // Sum all the projections of vk on w1 to wk-1
325         and sumProj w vk =
326             List.map (fun x -> proj x vk) w
327             |> matrix
328             |> sumRows
329             |> matrixToVector
330
331         Gram_Schmidt m (fun _ -> M([], C))
332
333
334
335 // Checks if a vector is a zero vector

```

---

```

336 let isZeroVector (V(v, _)) =
337     List.forall (fun x -> Number.isZero x) v
338
339 // Checks if a matrix is a zero matrix
340 let isZeroMatrix (M(m, _)) =
341     List.forall (fun x -> isZeroVector x) m
342
343 // first non zero element of a vector
344 let firstNonZero (V(v, _)) =
345     if isZeroVector (V(v, C)) then failwith "firstNonZero: Vector does not
346         have a non zero element"
347     else
348         List.find (fun x -> not (Number.isZero x)) v
349
350 // Index of the first non zero element of a vector
351 let firstNonZeroIndex (V(v, _)) =
352     if isZeroVector (V(v, C)) then -1
353     else
354         List.findIndex (fun x -> not (Number.isZero x)) v
355
356 // string a vector
357 let stringVector (V(v, o)) =
358     let space = if o = C then "\n" else " "
359     List.map (fun x -> toString x) v |> String.concat space
360
361 // string a matrix
362 let rec stringMatrix m =
363     if not <| correctOrderCheck m R then stringMatrix (correctOrder m R)
364     else
365         let (M(vl, _)) = m
366         List.map (fun x -> stringVector x) vl |> String.concat "\n"
367
368 // Index of the first non zero element of a matrix
369 let rec firstNonZeroIndexMatrix (M(m, o)) idx (r, c) =
370     if m = [] then (r, c) else
371         let fnxi = List.head m |> firstNonZeroIndex
372         match m with
373         | [] -> (r, c)
374         | _::vt when fnxi >= 0 && (fnxi < c || c = -1) -> firstNonZeroIndexMatrix
375             (M(vt, o)) (idx + 1) (idx, fnxi)
376         | _::vt -> firstNonZeroIndexMatrix (M(vt, o)) (idx + 1) (r, c)
377
378 // Switches two vectors in a matrix
379 let swapFirstWith (M(m, o)) i =
380     if i = 0
381     then M(m, o)
382     else
383         let rec swapper h m i idx acc_m =
384             match m with
385             | [] -> failwith "swapFirstWith: Index out of range"
386             | x::xs when idx = i -> x :: acc_m @ (h::xs)
387             | x::xs -> swapper h xs i (idx + 1) (acc_m @ [x])
388
389         match m with
390         | [] -> failwith "swapFirstWith: Matrix is empty"
391         | h::tail -> M(swapper h tail i 1 [], o)
392

```

---

```

393 // Multiplies the i'th vector with a scalar
394 let scalarIthVector c i (M(m, o)) =
395     let rec sIV c m i idx acc_m =
396         match m with
397         | [] -> failwith "scalarIthVector: Index out of range"
398         | x::xs when idx = i -> acc_m @ c * x :: xs
399         | x::xs -> sIV c xs i (idx + 1) (acc_m @ [x])
400
401     M(sIV c m i 0 [], o)
402
403 // Alters row j with Rj <- Rj - c * Ri
404 let rec rowOperation i j c m =
405     if i = j then failwith "rowOperation: Row i and j must be different j <> i"
406     elif not <| correctOrderCheck m R then rowOperation i j c <| correctOrder m
407     R
408     else
409         replaceMatrixIthVector (j-1) m <| getMatrixIthVector (j-1) m - c *
410         getMatrixIthVector (i-1) m
411
412 // row echelon form of a matrix
413 let rec rowEchelonForm A =
414     if not <| correctOrderCheck A R then rowEchelonForm (correctOrder A R)
415     else
416         let (D(r, c)) = dimMatrix A
417         match A with
418         | M([], _) -> A
419         | _ when isZeroMatrix A -> A
420         | M(v::_, _) when r = 1 -> firstNonZero v |> inv |> scalarMatrix A
421         | M(_, o) ->
422             let (i, j) = firstNonZeroIndexMatrix A 0 (-1, -1)
423             let (M(B, _)) = swapFirstWith A i
424             let b = List.head B |> firstNonZero
425             let (M(B, _)) = scalarIthVector (inv b) 0 (M(B, o))
426             let R1 = List.head B
427             let R2m = List.tail B
428             let B = rowOps j 1 r R1 (M(R2m, o))
429             let (M(Cm, _)) = rowEchelonForm B
430             M(R1::Cm, o)
431
432 // Ri <- Ri - b * R1 - possible error i mat 1 notes Algorithm 1 (should be b <-
433 // the jth entry of the ith row if B)
434 and rowOps coloumn i n rows R1 acc_m =
435     if i >= n rows then acc_m else
436     let Ri = getMatrixIthVector (i - 1) acc_m
437     let b = getVectorIthNumber coloumn Ri
438     rowOps coloumn (i + 1) n rows R1 <| replaceMatrixIthVector (i-1) acc_m (Ri
439     - b * R1)
440
441 // A standard basis vector of length n with 1 at i
442 let standardBasisVector n i =
443     let rec sbv idx =
444         match idx with
445         | _ when idx < 1 -> []
446         | 1 when i <> 1 -> [zero]
447         | x when x = i -> one :: sbv (x - 1)
448         | _ -> zero :: sbv (idx - 1)
449     vector <| sbv n

```

```

448
449 // A standard basis matrix of  $F^n$ 
450 let standardBasis n =
451   let rec sb idx =
452     match idx with
453     | _ when idx < 1 -> []
454     | 1 -> [standardBasisVector n 1]
455     | _ -> standardBasisVector n idx :: sb (idx - 1)
456   matrix <| sb n
457
458 // fullranked diagonal matrix
459 let fullrankedDiagonalMatrix m n =
460   if m > n then failwith "fullrankedDiagonalMatrix: number of rows must be
less than or equal to the number of columns"
461   else
462     let rec frdm i frm =
463       match i with
464       | _ when i < 0 -> failwith "fullrankedDiagonalMatrix: i must be
greater than or equal to 0"
465       | 0 -> frm
466       | _ ->
467         let (V(zeroV, _)) = vectorOf zero m
468         frdm (i - 1) <| extendMatrix frm zeroV
469     frdm (n - m) <| standardBasis m
470
471
472 // Checks if a matrix is upper triangular
473 let rec isUpperTriangular (M(vl, o)) =
474   if not <| correctOrderCheck (M(vl, o)) R then isUpperTriangular (
correctOrder (M(vl, o)) R)
475   else
476     let (D(_, m)) = dimMatrix (M(vl, o))
477     let rec iut vl idx =
478       match vl with
479       | [] -> true
480       | x::xs when idx >= m -> isZeroVector x && iut xs (idx + 1)
481       | x::xs -> firstNonZeroIndex x = idx && iut xs (idx + 1)
482     iut vl 0
483
484 // rank of a matrix
485 let rank m =
486   let (D(r, c)) = dimMatrix m
487   let maxRank = if r < c then r else c
488   let rec ra m rowi =
489     if rowi > maxRank || isZeroMatrix m then
490       rowi
491     else
492       match m with
493       | M([], _) -> rowi
494       | M(v::vl, o) ->
495         let fnz = firstNonZeroIndex v
496         if fnz < rowi then
497           ra (M(vl, o)) rowi
498         else
499           ra (M(vl, o)) (rowi + 1)
500   ra (correctOrder (rowEchelonForm m) R) 0
501
502 // determines if a matrix has full rank
503 let hasFullRank m =

```

```

504 let (D(r, c)) = dimMatrix m
505 let maxRank = if r < c then r else c
506 rank m = maxRank
507
508 // determines if a list of numbers is all zero
509 let isZeroNumberList nl =
510   List.forall (fun x -> Number.isZero x) nl
511
512 // determines if a matrix is Identity matrix
513 let isDiagonalMatrix m =
514   let (D(r, c)) = dimMatrix m
515   if r <> c then false
516   else
517     let rec iim m =
518       let nl, (M(vl, o)) = firstElemetsVectors m [] []
519       match vl, nl with
520       | [], [] -> true
521       | v::vTail, n::nTail ->
522         isZeroVector v && isZeroNumberList nTail && not (Number.isZero n)
523         && iim (M(vTail, o))
524       | _, _ -> false
525     iim m
526
527 // checks is if every vector has inner product of zero with the next vector
528 let rec isOrthogonalBasis m =
529   (transposeMatrix m |> conjugateMatrix) * m |> isDiagonalMatrix
530
531 // split a matrix into two matrices
532 let rec splitMatrix o i m =
533   if not <| correctOrderCheck m o then splitMatrix o i (correctOrder m o)
534   else
535     let (M(vl, _)) = m
536
537     let rec sm vl idx acc =
538       match vl with
539       | [] -> failwith "splitMatrix: Index out of range"
540       | x::xs when idx = 0 -> matrix <| acc @ [x], matrix xs
541       | x::xs -> sm xs (idx - 1) (acc @ [x])
542     sm vl i []
543
544 // Determines if a matrix has a zero column
545 let rec dontHaveZeroCols m =
546   if not <| correctOrderCheck m C then dontHaveZeroCols (correctOrder m C)
547   else
548     let (M(vl, _)) = m
549     List.forall (fun x -> not <| isZeroVector x) vl
550
551 // Determines if a matrix has the same span as another matrix
552 let hasSameSpan m1 m2 =
553   let (D(r, c)) = dimMatrix m1
554   if (D(r, c)) <> dimMatrix m2 then failwith "Matrices must have the same dimension"
555   else
556
557     let hss m1 m2 =
558       let s1, s2 = extendMatrixWithMatrix m1 m2 |> rowEchelonForm |>
559         splitMatrix C (c-1)
560       hasFullRank s1 && hasFullRank s2

```

```

560      hSS m1 m2 && hSS m2 m1
561
562
563
564
565
566
567 ///////////////////////////////////////////////////
568 /// Functions to solve Ax = b ///
569 ///////////////////////////////////////////////////
570
571 type ExprVector = list<Expr<Number>>
572 type ExprMatrix = list<ExprVector>
573
574
575
576 // multiplies a expression with a vector and returns a Expr list
577 let scalarWithExpr (V(nl, _)) e =
578     List.map (fun x -> N x * e) nl
579
580 // matrix product with expression vector
581 let matrixProductExprList vl el =
582     let znl = List.head vl |> vectorLength |> vectorOf zero
583     let zeroExprList = scalarWithExpr znl (N zero) // zero expr vector
584     List.map2 (fun mc n -> scalarWithExpr n mc) el vl
585     |> List.fold (fun a b -> (List.map2 (+) a b)) zeroExprList
586
587 // Creates a vector of n variables
588 let rec charVector n =
589     match n with
590     | _ when n <= 0 -> []
591     | _ -> X (char (n - 1)) :: charVector (n - 1)
592
593 // inserts an enviroment into a vector
594 let rec vectorEnv n env =
595     match n with
596     | _ when n <= 0 -> V([], C)
597     | _ ->
598         let (V(nl, _)) = vectorEnv (n - 1) env
599         Map.find (char (n - 1)) env :: nl |> vector
600
601 // solves the system of lineary equations
602 let rec solveEquations el bl cl =
603     match el, bl, cl with
604     | [], [], [] -> Map.empty
605     | e::es, b::bs, _ when isZero e && isZero b ->
606         solveEquations es bs cl
607     | e::es, b::bs, c::cs ->
608         let env = solveEquations es bs cs
609         let (lhs, rhs) = isolateX (insertEnv e env) b c
610         Map.add (getVariable lhs) (getNumber rhs) env
611     | _, _, _ -> failwith "solveEquations: The number of equations and
612         variables must be the same"
613
614 // Solves the equation Ax = b
615 let matrixEquation A (V(nlb, ob)) =
616     let (D(r, c)) = dimMatrix A
617     if r <> List.length nlb || ob = R then
618         failwith "matrixEquation: b must be a column vector or have same

```



---

```

length as rows of A"
618
619 // Perform row echelon form on the total matrix
620 let ef_t = correctOrder (rowEchelonForm <| extendMatrix A n1b) C
621
622 // Extract the last vector of the matrix
623 let (V(ef_b, _), M(ef_v1, o)) = extractlastVector ef_t
624
625 // Check if the system of equations has a solution
626 if hasFullRank ef_t then
627     failwith "matrixEquation: The system of equations has no solution"
628 elif not <| hasFullRank (M(ef_v1, o)) then
629     failwith "matrixEquation: The system of equations has infinite
solutions"
630
631 let varlist = charVector c
632 let b = scalarWithExpr (vector ef_b) (N one)
633 solveEquations (matrixProductExprList ef_v1 varlist) b varlist |>
vectorEnv c

```

Listing 69: Matrix.fs