# Danmarks Tekniske Universitet



# Bachelor Projekt

Funktionel Modellering af Matematiske Systemer i F#

Jonas Dahl Larsen (s205829)

8. april 2024

# ${\bf Indhold}$

1	Intr	$\operatorname{roduktion}$	2
2	Fun	damentale koncepter	2
	2.1	Introduktion til Funktions Programmering	2
		2.1.1 Typer	3
	2.2	Signatur filer og implementerings filer	3
	2.3	Overloading operatorer	4
	2.4	Property Based Testing	4
3	Svn	nbolske lignings udtryk	5
	3.1	Tal mængder	5
		3.1.1 Rationelle tal Mondul	5
		3.1.2 Komplekse tal Mondul	6
		3.1.3 Tal Mondulet	6
	3.2	Matematiske ligninger	8
		3.2.1 Polsk notation	8
		3.2.2 Ligninger som træer	8
		3.2.3 Ligningsudtryk mondulet	10
		3.2.4 Generering af Ligningsudtryk	12
	3.3	Simplifikation af Ligningsudtryk	$12^{-1}$
		3.3.1 Evaluering af ligningsudtryk	$12^{-1}$
		3.3.2 PBT af simplifikationen	12
	3.4	differentiering af Ligningsudtryk	12
4	<b>1</b> /21-	torer og Matricer	12
4	4.1	Matrix operationer	13
	4.1	4.1.1 Matematiske operationer	13
		4.1.1.1 Skalering af en matrix	13
		4.1.1.2 Addition af matricer	13 14
			14 15
		±	
	4.0		17
	4.2	PBT af matrix operationer	18
		Række-echelon form	19
	4.4	Gram-Schmidt	19
	4.5	PBT af Gram-Schmidt	20
5	App	pendiks	22
	5.1	Matrix fs	22

## 1 Introduktion

I 2023 valgte Danmarks Tekniske Universitet at anvende Python som et hjælpeværktøj i deres grundlæggende matematikkursus "01001 Matematik 1a (Polyteknisk grundlag)". Python er et af de mest anvendte programmeringssprog<sup>1</sup>, kun overgået af to sprog, der primært bruges sammen til at udvikle hjemmesider. Derfor har Python, med en række matematiske programudvidelser som SymPy<sup>2</sup>, været et oplagt valg som programmeringssprog til det grundlæggende matematikkursus tilbudt af DTU.

Projektet vil undersøge, hvordan et funktionsprogrammeringssprog, kan gavne de studerendes forståelse af de grundlæggende matematiske koncepter. Formålet er at guide læseren gennem opbygningen af en række funktionsprogrammer baseret på grundlæggende matematik³ og dermed illustrere anvendelser. Projektet beskriver en generel struktur til opbygning og anvendelse af et funktions programmeringsprogram. Der tages udgangspunkt i  $F\#^4$ , men beskrivelserne af programmerne vil også kunne anvendes i lignende funktionsprogrammeringssprog.

Rapporten begynder med at forklare nogle Fundamentale koncepter inden for funktionsprogrammering samt metoder til validering af programmerne.

# 2 Fundamentale koncepter

### 2.1 Introduktion til Funktions Programmering

Det er forventet af læseren har kendskab til programmering, der gives derfor kun en kort beskrivelse af syntaks og notationen, så læser ikke bekendt med F# kan forstå de eksempler der løbende vil forkomme i rappoten.

$$f(n) = \begin{cases} 1 & n = 0\\ n \cdot f(n-1) & n > 0\\ \text{undefined} & n < 0 \end{cases}$$
 (1)

Vi begynder derfor med at betragte funktionen for fakultet (1), et eksempel på en implementering i F# er givet i Listing 1 som kan sammelignes med Python kode i Listing 2, da Python og pseudokode er næsten det samme.

Listing 1: Eksempel på Fakultet i F#

<sup>&</sup>lt;sup>1</sup>Most used programming languages among developers worldwide as of 2023.

 $<sup>^2</sup>SumPu$ 

 $<sup>^3</sup> Kursus\ side\ for\ Mathematics\ 1a.$ 

 $<sup>^4</sup>$ fsharp.org.

```
// Fakultet i Python
def factorial(n):
    if n == 0:
        return 1
    elif n > 0:
        return n * factorial(n - 1)
    else:
        raise ValueError("Negative argument")
```

Listing 2: Eksempel på Fakultet i Python

I F# anvendes let til at definere en ny variabel eller, i dette tilfælde, en funktion kaldet factorial. Næste nøgleord er rec, hvilket indikerer, at funktionen er rekursiv. Funktionen tager et inputargument n, og i linje 3 starter et match-udtryk. Her er n vores udtryk, og efter with begynder en række mønstre, som udtrykket forsøger at genkende på, separeret med '|'. Resultatet for funktionen vil blive koden som er eksekveret efter ' $\rightarrow$ '.

#### 2.1.1 Typer

Typer er tildelt ved kompilering, i modsætning til python som kør det under kørsel. Alle udtryk funktioner inkluderet har en type i F#, typen for Listing factorial1 er  $int \rightarrow int$ . Det er derfor ikke muligt at kalde funktionen med et argument der ikke er af typen int. Typen for funktionen skrives som  $Factorial: int \rightarrow int$ . Vi kan dermed formulere følgende omkring typer<sup>5</sup>:

$$f: T_1 \to T_2$$

$$fe: T_2 \iff e: T_1$$

Er en funktion kaldt med et argument der ikke passer til funktionens type, gives en fejlmeddelelse. Derudover kan en type også være bestående at en tuple af typer.

$$f: T_1*T_2*..*T_n \to T_{n+1}$$
 
$$f(e_1,e_2,..,e_n): T_{n+1} \iff e_1: T_1 \wedge e_2: T_2 \wedge .. \wedge e_n: T_n$$

En tuple som kun består af to typer, kaldes et par.

I F# er det ikke nødvendigt at anvende parenteser som i andre programmeringssprog. De vil derfor kun anvendes hvor det er nødvendige, gennem rapporten. Når man laver en tuple er det nødvendigt. Givet en funktion  $g:T_1\to T_2\to T_3$  betyder den tager et udtryk af typen  $T_1$  som giver en funktion af typen  $T_2\to T_3$  hvor evalueringen af funktionen giver  $T_3$ 

# 2.2 Signatur filer og implementerings filer

En standard F# fil er lavet med .fs extension, denne fil indeholder alt den kode som er nødigt for at kunne køre programmet. En implementerings fil kan have en signatur fil med .fsi extension, denne fil indeholder en beskrivelse af de typer og funktioner i implementerings filen som er tilgængelige for andre filer. En signatur fil kan derfor bruges som et blueprint for andre der ønsker at anvende eller replicere implementerings filen. I andre programmerings sprog vil man anse funktionerne i signatur filen som værende "public" og de funktioner der ikke er i signatur filen, men er i implementerings filen som værende "private".

<sup>&</sup>lt;sup>5</sup>Functional Programming Using F#, s. 14.

# 2.3 Overloading operatorer

I F# er det muligt at overskrive standard operatorer, så man kan anvende dem på egne typer. Det vil igennem rapporten blive anvendt til at definere matematiske operationer på de typer som vi kommer til at bygge.

### 2.4 Property Based Testing

Property Based Test (PBT) er en teknik til at teste korrekthed af egenskaber som man ved altid skal være opfyldt. En PBT test generer en række tilfældige input til en funktion og tester om en egenskab er opfyldt. Hvis en egenskab ikke er opfyldt, vil PBT give et eksempel på en fejl. De matematiske studerende på DTU, begynder med at lære om logik. I den forbindelse lærer man at en udsagnslogisk formel er gyldig (tautologi) hvis den altid er sand. Der eksistere mange teknikker til at vise at en udsagnslogisk formel er gyldig, i DTU's matematiske kursus lærer man at anvende sandhedstabellen. De viser hvordan 2 er gyldig.

$$P \wedge (Q \wedge R) \iff (P \wedge Q) \wedge R$$
 (2)

Vi vil også kunne anvende PBT til at undersøge om (2) er gyldig, ved at udtrykke egenskaben som en funktionen af P, Q og R se Listing 3 3.

Listing 3: PBT af (2), for at undgår shortcircuting har begge sider af lighedstegnet omgivet af parenteser.

```
> let _ = Check.Quick propositional_formula;;
0k, passed 100 tests.
```

Listing 4: Output ved PBT af (2)

Check.Quick er en del af "FsCheck"biblioteket, den tager en funktion som argument, og generere en række tilfældige input til funktionen på baggrund af funktionens type. Hvis funktionen returnere "true"for alle input, vil testen lykkedes. Hvis funktionen returnere "false"for et input, vil testen fejle og give et eksempel på et input der fejlede. I Listing 3 er der anvendt "Check.Quick"til at teste om (2) er gyldig. Funktionen "Check.Quick"returnere "Ok, passed 100 tests."hvilket indikerer at (2) er gyldig. Det vigtigt her at forstå dette ikke er det samme som at bevise at den er gyldig, da ikke alle muligheder er blevet testet. I dette tilfælde kan vi regne sandsynligheden for hvor vidt alle kombinationer er testet.  $P, Q, R \in \{True, False\}$ , derfor er der  $2^3 = 8$  mulige kombinationer for input til funktionen. Hvis vi antager at alle kombinationer er lige sandsynlige, er sandsynligheden for at alle kombinationer er blevet testet  $1 - (\frac{7}{8})^{100} = 0.999998$ . Derfor er det meget sandsynligt at (2) er gyldig.

Det vil generelt ikke være muligt at regne denne sandsynlighed, da vi senere vil anvende PBT til at teste funktioner der tager argumenter som ikke har et endeligt antal kombinationer. Dog vil det stadig give en god indikation hvorvidt en egenskab er overholdt. I nogle tilfælde vil det være

en fordel at opskrive en PBT før implementeringen af en funktion som man ved skal overholde en egenskab, på den måde anvende Test Driven Development  $(TDD)^6$  til at teste om ens egenskab forbliver overholdt, under implementering.

# 3 Symbolske lignings udtryk

Det ønskes at kunne repræsentere simple ligninger som en type i F#. Vil derfor gennemgå en del teori og funktion som er nødvendige for at kunne dette. Det vil give os et grundlæggende fundament for at kunne udføre matematiske evalueringer som differentiering i F#. Som de fleste andre programmer har F# kun float og int som kan repræsentere tal. Derfor vil vi begynde med at definere et mondul som indeholder en type for tal. Tanke gangen her at gennemgå en opbygning af en måde at kunne repræsentere ligninger samt simplificere dem. Vi begrænset os selv til at kun have matematiske operationer som addition, subtraktion, negation, multiplikation og division.

## 3.1 Tal mængder

Vi begynder med opbygningen af et mondul som kan repræsentere tal mængder. Typen for tal, består af tre konstruktører, for henholdvis heltal, rationale tal og komplekse tal. Dog er mondulet lavet med henblik på at kunne udvides med flere typer af tal. Måden resten af programmet er lavet på, gør de eneste grav til tal er at der er definerede matematiske operationer i form af addition, subtraktion, negation, multiplikation og division. Samt at tallet inden for addition og multiplikation er associative. Dette gælder blandt andet ikke for en vektor, derfor vil vi senere betragte at udvide programmet med en type for vektorer. En udvidelse kunne være for reele tal, som kan håndtere "floating point errors", men for ikke at komplicere programmet vil vi i denne opgave ikke betragte floats.

#### 3.1.1 Rationelle tal Mondul

Repræsentationen af rationale tal kan laves ved hjælp af danne et par af integers, hvor den ene integer er tælleren og den anden er nævneren.

```
[language={FSharp},
label={type_rationel},
caption={Typen for rationelle tal}]
type rational = R of int * int
```

Nedestående er der givet en signatur fil for rational mondulet 5. i Implementerings filen overloades de matematiske operatorer, ved hjælp af de klassiske regneregler for brøker<sup>8</sup>.

```
module rational

[<Sealed>]
type rational =
static member (~-): rational -> rational
```

<sup>&</sup>lt;sup>6</sup> Test-Driven Development.

<sup>&</sup>lt;sup>7</sup>Floating-point error.

<sup>&</sup>lt;sup>8</sup>Rational Number wikipedia.

```
static member ( + ) : rational * rational -> rational
      static member ( + )
                           : int * rational -> rational
      static member (
                      - )
                           : rational * rational -> rational
      static member ( - )
                           : int * rational -> rational
9
      static member ( * )
                           : int * rational -> rational
      static member ( * )
                           : rational * int -> rational
      static member ( * )
                           : rational * rational -> rational
12
13
      static member ( / )
                           : rational * rational -> rational
      static member ( / )
                           : int * rational -> rational
14
      static member ( / )
                           : rational * int -> rational
15
      static member ( / )
                          : int * int -> rational
16
                   : int * int -> rational
  val make
  val equal
                   : rational * rational -> bool
19
      posetive
                   : rational -> bool
21 val toString
                   : rational -> string
                   : rational -> bool
22 val isZero
23 val isOne
                   : rational -> bool
24 val isInt
                   : rational -> bool
val makeRatInt
                  : rational -> int
val greaterThan : rational * rational -> bool
val isNegative
                   : rational -> bool
28 val absRational
                  : rational -> rational
```

Listing 5: Signatur filen for rational mondulet

For at kunne sammenligne, men også for nemmere at undgå for store brøker, vil alle rationelle tal blive reduceret til deres simpleste from. Dette kan gøres ved at finde den største fælles divisor (GCD)<sup>9</sup>. Der udover er det vigtigt at være opmærksom på man ikke foretager nul division. Derfor vil implementerings filen kaste en "System.DivideByZeroException" hvis nævneren er eller bliver nul. Signatur filen indeholder en række funktioner som bliver anvendt af andre filer.

#### 3.1.2 Komplekse tal Mondul

#### 3.1.3 Tal Mondulet

Vi har nu beskrevet en måde at kunne repræsentere bruger definere tal på ved brug af typer i F#. Det vil derfor være oplagt at have en type som indeholder alle de typer tal vi ønsker at kunne anvende i de matematiske udtryk vi er ved at opbygge. Fordelen ved at samle dem til en type er at vi kan lave en række funktioner blandt andet matematiske operationer som kan anvendes på alle type tal. Vi begynder med at definere en type for tal 6, som indeholder konstruktører for de tal typer vi har definerede samt en for heltal.

```
type Number = | Int of int | Rational of rational
Listing 6: Typen for Number
```

Betragtes signatur filen for Number mondulet 7, ses det at der igen er defineret overloading af de anvendte matematiske operationer. Derudover er der defineret en række funktioner som kan anvendes på Number typen.

<sup>&</sup>lt;sup>9</sup> Greatest common divisor wikipedia.

```
1 module Number
2 // open rantionalAndComplex
3 open rational
4 open complex
6 type Number =
     | Int of int
      | Rational of rational
      | Complex of complex
      with
10
      static member ( + )
                          : Number * Number -> Number
      static member ( - ) : Number * Number -> Number
      static member ( * ) : Number * Number -> Number
13
      static member ( / ) : Number * Number -> Number
14
      static member ( ~- ) : Number -> Number
15
16
17
18
19
20 val zero
                  : Number
21 val one
                  : Number
22 val two
                  : Number
                  : Number -> bool
23 val isZero
24 val isOne
                  : Number -> bool
val isNegative : Number -> bool
26 val absNumber : Number -> Number
27 val greaterThan : Number -> Number -> bool
28 val tryReduce
                   : Number -> Number
29 val toString
                   : Number -> string
30 val conjugate
                   : Number -> Number
31 val inv
                   : Number -> Number
```

Listing 7: Signatur filen for Number mondulet

Ved implementeringen af de matematiske operationer, hvis der eksistere en konstruktør i Number, der repræsentere en tal mængde hvor alle andre konstruktører er delmængder af denne mængde. Er det muligt at definere en enkelt funktion som kan udføre alle binærer operationer. Som et eksempel er funktionen 8 givet, som tager to tal og en funktion i form af den ønskede binærer operation som parameter. Funktionen vil derefter matche på de to tal og anvende den operation på de to tal.

```
f (makeRational a) (makeRational b) |> Rational
```

Listing 8: Number.operation funktionen

Det vil her til være oplagt på alle de matematiske operationer at anvende en funktionen til at forsøge at konvenere tal typen til den simpleste talmængde, som i vores tilfælde er heltal. Dette er gjort ved at anvende funktionen tryMakeInt på alle de matematiske operations overloadnings 9.

```
// tryMakeInt: Number -> Number
let tryMakeInt r =
    match r with
    | Rational a when isInt a -> Int (makeRatInt a)
    | _ -> r

type Number with
    static member (+) (a, b) = operation a b (+) |> tryMakeInt
    static member (-) (a, b) = operation a b (-) |> tryMakeInt
    static member (*) (a, b) = operation a b (*) |> tryMakeInt
    static member (*) (a, b) = operation a b (*) |> tryMakeInt
    static member (*) (a, b) = operation a b (*) |> tryMakeInt
    static member (*) (a) = neg a |> tryMakeInt
```

Listing 9: Overloadnings funktionerne for Number

Dermed har vi et mondul som kan repræsentere tal, samt udføre matematiske operationer på dem. Vi vil nu begynde at betragte hvordan vi kan anvende den i et lignings udtryk.

## 3.2 Matematiske ligninger

#### 3.2.1 Polsk notation


Matematiske ligningsudtryk som vi normalt kender dem er skrevet med infix notation. I infix notation skrives en binær operator mellem to operandere, kende tegnet for sproget er at det indeholder parenteser samt præcedens regler. Dette gør det generalt kompliceret at evaluere og håndtere matematiske udtryk i et programmeringssprog. Derfor er det mere oplagt at kunne anvende polsk notation (prefix) istedet, hvor operatoren skrives før operandere eller omvendt polsk notation (postfix). Da de hverken indeholder parenteser eller præcedens regler<sup>10</sup>.

```
Infix Notation: (A+B) \cdot C
Prefix Notation: \cdot +ABC
Postfix Notation: AB+C
```

#### 3.2.2 Ligninger som træer

Et matematisk udtryk kan repræsenteres som et binært træ, hvor bladene er operander i det anvendte matematiske rum og alle andre noder er operationer. Som eksempel kan udtrykket  $-(3 \cdot x - 5) \cdot 2$  repræsenteres som følgende træ Figur 1.


 $<sup>^{10}</sup> Polsk\ notation\ wikipedia.$ 



Figur 1: Et binært træ der repræsenterer udtrykke<br/>t $-(3\cdot x - 5)\cdot 2$ 

Det skal bemærkes der er forskel på den unære og binære operator '–' i træet, den unære betyder negation og den binære er subtraktion. Givet et binært træ for en matematisk ligning, vil det være muligt omdanne dem til infix, prefix eller postfix notation. Dette kan gøres ved at anvende modificeret Inorder, Preorder eller Postorder Traversal<sup>11</sup>, algorithmerne er illustreret i Figur 2.

<sup>&</sup>lt;sup>11</sup> Tree Traversal Techniques – Data Structure and Algorithm Tutorials.



Modificeret Inorder Traversal:  $-(3 \cdot x - 5) \cdot 2$ 

Preorder Traversal:  $\cdot - - \cdot 3x52$ Postorder Traversal:  $3x \cdot 5 - -2$ 

Figur 2: Træet fra Figur 1 med forskellige travesal metoder

Vi vil i 3.2.3 betragte hvordan vi kan implementere et mondul som kan repræsentere ligningsudtryk ved brug af prefix notation. Postorder Traversal blive anvendt til at kunne rekursivt simplificere og evaluere ligningsudtryk.

Grundet præcedens regler i infix notation, er det nødvendigt at modificere Inorder Traversal, da unære noder altid skal håndteres før dens børn. Desuden vil det også være nødvendigt at implementere regler for at håndtere parenteser, hvis der ønskes en symbolsk ligning. Den modificeret Inorder Traversal anvendes til at kunne visualisere ligningsudtryk i infix notation.

#### 3.2.3 Ligningsudtryk mondulet

Efter at have udviklet et modul til repræsentation af talmængder, er det nu muligt at videreudvikle et modul til at repræsentere ligningsudtryk. Vi starter med at definere en polymorfisk
type for et ligningsudtryk, som vist i Listing 10. Denne type inkluderer flere konstruktører, der
hver især repræsenterer de matematiske operationer, vi ønsker at anvende. Desuden indfører vi
konstruktøren N til at repræsentere en matematisk struktur; i dette tilfælde anvender vi vores
talmængder (se Listing 6). Det er dog muligt at udvide dette til at omfatte andre matematiske
strukturer, hvor det er muligt at definere de samme operationer. Sidst haves en konstruktør X
til repræsentation af variable.

Listing 10: Typen for Expr

Expr<'a> typen er dermed en polymorfisk type, hvor 'a er typen for den matematiske struktur hvor vi kan lave brugerdefinerede matematiske operationer. Et exemplar på en Expr<Number> er givet i 11.

```
> tree "-(3*x-5)*2";;
val it: Expr<Number> = Mul (Neg (Sub (Mul (N (Int 3), X 'x'), N (Int 5))), N (Int 2))
Listing 11: -(3 \cdot x - 5) \cdot 2 som et udtryks træ. Funktionen tree bliver beskrevet i 3.2.4.
```

Signatur filen indeholder overloadings på de matematiske operationer, så de kan anvendes mellem ligningsudtryk. Samt en funktion eval til at evaluere et ligningsudtryk.

```
1 module Expression
2 open Number
4 type Expr<'a> =
   | X of char
    | N of 'a
    | Neg of Expr<'a>
    | Add of Expr<'a> * Expr<'a>
    | Sub of Expr<'a> * Expr<'a>
    | Mul of Expr<'a> * Expr<'a>
    | Div of Expr<'a> * Expr<'a>
12
    static member ( ~- ) : Expr<Number> -> Expr<Number>
13
    static member ( + ) : Expr<Number> * Expr<Number> -> Expr<Number>
14
    static member ( - ) : Expr<Number> * Expr<Number> -> Expr<Number>
15
    static member ( * ) : Expr<Number> * Expr<Number> -> Expr<Number>
16
17
    static member ( / ) : Expr<Number> * Expr<Number> -> Expr<Number>
18
19 val eval : Expr < Number > -> Map < char , Number > -> Number
20 val containsX : Expr<Number> -> Expr<Number> -> bool
val getNumber : Expr<Number> -> Number
val getVariable : Expr<Number> -> char
```

Listing 12: Signatur filen for Expression mondulet

De overloadede matematiske operatorer i Expressions, laver overflade evalueringer samt simplifikationer på deres respektive argumenter. Overfalde evalueing vil sige at de individuellen funktioner kun betragter de to øverste niveauer på de lignings udtryk træer de tager som input, mullige implementeringer af addition og multiplikation er givet i Listing 13.

```
1 // add: Expr < Number > -> Expr < Number > -> Expr < Number >
 let rec add e1 e2:Expr<Number> =
    match e1, e2 with
    | Na, Nb
                                               \rightarrow N (a + b)
    | N a, b | b, N a when isZero a
    | Mul(a, X b), Mul(c, X d)
    | Mul(X b, a), Mul(c, X d)
    | Mul(a, X b), Mul(X d, c)
    | Mul(X b, a), Mul(X d, c) when b = d \rightarrow Mul(add a c, X b)
                                              -> Add(e1, e2)
10
12 // mul: Expr < Number > -> Expr < Number > -> Expr < Number >
 let mul e1 e2:Expr<Number> =
    match e1, e2 with
    |Na, Nb
                                        \rightarrow N (a * b)
15
    |N a, b | b, N a when isOne a
                                        -> b
16
    |N a, _ | _ , N a when isZero a
                                        -> N zero
17
                                        -> Mul(e1, e2)
```

Listing 13: Addition og multiplikation af to ligningsudtryk

- 3.2.4 Generaring af Ligningsudtryk
- 3.3 Simplifikation af Ligningsudtryk
- 3.3.1 Evaluering af ligningsudtryk
- 3.3.2 PBT af simplifikationen
- 3.4 differentiering af Ligningsudtryk

# 4 Vektorer og Matricer

Vi vil nu betragte et modul for vektorer og matricer. Da en Vector også kan betragtes som en matrix, vil vi herfra når der omtalles begge kun referare til en matrix. For at kunne håndtere matricer på en systematik måde begynder vi med at defienre en type for major order.

Vi vil nu betragte opbyggelsen af et modul for vektorer og matricer. Eftersom en vektor også kan opfattes som en matrix, vil vi i det følgende, når begge dele omtales, udelukkende referere til matricer. For systematik at kunne håndtere matricer, starter vi med at definere en type for lagringsordning.

```
type Order = | R | C
```

Listing 14: Typen for order

Typen Order (se Listing 14), anvendes til at angive, om en matrix er i rækkefølge (row-major) eller kolonnefølge (column-major)<sup>12</sup>. En vektor, der er lagret i kolonnefølge, kan betragtes som den transponeret rækkefølge vektor. Vi kan derfor nu definere en type for matricer, ved hjælp af en type for vektore (Listing 15).

 $<sup>^{12}</sup>Row$ - and column-major order.

```
type Vector = V of list<Number> * Order
type Matrix = M of list<Vector> * Order
```

Listing 15: Typen for Matricer

Derudover er det en fordel at kunne kende dimissionen af en matrix. Derfor er der også defineret en type for dimissionen(se Listing 16).

```
1 // Rows x Cols
2 type Dimension = D of int * int
```

Listing 16: Typen for dimissionen

.-

### 4.1 Matrix operationer

Der vil i denne sektion beskrives en række funktioner som er nødvendige før vi kan betragte nogle rekursive algoritmer som kan anvendes på en matrice. Da modulet indeholder mange hjælpe funktioner, vil der fokuseres på de funktioner med matematisk relevans.

Det muligt at definere en funktion til at finde dimissionen af en matrix (se Listing 17). Funktionen laver et kald til matrixValidMajor genere en fejl hvis ikke alle vektorer og matrien har samme lagringsordning. matrixVectorLength finder længden af en vektor i matricen.

```
// dimMatrix : Matrix -> Dimension
let dimMatrix (M(vl, o)) =
    if vl = [] then D (0, 0)
    else
    let _ = matrixValidMajor (M(vl, o))
    let d1 = List.length vl
    let d2 = matrixVectorLength (M(vl, o))
    match o with
    | R -> D (d1, d2)
    | C -> D (d2, d1)
```

Listing 17: Funktion til at finde dimissionen af en matrix

Hvis en matrix er gemt som rækkefølge, vil antallet af rækker være længden af en vektor og antallet af kolonner være længden af vektor listen, og omvendt for kolonnefølge.

## 4.1.1 Matematiske operationer

I denne section bør bemærkes flere Listings ikke anvender fejlhåndtering, dette er udelukket for læsbarhedens skyld. De anvendte funktioner i mondulet har passende dimensions tjek på matricerne se evt. appendiks 5.1, der sikre at operationerne altid er lovlige.

## 4.1.1.1 Skalering af en matrix

Vi begynder med at betragte en funktion til at skalere en matrix (se Listing 18).

```
// scalarVector : Number -> Vector -> Vector
let scalarVector (n:Number) (V (nl, o)) =
    V ((List.map (fun x -> x * n) nl), o)

// scalarMatrix : Matrix -> Number -> Matrix
let scalarMatrix (M (vl, o)) n =
    M ((List.map (fun x -> scalarVector n x) vl), o)
```

Listing 18: Funktion til at skalere en matrix

Det at skalere en matrice er svare til at skalere hvert element i matricen. Derfor ved at have en funktion scalarVector, der skalere hvert element i en givet vektor bliver scalarMatrix at skalere hver vektor i en givet matrice. List.map svare til at lave en list comprehension i Python<sup>13</sup>.

```
// addVector : Vector -> Vector -> Vector
2 let addVector (V (v1, o1)) (V (v2, _)) =
      V ((List.map2 (+) v1 v2), o1)
5 // addMatrix : Matrix -> Matrix -> Matrix
6 let addMatrix (M(vl1, o)) (M(vl2, _)) =
      M (List.map2 addVector vl1 vl2, o)
9 // subVector : Vector -> Vector -> Vector
 let subVector x y =
      scalarVector (-one) y |> addVector x
12
13 // sumRows : Matrix -> Matrix
14 let rec sumRows m =
      if not <| corectOrderCheck m C
1.5
      then sumRows < | correctOrder m C
16
      else
17
      let zeroVector = vectorOf zero <| matrixVectorLength m</pre>
18
      let (M(vl, _)) = m
19
      matrix [List.fold (addVector) zeroVector vl]
20
```

Listing 19: Funktion til at addere matricer og substraktion af vektorer

#### 4.1.1.2 Addition af matricer

Addition af vektorer kommer ned til at fortage additionen elementvis i Listing 19, ved brug af List funktionen map2. Vi kan bruge addVector til at definere, matrix addition og subtraktion af vektorer sidst bruges den også til at summere rækkerne i en matrice (sumRows) som vil blive brugt i implementeringen af Gram-Schmidt processen i sektion 4.4, funktionen bliver yderligere beskrevet i defination 1.

**Defination 1** (Summering af rækker i en matrix).

 $<sup>^{13}</sup>Python$  - List Comprehension.

Lad A være en matrix med m rækker og n søjler, hvor

$$A = \begin{bmatrix} a_{11} & a_{12} & \cdots & a_{1n} \\ a_{21} & a_{22} & \cdots & a_{2n} \\ \vdots & \vdots & \ddots & \vdots \\ a_{m1} & a_{m2} & \cdots & a_{mn} \end{bmatrix}$$

Så gælder om sumRows at

$$sumRows(A) = v = \begin{bmatrix} \sum_{j=1}^{n} a_{1j} \\ \sum_{j=1}^{n} a_{2j} \\ \vdots \\ \sum_{j=1}^{n} a_{mj} \end{bmatrix}$$

Dermed er  $v_i = \sum_{j=1}^{n} a_{ij} \text{ for } i = 1, 2, ..., m.$ 

#### 4.1.1.3 Matrix multiplikation

**Defination 2** (Matrix-Vektor Multiplikation).

Lad A være en matrix med m rækker og n søjler, hvor

$$A = \begin{bmatrix} a_{11} & a_{12} & \cdots & a_{1n} \\ a_{21} & a_{22} & \cdots & a_{2n} \\ \vdots & \vdots & \ddots & \vdots \\ a_{m1} & a_{m2} & \cdots & a_{mn} \end{bmatrix}$$

og  $\mathbf{v} = (v_1, v_2, \dots, v_n)^T$  være en vektor med n elementer. Så er matrix-vektor Av produktet defineret som

$$\begin{bmatrix} a_{11} & a_{12} & \cdots & a_{1n} \\ a_{21} & a_{22} & \cdots & a_{2n} \\ \vdots & \vdots & \ddots & \vdots \\ a_{m1} & a_{m2} & \cdots & a_{mn} \end{bmatrix} \begin{bmatrix} v_1 \\ v_2 \\ \vdots \\ v_n \end{bmatrix} = \begin{bmatrix} a_{11}v_1 + a_{12}v_2 + \cdots + a_{1n}v_n \\ a_{21}v_1 + a_{22}v_2 + \cdots + a_{2n}v_n \\ \vdots \\ a_{m1}v_1 + a_{m2}v_2 + \cdots + a_{mn}v_n \end{bmatrix} = \begin{bmatrix} \sum_{j=1}^n a_{1j}v_j \\ \sum_{j=1}^n a_{2j}v_j \\ \vdots \\ \sum_{j=1}^n a_{mj}v_j \end{bmatrix}$$

Sætning 1 (Matrix-Vektor Multiplikation).

Lad A være en matrix med m rækker og n søjler, og lad  $\mathbf{v}$  være en vektor med n elementer. Så gælder der

$$Av = sumRows \begin{bmatrix} | & | & | \\ a_1v_1 & a_2v_2 & \cdots & a_nv_n \\ | & | & | \end{bmatrix}$$

Bevis. Lad B være resultatet af at skalere søjlerne i matrix A med de tilsvarende elementer i

vektoren  $\mathbf{v}$ . Vi har

$$B = \begin{bmatrix} | & | & | \\ a_1v_1 & a_2v_2 & \cdots & a_nv_n \\ | & | & | \end{bmatrix}$$

$$= \begin{bmatrix} a_{11}v_1 & a_{12}v_2 & \cdots & a_{1n}v_n \\ a_{21}v_1 & a_{22}v_2 & \cdots & a_{2n}v_n \\ \vdots & \vdots & \ddots & \vdots \\ a_{m1}v_1 & a_{m2}v_2 & \cdots & a_{mn}v_n \end{bmatrix}$$

Ved brug af definition 1 for *sumRows* og definition 2 ses det at

$$sumRows(B) = \begin{bmatrix} \sum_{j=1}^{n} a_{1j}v_j \\ \sum_{j=1}^{n} a_{2j}v_j \\ \vdots \\ \sum_{j=1}^{n} a_{mj}v_j \end{bmatrix} = A\mathbf{v}$$

Vi kan dermed anvende Sætning 1 til at definere en funktion matrixMulVector for matrix-vektor multiplikation (se Listing 20). Som først skalere søjlerne i matricen med de tilsvarende element i vektoren, og derefter summe rækkerne i matricen.

Listing 20: Funktion til matrix-vektor multiplikation

**Defination 3** (Matrix multiplikation <sup>14</sup>).

Lad  $A \in \mathbb{F}^{m \times n}$  og  $B \in \mathbb{F}^{n \times \ell}$ . Antag at søjlerne i B er givet ved  $b_1, \ldots, b_\ell \in \mathbb{F}^n$ , dermed

$$B = \begin{bmatrix} | & & | \\ b_1 & \cdots & b_{\ell} \\ | & & | \end{bmatrix}.$$

Så defineres matrice produktet som

$$A \cdot B = \begin{bmatrix} | & & | \\ A \cdot b_1 & \cdots & A \cdot b_\ell \\ | & & | \end{bmatrix}.$$

Udfra defination 3, ses det at funktion matrixProduct i Listing 21 til at multiplicere to matricer, bliver derfor at lave matrix-vektor multiplikation på hver søjle i matricen. Da matrixMulVector returnere en matrix, skal der bruges en funktion til at konvertere matricen til en vektor, hvilket matrixToVector gør.

 $<sup>^{14}</sup>$  Mathematics 1a.

```
// matrixProduct : Matrix -> Matrix
let rec matrixProduct a (M(vlb, _)) =
let product = List.map (
fun bv -> matrixMulVector a bv |> matrixToVector ) vlb
M(product, C)
```

Listing 21: Funktion til at multiplicere matricer

#### 4.1.1.4 Projektion af en vektor

Som beskrevet i afsnit 2.3 'Projections onto a line' i 'Mathematics 1b' <sup>15</sup>, kan projektionen af en vektor defineres som følgende, hvor  $Y = \text{span}\{y\}$ .

$$\operatorname{proj}_Y: V \to V, \quad \operatorname{proj}_Y(x) = \frac{\langle x, y \rangle}{\langle y, y \rangle} y$$
 (3)

Med det standard indreprodukt

$$\langle x, y \rangle = y^* x = \sum_{k=1}^n x_k \overline{y}_k \tag{4}$$

Den første funktion vi skal bruge er derfor en funktionen til at konjugere en vektor conjugate-Vector, det gøres ved at konjugere elementerne i vektoren. Udover dette defineres en funktion til at multiplikere to vektorer element vis vectorMulElementWise.

Listing 22: Funktioner til projektere en vektor på en anden

Evalueringen af det standard indre produkt innerProduct mellem to vektore, bliver derfor at konjugere den ene vektor og derefter multiplicere elementvis med den anden vektor. Hvortil summen af elementerne i den resulterende vektor er det indre produkt.

Sidst kan funktionen proj skrives direkte som den er defineres i ligning 3.

<sup>&</sup>lt;sup>15</sup>Mathematics 1b - Functions of several variables.

# 4.2 PBT af matrix operationer

Det er nu muligt at opstille nogle PBT af der sikre at matricerne overholder matematiske egenskaber i sætning 2. Først defineres en generator for matricer, som generere matricer med tilfældige tal fra vores talmængde 6.

```
1 // vectorGen : int -> Gen<Vector>
2 let vectorGen n =
      Gen.listOfLength n numberGen |> Gen.map (fun x -> vector x)
5 // matrixGen : Gen<Matrix>
6 let matrixGen =
      gen {
          let! row = Gen.choose(1, 6)
          let! col = Gen.choose(1, 6)
          let! vectors = Gen.listOfLength col (vectorGen row)
10
          return matrix vectors
12
13
  type MaxtrixGen =
14
      static member Matrix() =
1.5
          {new Arbitrary < Matrix > () with
16
               override \_.Generator = matrixGen
17
               override _.Shrinker _ = Seq.empty}
18
19
  type NumberGen =
20
      static member Number() =
21
          {new Arbitrary < Number > () with
22
               override _.Generator = numberGen
23
               override _.Shrinker _ = Seq.empty}
```

Listing 23: Generatorene anvendt til PBT af matrix operationer i

Sætning 2 (Vektor Aksiomer<sup>16</sup>).

Lad  $c, d \in \mathbb{F}$  og  $v_i \in \mathbb{F}^n$  for  $i = 1 \dots m$  så gælder:

3.  $c \cdot (v_1 + \cdots + v_m) = c \cdot v_1 + \cdots + c \cdot v_m$ 

1. 
$$(v_1 + \dots + v_{m-1}) + v_m = v_1 + (v_2 + \dots + v_m)$$
  
2.  $c \cdot \begin{pmatrix} d \cdot \begin{bmatrix} | & | \\ v_1 & \dots & v_m \\ | & | \end{pmatrix} \end{pmatrix} = (c \cdot d) \cdot \begin{bmatrix} | & | \\ v_1 & \dots & v_m \\ | & | \end{pmatrix}$ 

Vi kan dermed nu lave definere egenskaberne fra sætning 2 som nogle funktioner, og teste dem med PBT.

```
vectorCom : Matrix -> bool
let vectorCom m =
    sumRows m = sumRows (flip m)

vectorScalarAss : Matrix -> Number -> bool
```

 $<sup>^{16}</sup>$ theorem 7.2 i mat 1 noterne

Listing 24: Egenskaberne fra sætning 2 som funktioner

```
- Arb.register < MaxtrixGen > ()
- Arb.register < NumberGen > ()
- let _ = Check.Quick vectorCom
- let _ = Check.Quick vectorScalarAss
- let _ = Check.Quick vectorAssCom;;
Ok, passed 100 tests.
Ok, passed 100 tests.
Ok, passed 100 tests.
```

Listing 25: Outputtet fra PBT af vektor Listing 24

#### 4.3 Række-echelon form

#### 4.4 Gram-Schmidt

Vi kan nu betragte implementeringen af Gram-Schmidt processen. Denne proces kan anvendes rekursivt til at finde en ortonormal basis for et underrum udspændt af en liste af vektorer  $v_1, v_2, \ldots, v_n$ . Processen kan implementeres rekursivt idet de nye vektorer  $w_k$  for  $k = 2, 3, \ldots, n$  konstrueres baseret på alle de tidligere vektorer  $w_1, \ldots, w_{k-1}$ .

Før vi implementerer Gram-Schmidt processen, er vi dog begrænset af vores Number type 6, idet  $x \in \{\text{Number}\} \implies \sqrt{x} \in \{\text{Number}\}$ . Derfor vil vi ikke normalisere vektorerne, hvilket medfører, at vi kun vil finde en ortogonal basis, fremfor en ortonormal basis.

```
1 // orthogonalBacis : Matrix -> Matrix
2 let orthogonalBacis m =
      if not < | corectOrderCheck m C
      then orthogonalBacis (correctOrder m C)
      else
      // Gram_Schmidt : Matrix -> (Vector list -> Matrix) -> Matrix
      let rec Gram_Schmidt vm acc_wm =
          match acc_wm [], vm with
9
          | x, M([], _) -> x
10
          | M([], _), M(v1::vrest, o) ->
              Gram_Schmidt (M(vrest,o))
               <| fun x -> extendMatrix (M([v1], C)) x
13
          | M(w, _), M(vk::vrest, o) ->
14
              let (V(wk, _)) = vk - sumProj w vk
              Gram_Schmidt (M(vrest,o))
16
              <| fun x -> extendMatrix (acc_wm wk) x
17
```

```
// sumProj : Vector list -> Vector -> Vector
and sumProj w vk =

List.map (fun x -> proj x vk) w
   |> matrix
   |> sumRows
   |> matrixToVector

Gram_Schmidt m (fun _ -> M([], C))
```

Listing 26: Dannelsen af en ortogonal basis, ved hjælp af Gram-Schmidt processen

Funktionen sumProj tager en liste med vektorer w, som i Gram-Schmidt-processen er de tidligere behandlede vektorer  $w_1, \ldots, w_{k-1}$ , og en vektor  $v_k$  som er den k'te vektor.  $v_k$  projiceres på alle vektorerne i w, hvorefter der tages summen af disse projektioner.

Funktionen Gram\_Schmidt, tager en matrix hvor søjlerne er de vektores som ønskes at finde en ortogonal basis for. Der udover tager den en akkumulerende funktion som indeholder de behandlede vektorer. Hvis der ikke er flere vektorer i matricen, gives den akkumulerede funktion. Hvis der ikke er nogle vektorer i akkumulatoren, tages den første vektor fra matricen og tilføjes til akkumulatoren. Hvis der er vektorer i både akkumulatoren og matricen, kaldes sumProj på den akkumulerede liste og den første vektor i matricen. Resultatet trækkes fra den første vektor i matricen, og dette bliver den nye vektor som tilføjes til akkumulatoren.

Funktionen orthogonalBacis tager en matrix og tjekker om matricen er i kolonnefølge, hvis ikke kalder funktionen sig selv, med den korrekte lagringsordning. Ellers kaldes Gram\_Schmidt med matricen og en tom akkumulator. Resultatet bliver derfor en matrix med en ortogonal basis for underrummet udspændt af de givne vektorer, givet at vektorerne er lineært uafhængige.

#### 4.5 PBT af Gram-Schmidt

Udfrodringen ved at lave en PBT af Gram-Schmidt er at vektorsættet skal være lineært uafhængige. Derfor laves der en generator som ved at udføre tilfældige række operationer på en diagonal matrix, kan generere en matrix med lineært uafhængige vektorer. #TODO: Lav et bevis for det beholder enskaben for lineært uafhængighed.

```
// getBacismatrixGen : int -> Gen<Matrix>
  let getBacismatrixGen n =
      Gen.map (fun x -> standardBacis x) (Gen.choose (2, n))
3
 // performRowOperationGen : Matrix -> Gen<Matrix>
  let performRowOperationGen m =
      let (D(n, _)) = dimMatrix m
      gen {
          let! i = Gen.choose(1, n)
          let! j = match i with
10
               | 1 -> Gen.choose(2, n)
11
                _ when i = n \rightarrow Gen.choose(1, n-1)
                   -> Gen.oneof [
                       Gen.choose(1, i-1);
14
                       Gen.choose(i+1, n)]
          let! a = numberGen
16
```

```
return rowOperation i j a m }
18
19
20 // multipleRowOperationsGen : Matrix -> int -> Gen<Matrix>
 let rec multipleRowOperationsGen m count =
      if count <= 0 then Gen.constant m</pre>
      else
23
          gen {
               let! newMatrix = performRowOperationGen m
               return! multipleRowOperationsGen newMatrix (count - 1)
26
          }
27
28
  // getIndependetBacisGen : Gen<Matrix>
  let getIndependetBacisGen =
      gen {
31
          let! m = getBacismatrixGen 5
          let! numberOfOperations = Gen.choose(1, 10)
33
          let! span = multipleRowOperationsGen m numberOfOperations
34
          return span }
35
36
37 type IndependetBacis = Matrix
  type IndependetBacisGen =
38
      static member IndependetBacis() =
39
          {new Arbitrary < Matrix > () with
40
               override _.Generator = getIndependetBacisGen
41
               override _.Shrinker _ = Seq.empty}
```

Listing 27: Generatorene anvendt til PBT af Gram-Schmidt

Listing 27 viser de forskellige generatorer, som anvendes til PBT (Property-Based Testing) af Gram-Schmidt-processen. Først genereres en tilfældig basis matrix. Dernæst udvælges to tilfældige rækker, i og j, hvorefter der udføres en rækkeoperation på  $R_j$ , således at  $R_j \leftarrow R_j - aR_i$ , hvor a er et tilfældigt Number. Denne proces gentages et tilfældigt antal gange.

Dernæst skal vi bruge en funktion til at tjekke om en matrix er en ortogonal basis. isOrthogonalBacis i Listing 28 tjekker om alle vektorerne i en matrix er ortogonale, ved at tjekke om søjle  $v_i$  er ortogonal med  $v_{i+1}$ , for alle  $i \in [1, n-1]$  hvor n er længden på søjlerne. To søjler er ortogonale hvis deres indreprodukt er 0.

```
// isOrthogonalBacis : Matrix -> bool
let rec isOrthogonalBacis (M(vl, o)) =
    if not <| corectOrderCheck (M(vl, o)) C
    then isOrthogonalBacis <| correctOrder (M(vl, o)) C
    else
    match vl with
    | [] -> true
    | _::[] -> true
    | v::vnext::vrest -> innerProduct v vnext = zero &&
    isOrthogonalBacis (M(vnext::vrest, o))
```

Listing 28: Funktion til at tjekke om søjlerne i en matrix er en ortogonal basis

PB testen gramSchmidtIsOrthogonal bliver derfor blot at tjekke om en matrix bestående af lineært uafhængige vektorer, der udspænder et underrum, er orthogonale efter Gram-Schmidt processen er blevet anvendt. Grundet tilfældige matematiske operationer, opstår der en støre mængde opstå overflow fejl, derfor godtages disse men klassificeres som overflow.

Listing 29: PBT af Gram-Schmidt processen

```
- Arb.register < IndependetBacisGen > ()
- let _ = Check.Quick gramSchmidtIsOrthogonal;;
Ok, passed 100 tests.
69% PropertyHolds.
31% OverflowException.
```

Listing 30: Output fra PBT af Gram-Schmidt processen

Outputtet fra PBT af Gram-Schmidt processen kan ses i Listing 30. Som sædvanligt indikere testen kun korrekthed, men ikke garanteret korrekthed.

# 5 Appendiks

#### 5.1 Matrix.fs

# Litteratur

- [1] Floating-point error. URL: https://en.wikipedia.org/wiki/Floating-point\_error\_mitigation (hentet 30.03.2024).
- [2] fsharp.org. URL: https://fsharp.org/ (hentet 30.03.2024).
- [3] Greatest common divisor wikipedia. URL: https://en.wikipedia.org/wiki/Greatest\_common\_divisor (hentet 30.03.2024).
- [4] Michael R. Hansen og Hans Rischel. Functional Programming Using F#. Cambridge University Press, 2013. ISBN: 9781107019027, 1107019028.
- [5] Ole Christensen og Jakob Lemvig. *Mathematics 1b Functions of several variables*. URL: https://01002.compute.dtu.dk/\_assets/notesvol2.pdf (hentet 09.02.2024).
- [6] Kursus side for Mathematics 1a. URL: https://matla.compute.dtu.dk/intro.html (hentet 30.03.2024).
- [7] Mathematics 1a. URL: https://01001.compute.dtu.dk/\_assets/enotesvol1.pdf (hentet 09.02.2024).
- [8] Polsk notation wikipedia. URL: https://en.wikipedia.org/wiki/Polish\_notation.
- [9] Python List Comprehension. URL: https://www.w3schools.com/python/python\_lists\_comprehension.asp (hentet 31.03.2024).
- [10] Rational Number wikipedia. URL: https://en.wikipedia.org/wiki/Rational\_number (hentet 30.03.2024).
- [11] Row- and column-major order. URL: https://en.wikipedia.org/wiki/Row-\_and\_column-major\_order (hentet 31.03.2024).
- [12] Statista. Most used programming languages among developers worldwide as of 2023. URL: https://www.statista.com/statistics/793628/worldwide-developer-survey-most-used-languages/(hentet 30.03.2024).
- [13] SymPy. URL: https://www.sympy.org/en/index.html (hentet 30.03.2024).
- [14] Test-Driven Development. URL: https://en.wikipedia.org/wiki/Test-driven\_development (hentet 31.03.2024).
- [15] Tree Traversal Techniques Data Structure and Algorithm Tutorials. URL: https://www.geeksforgeeks.org/tree-traversals-inorder-preorder-and-postorder/ (hentet 30.03.2024).